

TEMPEST



TEMPEST

THE MAKING AND REMAKING
OF ATARI'S ICONIC VIDEOGAME

TEMPEST
VS
TEMPEST

For Edna.

© Rob Hogan 2025, All Rights Reserved.

Edition Date: Saturday 24th May, 2025

This work is licensed under a Creative Commons
"Attribution-NonCommercial-ShareAlike 3.0 Unported"
license.




Contents

how to make tempest in 1981	9
building tempest 2000 in 1994	15
mainline	21
mainloop	29
cry if i want to	41
what is blitting	47
character assassination	55
my first shader	63
unused stars	69
flipper	77
wells	83

webs	87
object list	93
cursors	97
claws	103
more claws	107
meltovision	113
bullet	123
activeobjects	129
rotary club	137
things hidden	143
space game reenactment society	149
sexy yes	155
story of a beep	163
unused explosions	167
play us a tune pal	169
ntsc pal	177

approaching logo process **181**

player charges **191**



Artefacts from Tempest (1981).

about tempest & tempest 2000

Tempest was a coin-operated arcade game released in 1981 by Atari Corporation. At the time, its graphics and gameplay were at the very edge of what was possible with contemporary hardware. Its unique appearance was in part attributable to its use of vector graphics, a technology that drew lines of color, rather than pixels, on the screen and enabled the earliest 3D graphics in the arcade games of the time.

Some 15 years later, its successor *Tempest 2000* was likewise a classic game. To its misfortune it was made for a console that crashed and burned with no survivors, and very few customers. The Jaguar was supposed to be Atari Corporation's first great leap into the 64 bit era. Instead it promptly tripped over its shoelaces and left a number of great games, *Tempest 2000* among them, behind as orphans.

This book isn't about the history of *Tempest* or its importance in the history of computer games. This is a book about computer code. Each chapter aims to be an easily digested sketch of how a specific feature or effect was achieved. In the case of *Tempest* this means sifting through the 6502 assembly code David Theurer wrote by hand(!) to create the game, and wrangling with the bits and bytes of Atari's Quadrascan Vector Graphics. For *Tempest 2000* it means familiarizing ourselves with Motorola 68K assembly code and the Jaguar's powerful hardware platform.

There are lots of fascinating bits and pieces to pore over. And it is interesting to see how the years that separate the two games leave some things unchanged and many other things changed utterly. I've put together this book of squibs in the hope they will allow you, the reader, to digest small bite-sized insights into each game. And I hope, by interleaving elements from *Tempest* and *Tempest 2000*, to enable you to compare the parallel lives in the design and execution of separate incarnations of the same classic video game.

Rob Hogan (@mwenge)
2025



Artefacts from Tempest 2000 (1994).

**TEMPEST
VS
TEMPEST**

how to make tempest in 1981

It is 1981 and Atari Corporation has tasked you with writing a state-of-the-art video game. To complete your assignment they have equipped you with a pencil and a bundle of sheets. You wonder where the computer is, and at what point they will give you one.

You can sit in your office and wonder in vain, because they are never going to give you one. You must devise and program the game entirely with pencil and paper. There is only one computer in the building, and it is the price of a house. They are not going to let you anywhere near that computer. You are in building 1272 and the computer is in building 1360. You are not even on the same block as the computer.




Figure .1: The computer is in another building.

Instead you are confronted with a ream of blank sheets such as this one; into which you must pencil the assembly language instructions that will make your game happen.

Assembler Coding Form												PAGE OF		CARD ELECTRO NUMBER				
PROGRAM: HELOWRLD			PROGRAMMER			DATE			PUNCHING INSTRUCTIONS			GRAPHIC						
STATEMENT												PUNCH						
1	Name	10	Dimension	14	18	20	25	30	35	40	45	50	55	60	65	70	75	Identification Sequence
																		80
HELOWRLD	CSECT																	
	STM	14,12,12(13)																
	LR	12,15																
	USING	HELLOWRLD,12																
	LR	14,13																
	LA	13,SAVEAREA																
	ST	13,8(14)																
	ST	14,4(13)																
	OPEN	(SYSPRINT,(OUTPRT))																
	LA	2,=CL12'HELLO WORLD'																
	PUT	SYSPRINT,(2)																
	CLOSE	SYSPRINT																
	XR	15,15																
	L	13,4(13)																
	L	14,12(13)																
	LM	0,12,20(13)																
	BR	14																
	SAVEAREA	DS	18F															
	SYSPRINT	DCB	DDNAME=SYSPRINT, DSORG=PS, MACRF=PM, LRECL=121, RECFM=FBA															
		LTORG	1															
		END																

*A punched card form, IBM Series 400, is available for punched source programs. Write IBM Systems Division, Dept. M-100, Indianapolis, Indiana 46285 or to any IBM Systems/360 assembler language reference manual. Address comments concerning this form to IBM Corporation, P.O. Box 8000, Programming Publishing, San Jose, California, U.S.A. 95108.

*Max. of lines per card may be 128

Figure .2: An example of a coding sheet to give you an idea of what programming was like in 1981.

If you fill out enough of these sheets with 6502 assembly you have a program that can be typed out and assembled.

dave says



I'd write the code on programming sheets and turn it in to the typists who'd type them in to a DEC computer, then give us a tape with the resulting compiled/linked program.

Once the typists had entered your program into the PDP-11 in building 1360 the would see a listing of the files such as this on a VT-100 terminal.

```
.DIR RK1:

ALVGUT.MAC      19          ALCOIN.MAC      1
HLL65 .MAC       4           ALHARD.MAC     7
ALSOUN.MAC      17          ALWELG.MAC    129
ALHAR2.MAC      7           ALDISP.MAC    111
ALSC02.MAC      50          ALDIS2.MAC    111
ALSCOR.MAC      50          ALCOMN.MAC    52
ALVROM.MAC      77          ALLANG.MAC    14
ALTES2.MAC      34          ALEARO.MAC    12
VGMC .MAC       8           STATE2.MAC    2
ANVGAN.MAC     12          ALEXEC.MAC   24
ALDIAG.MAC      6           COIN65.MAC   47
ALTEST.MAC     32          ASCVG .MAC    2
24 Files, 828 Blocks
5200 Free blocks
.
```

Figure .3: Listing the Tempest source files.

Their next step was to assemble each of the files using the MAC65 command on each:

.R MAC65 RK1:ALWELG=ALWELG *ERRORS DETECTED: 0 FREE CORE: 11479. WORDS	.R MAC65 RK1:ALCOIN=ALCOIN *ERRORS DETECTED: 0 FREE CORE: 11118. WORDS
RK1:ALSC02=ALSC02 *ERRORS DETECTED: 0 FREE CORE: 12479. WORDS	RK1:ALHAR2=ALHAR2 *ERRORS DETECTED: 0 FREE CORE: 11892. WORDS
RK1:ALDIS2=ALDIS2 *ERRORS DETECTED: 0 FREE CORE: 11890. WORDS	RK1:ALLANG=ALLANG *ERRORS DETECTED: 0 FREE CORE: 13186. WORDS
RK1:ALEXEC=ALEXEC *ERRORS DETECTED: 0 FREE CORE: 13003. WORDS	RK1:ALTES2=ALTES2 *ERRORS DETECTED: 0 FREE CORE: 12298. WORDS
RK1:ALSOUN=ALSOUN *ERRORS DETECTED: 0 FREE CORE: 12597. WORDS	RK1:ALEARO=ALEARO *ERRORS DETECTED: 0 FREE CORE: 13010. WORDS
RK1:ALVROM=ALVROM *ERRORS DETECTED: 0 FREE CORE: 12599. WORDS	RK1:ALVGUT=ALVGUT *ERRORS DETECTED: 0 FREE CORE: 13178. WORDS

Figure .4: Assembling the Tempest source files.

Once that was done they could run the RMAC command to link the assembled files into a single object file ALEXEC.LDA.

```
.R LINKM
*RK1:ALEXEC/L,ALEXEC/A=RK1:ALWELG,ALSC02,ALDIS2,ALEXEC,ALSOUN,ALVROM/C
*ALCOIN,ALLANG,ALHAR2,ALTES2,ALEARD,ALVGUT
MULT DEF OF VGBRIT IN MODULE: 000C
MULT DEF OF VGLIST IN MODULE: 000C
MULT DEF OF XCOMP IN MODULE: 000C
```

Figure .5: Linking the Tempest object files.

dave says



We'd then take that to a blue box [which used the FORTH programming language] for debugging. We'd mark changes to the code on a listing, give it back to the typists who'd edit our files and give us a new tape. Repeat ad infinitum.

Once you have completed the ad-infinitum phase of this process you may have a game that is ready for production and distribution. For this you must talk to your project manager, Morgan, and together decide how to split the ALEXEC.LDA file into a number of different files. Each of these will be written to a specific ROM chip as detailed by the parts manifest you devise together and distribute to the guys in building 1320:

DATE 17-12-1981 16 23 14 USER THEURER JOB TEMPEST						PAGE 0001
PROJECT # 28903			PROJECT PROGRAMMER DAVE THEURER		PROJECT LEADER MORGAN HOFF	
DOCUMENTATION DISK S 36A,36B,36C			PROJECT NAME TEMPEST		DATE 12-17-81	VERSION 2A ALT
VERIFICATION CONTROL FILENAME 002X2.DAT VERIFICATION FILE ALEXEC.LDA						
PART #	PCR LOC.	DESCRIPTION	START ADDR	GENERIC NAME	PART SIZE	BITS USED
136002-138	N/P3	VG EROM	3000	2532	4XX8	7- 0
136002-237	P1	PROGRAM EROM	9000	2532	4XX8	7- 0
136002-136	L/M1	PROGRAM EROM	4000	2532	4XX8	7- 0
136002-235	J1	PROGRAM EROM	8000	2532	4XX8	7- 0
136002-134	F1	PROGRAM EROM	C000	2532	4XX8	7- 0
136002-133	D1	PROGRAM EROM	0000	2532	4XX8	7- 0
IND. SELF TEST						
GENERAL INFORMATION						
ASSEMBLER USED MAC65						
MAIN PROGRAM SOURCE FILES INCLUDES ALCOMM, HLL65, COIN65, VGM65, ANVGAN						
FILES ALWELG,ALSC02,ALDIS2,ALEXEC,ALSOUN,ALVROM,ALCOIN,ALLANG,ALHAR2,ALTES2,ALEARD,ALVGUT						
PICTURE FILE						
LINKER USED LINKM						
LINKER COMMAND ALEXEC/L,ALEXEC/A ALWELG,ALSC02,ALDIS2,ALEXEC,ALSOUN,ALVROM,ALCOIN,ALLANG,ALHAR2,ALTES2,ALEARD,ALVGUT						
COMMAND FILES						
DOCUMENTATION FILE TEMPST.DOC						

Figure .6: The actual build manifest for Tempest, recreated in the style of the time.

You might expect that splitting ALEXEC.LDA is as simple as carving up its raw contents into these component files. You would be wrong. ALEXEC.LDA actually consists of a long sequence of triplets, each giving you the ROM address a set of bytes are to be written to, the number of bytes, followed by the bytes themselves. This table shows the first ten or so triplets to give you an idea of what we mean. ALEXEC.LDA contains 2000 such triplets in total!

Address	Length	Data
9000	38	02BB5A3050EE3DA84D20C592203492202B902031A8A9FA855BA9008D0601855FA90085016020
9026	38	1B9220C592208F92206F92204692209F9220AD92206EC1A9FF8D24018D4801A9008D230160A9
904C	19	108D0202A9008529852BAD2101852A1000C62B
905C	1	02
905F	38	A201A52A0A662A6629CA10F6A529186D2201A52A65688568A52B65698569A55F186918
9085	17	855FA55B6900855BC9FC9000A9018D1501
9090	1	05
9096	11	A55F38E55DA55BF000E9FF
909E	1	02
90A1	18	D000A55D855FA9FF855BA90424053000A908
90B0	1	02
90B3	9	8500A63DA9009D0201
90A2	1	19
90BC	36	A9FF8D14014C4997AD2601A21CCADDDE9190FAA004AD6A012904F000AD1D07C9309000C8
90DE	1	01
90E0	5	C9509000C8
90E3	1	01
90E5	5	C9709000C8
90E8	1	01
90D7	1	12
90EA	10	A5092943C940D000A01B
90F1	1	02
90F4	8	8429E429B000A629
90F9	1	02
90FC	12	8E2701A5051000A9008D2601
9102	1	05
9108	9	A63F863DF00020B292
910D	1	03
9111	38	A904857CA9FF855BA9008D00028551857B8D0506A6051000A9148D0506A9FF8D1101A9168500
9137	13	A9088501A900859F2096C1A910
9128	1	1B

The triplets in ALEXEC.LDA for ROM address 9000 written to ROM Part 136002-237.

Notice how apparently wasteful this file format is: some of the triplets contain only byte! You can also notice that the triplets can appear out of order. For example the bytes for address 90A2 are specified after the bytes for 90A1. So perhaps for some binaries this encoding scheme may be more efficient than just raw bytes, but not in the case of ALEXEC.LDA which uses 25,408 bytes to encode a total of 24,576 bytes that actually get written to the ROM parts on the circuit board.

Which leads nicely to our final diagram. If you look closely in the circuit board diagram on the following page you will find the final destination for these files in the printed circuit board shipped in the arcade cabinet. They are the parts indicated by stars at

the top of the board.




Figure 23 Tempest™ Analog Vector-Generator PCB Assembly
A037383-01 thru -04 C

Figure .7: The circuit board diagram containing the ROM chips that our ROM files are written to..

building tempest 2000 in 1994

It is 1994 and Atari have asked you create the flagship launch title for their new video game console, the Atari Jaguar. No need to come to Sunnyvale in California, you can stay in your valley in Wales and let us know when you're finished.



Figure .1: You are not in California, you are in Cwmcych in Wales and that's where you'll make Tempest 2000.

So instead of sitting two doors down from the computer room armed with just a pencil, you are instead expected to conjure a game that will launch a machine on which the future of Atari depends from your cottage. To get you started you will be sent a console

and a manual. You already have a computer we assume.

So because it's the way you've always done it, you get started writing Tempest 2000 in one big file of Motoral 68K assembly code. And because you always get to name things whatever you want, you name this file YAK.S, because that's the three-letter name you use on high score tables.



Figure .2: YAK has the top score in Tempest.

Assembling this code will be a simple matter of running the `madmac` assembler as follows:

```
$mac -fb -isrc -l src/yak.s -o yak.dat
```

We're not done yet though. In addition to a cryptic name for the main megafile (YAK.S will eventually run to almost 20,000 lines of code), we also need to write a bunch of programs in a custom flavour of 68K Motorola assembly for the Jaguar's GPU. These programs will handle all the hard mathematics for generating polygons in three dimensions. Because no-one else will ever need to read this code we can name them whatever we want. So we'll continue the zoological theme and name them after more beasts of the field. There will be nine GPU programs in total, each doing something specific (for example, scaling and rotating objects in `antelope.gas`) and some of them doing the same things but slightly differently (variations of a Robotron-style particle explosion in `camel.gas` and `xcamel.gas`).

This is roughly what it looks like when we assemble each of our GPU programs.

```
$ mac -fr -mtom -l -isrc donky.gas -o donky.dat  
$ mac -fr -mtom -l -isrc camel.gas -o camel.dat
```

```
$ mac -fr -mtom -l -isrc antelope.gas -o antelope.dat
$ mac -fr -mtom -l -isrc goat.gas -o goat.dat
$ mac -fr -mtom -l -isrc llama.gas -o llama.dat
$ mac -fr -mtom -l -isrc horse.gas -o horse.dat
$ mac -fr -mtom -l -isrc ox.gas -o ox.dat
$ mac -fr -mtom -l -isrc stoat.gas -o stoat.dat
$ mac -fr -mtom -l -isrc xcamel.gas -o xcamel.dat
```

Now is a good time to be dealing with less objects, so let's create another file called YAKGPU.S that collates all our GPU programs into one big GPU object. Here's how we do that in YAKGPU.S:

```
; ; CONSTANT DATA (GPU PROGRAMS)
;
fastvector:
    .include "llama.dat"
xvector:
    .include "goat.dat"
demons:
    .include "antelope.dat"
parrot:
    .include "camel.dat"
xparrot:
    .include "xcamel.dat"
texter:
    .include "stoat.dat"
bovine:
    .include "ox.dat"
equine:
    .include "horse.dat"
equine2:
    .include "donky.dat"
```

And here's how we assemble YAKGPU.S:

```
$ mac -fb -isrc yakgpu.s -o yakgpu.dat
```

We have one more object file to pull together. An omnibus containing all our image and sound data. For this we will create a file called IMAGES_SOUNDS.S. At the top we will put all our image files:

```
.incbin "beasty3.cry"
.incbin "beasty4.cry"
.incbin "beasty5.cry"
.incbin "beasty6.cry"
.incbin "beasty7.cry"
.incbin "beasty8.cry"
```

We are still in a proverbial farmyard with our naming conventions. Let's take a look at what's actually in these files:



beasty3.cry



beasty5.cry



beasty5.cry



beasty6.cry



beasty7.cry



beasty8.cry

Apart from attractive image data we also have some music to play. There are seven

tunes for us to include. These are:

```
.incbin "sounds/tune13.mod"
.DC.L $0000
.incbin "sounds/tune7.mod"
.DC.L $0000
.incbin "sounds/tune1.mod"
.DC.L $0000
.incbin "sounds/tune3.mod"
.DC.L $0000
.incbin "sounds/rave4.mod"
.DC.L $0000
.incbin "sounds/tune5.mod"
.DC.L $0000
.incbin "sounds/tune12.mod"
.DC.L $0000
```

We also have a whole bunch of sound effects. But we'll get into those later. For now, let's assemble our audiovisual bag of tricks into a single object file:

```
$ mac -fb -isrc images_sounds.s -o images_sounds.dat
```

This leaves us with just one final step to perform: linking all of our products, YAK.DAT, IMAGES_SOUNDS.DAT, and YAKGPU.DAT together to give us a game we can actually play. For that we use Atari's purpose built linker aln:

```
$ aln -z -e -a 802000 4000 efa8 yak.dat vidinit.dat yakgpu.dat images_sounds
.dat -o t2000.abs
```


mainline

Tempest's 'mainline' is a short routine that runs the entirety of the game from a handful of lines of assembly code. Beneath lies an iceberg of strange machinery but at its core the game consists of running these 15 or so lines many hundreds of times per second:

```
; INPUT:           POWER ON RESET PREPARATION
; OUTPUT:          NONE

MAINLN: JSR INISOU           ; INITIALIZE SOUNDS
          LDA I,CNEWGA
          STA QSTATE
          BEGIN             ; MAINLOOP
          BEGIN             ; LOOP UNTIL CURRENT FRAME HAS BEEN UP X MS.
          LDA FRTIMR
          CMP I,9
          CSEND
          LDA I,0            ; RESTART FRAME TIMER
          STA FRTIMR
          JSR EXSTAT          ; EXECUTE APPROPRIATE GAME STATE
          JSR NONSTA          ; EXECUTE NON-STATE DEPENDENT CODE
          JSR DISPLA          ; EXECUTE CODE TO DISPLAY NEW SCREEN
          CLC
          CSEND             ; LOOP ALWAYS
```

The game's framerate is controlled by this tight little loop. As soon as FRTIMR reaches 9 a frame of the game is executed:

```
LDA FRTIMR
CMP I,9
CSEND
```

tempest

```

DATE 17-04-1981 19 00 12   USER THEURER   JOB TEMPEST          PAGE 0004
1 ALEXEC - ALIENS EXECUTIVE ATARI MAC65 VM03.09 00 00 02 PAGE 17
2 MAINLOOP
3
4
5
6     0000 20 0006  MAINLN JSR IN100U ;INITIALIZE SOUNDS
7     0003 A9 00      LDA I,CHEWGA
8     0005 85 00      STA QSTATE
9     0007 A5 53      BEGIN ;MAINLOOP
10    0009 C9 09      LDY FRTIMR
11    000B 90 FA      CSEND
12    000D A9 00      LDA I,O
13    000F 85 53      STA FRTIMR ;RESTART FRAME TIMER
14    0011 20 001D      JSR EXSTAT ;EXECUTE APPROPRIATE GAME STATE
15    0014 20 00F1      JSR XINSTA ;EXECUTE NON-STATE DEPENDENT CODE
16    0017 20 0006      JSR DISPLAY ;EXECUTE CODE TO DISPLAY NEW SCREEN
17
18    001A 18          CLC
19    001B 90 EA      CSEND ;LOOP ALWAYS
20
21
22
23
24
25
26
27
28
29
30
31
32
33
34
35
36
37
38
39
40
41
42
43
44

```

Figure .1: The MAINLN routine in an assembly listing.

```

DATE 17-04-1981 19 00 12   USER THEURER   JOB TEMPEST          PAGE 0005
1 ALEXEC - ALIENS EXECUTIVE ATARI MAC65 VM03.09 00 00 02 PAGE 18
2 STATE ROUTINE EXECUTOR
3
4
5
6     001D          EXSTAT
7     0020 AD 0000      LDA INPO
8     0020 29 83      AND I,83
9     0022 C9 82      CMP I,82
10    0024 F0 00      IFNE PSTART ;FREEZE FREE PLAY
11    0024 20 0006      LDX PSTART
12    0029 46 00      LDX QSTATE
13    002B A5 4E      LDA SWFINA ;SET MUST PROCESS FLAG
14    002D 09 80      DRA I,MFAKE
15    002F 85 4E      STA SWFINA
16    0031 00 003B      PHA AX,ROUTAD+1
17    0034 49          PHA
18    0035 BD 003A      LDA AX,ROUTAD
19    0038 49          PHA
20    0025 13          ENDIF
21    0039 60          NOOPR RTS
22
23
24
25    003A 016B      ROUTAD ;STATE ROUTINE ADDRESS
26    003C 019F      WORD NEWGAM=1 ;NEW GAME
27    0151 FFFF      WORD NFWLIF=1 ;NEW LIFE AFTER LOSING A BASE
28    0040 020E      WORD ENDLIF=1 ;PLAY
29    0042 0250      WORD ENDGM=1 ;LIFE LOST
30    0044 005F      WORD PAUSE=1 ;END OF GAME
31    0046 0000      WORD ENDWV=1 ;PAUSE
32    0048 016B      WORD ENDWAV=1 ;NEW WAVE AFTER SHOOTING ALL INVADERS
33    004A FFFF      WORD HISCHK=1 ;END WAVE
34    004C FFFF66      WORD GETINI=1 ;CHECK FOR HI SCORES
35    004E 0277      WORD GETINI=1 ;GET HI SCORE INITIALS
36    0050 FFFF      WORD DLADR=1 ;DISPLAY HI SCORE TABLE
37    0052 FFFF       WORD PRORAT=1 ;REQUEST PLAYER RATE
38    0054 FFFF       WORD NEWWAV2=1 ;NEW WAVE PART 2
39    0056 FFFF       WORD LOWDW=1 ;DROP INVADERS
40    0058 01DA      WORD INIRAT=1 ;MONSTER DELAY/DISPLAY
41    005A FFFF       WORD PLDROP=1 ;DROP MODE
42    005C FFFF       WORD SYSTEM=1 ;END WAVE CLEAN UP AFTER BONUS
43    005E FFFF       WORD PRBDM=1 ;BOOM
44    0060 ROUTEN

```

Figure .2: The EXSTAT and ROUTEAD routines in an assembly listing.

Then it is reset to zero:

```
LDA I,0 ; RESTART FRAME TIMER
STA FRTIMR
```

Increasing the value from 9 to 0F (15) will slow the game down appreciably, while increasing it to 1F (31) will bring it to an unplayable crawl. Lowering it to 0 or 1 does make the game slightly faster, but not noticeably and it is still playable.

claw says



But what increments FRTIMR you ask? The answer is a piece of code in what is known as an 'interrupt handler'. This runs separately from the main loop and is invoked by the CPU up to 60 times a second.

Once we've decided to execute a frame we are down to just three lines orchestrating the entirety of the game:

```
JSR EXSTAT ; EXECUTE APPROPRIATE GAME STATE
JSR NONSTA ; EXECUTE NON-STATE DEPENDENT CODE
JSR DISPLAY ; EXECUTE CODE TO DISPLAY NEW SCREEN
```

The first routine we call is EXSTAT. It always updates the starfield (PRSTAR) but what it does next is dependent on the value contained in QSTATE. Take a brief look at the routine to see if you can identify how it uses QSTATE:

```
; INPUT:           QSTATE: CODE FOR STATE ROUTINE TO EXECUTE
; OUTPUT:          CONTROL PASSED TO ROUTINE

EXSTAT:
    LDA INOPO
    AND I,83
    CMP I,82
    IFNE             ; FREEZE & FREE PLAY?
    JSR PRSTAR       ; PROCESS STAR FIELD
    LDX QSTATE
    LDA SWFINA       ; SET MUST PROCESS FLAG
    ORA I,MFAKE
    STA SWFINA
    LDA AX,ROUTAD+1
    PHA
    LDA AX,ROUTAD
    PHA
    ENDIF
NOOPR:   RTS
```

The answer is in these lines:

```

LDX QSTATE
.
LDA AX,ROUTAD+1
PHA
LDA AX,ROUTAD
PHA

```

The value in QSTATE is loaded to the X register. This is then used as an index to retrieve the two-byte address stored in the Xth - 1 index of the ROUTAD table:

```

;
; STATE ROUTINE ADDRESS
;

ROUTAD: .WORD NEWGAM-1           ; NEW GAME
        .WORD NEWLIF-1          ; NEW LIFE (AFTER LOSING A BASE)
        .WORD PLAY-1            ; PLAY
        .WORD ENDLIF-1          ; LIFE LOST
        .WORD ENDGAM-1          ; END OF GAME
        .WORD PAUSE-1           ; PAUSE
        .WORD 0                  ; NEW WAVE (AFTER SHOOTING ALL INVADERS)
        .WORD ENDWAV-1           ; END OF WAVE
        .WORD HISCHK-1          ; CHECK FOR HI SCORES
        .WORD GETINI-1           ; GET HI SCORE INITIALS
        .WORD DLADR-1            ; DISPLAY HI SCORE TABLE
        .WORD PRORAT-1           ; REQUEST PLAYER RATE
        .WORD NEWAV2-1           ; NEW WAVE PART 2
        .WORD LOGINI-1           ; LOGO INIT
        .WORD INIRAT-1           ; MONSTER DELAY/DISPLAY
        .WORD NEWLF2-1           ; NEW LIFE PART 2
        .WORD PLDROP-1           ; DROP MODE
        .WORD SYSTEM-1            ; END WAVE CLEAN UP AFTER BONUS
        .WORD PRBOOM-1           ; BOOM

```

Pushing each of the two bytes to the stack (using PHA) means that whatever is at the address of those two bytes will be executed next. For example, if the value in QSTATE is 0 it will retrieve the address of NEWGAM-1: which is 016B, made up of 01 which it retrieves first, and 06 which it retrieves second. When both of these bytes are pushed to the stack they are treated as an address for execution: the result is that the EXTSTAT routine will execute NEWGAM once it exits, beginning a new game.

So depending on the current state of the game, Tempest will populate QSTATE with the index for the function it needs to call next. To give a flavour of what NEWGAM looks like we've listed it below. As expected it mainly consists of accounting operations such as clearing the score, assigning the number of lives and so on.

```

NEVGAM: JSR INICHK             ; INITIALIZE LANGUAGE PTRS, OPTIONS
        JSR INIDSP              ; INITIALIZE DISPLAY
        LDA QSTATUS
        IFMI                     ; ATTRACT?

```

```

JSR CLRSCO          ;NO. CLEAR SCORES
ENDIF
LDA I,0
STA LIVES2          ;ONE PLAYER GAME (DEFAULT: PLAYER 2 DEAD)
LDX NUMPLA          ;GIVE EACH PLAYER "NEW GAME" EQUIP
STX PLAYUP
BEGIN               ;LOOP FOR EACH PLAYER IN GAME (1 OR 2)
LDX PLAYUP
LDA LVSGAM          ;GET # LIVES
STA AX,LIVES1       ;INITIAL # OF LIVES (GUNS)
LDA I,-1
STA AX,WAVEN1       ;FORCE REQUEST RATE STATE
DEC PLAYUP
MIEND               ;ENDLOOP AFTER ALL PLAYERS PROCESSED
LDA I,0
STA NEWPLA          ;START GAME WITH 1ST PLAYER UP.
STA PLAGRO          ;DEACTIVATE STAR FIELD
LDA NUMPLA          ;INDUCE "PLAY PLAYER 1" MESSAGE
STA PLAYUP          ;IF 2 PLAYER GAME.
JMP INIRAO          ;INITIALIZE FOR PLAYER RATE REQUEST

```

You will notice at the end that we jump to a routine called INIRAO. This continues the accounting set-up for a new game but it also does something important, which is update QSTATE to direct the flow the next time it passes through MAINLN to a different function. In this case, it loads 16 to QSTATE, which will direct us to the routine PRORAT at the next iteration:

LDA I,CREQRAT	;GO TO REQUEST
STA QSTATE	;RATE STATE

Another way of describing what we have here is a 'state machine'. As the state of the system changes, EXSTAT will execute a different routine, appropriate to the current state of the game. The one executed most often is the third one listed in our ROUTEAD table. This is called PLAY and its contents, as well as its name, are almost self-explanatory:

PLAY:	
JSR MOVCUR	;MOVE CURSOR AROUND
JSR FIREPC	;FIRE PLAYER CHARGE
JSR PROSUZ	;PROCESS SUPER ZAP
JSR MOVNYM	;MOVE NYMPHS
JSR MOVINV	;MOVE INVADERS
JSR MOVCHA	;MOVE CHARGES
JSR FIREIC	;FIRE INVADER CHARGE
JSR COLLIS	;COLLISION DETECT
JSR PROEXP	;EXPLOSIONS
JMP ANALYZ	;ANALYZE PLAYER STATUS

Here every element of the game is updated and manipulated in some way. It is not

surprising that somewhere in the heart of a game's runtime the logic should ultimately settle on a list of items that need to be checked, one after the other. One important thing to note here is that we're not updating how any of the game elements are displayed. We are just updating their state, which we store separately from the details of how they are displayed. We will use this updated state to determine their display in a moment, but for now we are just updating our record of their position and condition. The second thing to note is that the order in which these operations are performed is not arbitrary. First we update the player and any bullets the player has fired. Then we move the enemies and any bullets they have fired. Now we can check if the player has collided with any of the enemies. If there has been a collision between any two objects we process an explosion and finally update the player's state to determine, for example, if they are dead. The purpose of this order is to ensure that information about the player and its enemies are updated before we detect to attempt any collisions.

A lot has happened underneath the single line..

JSR EXSTAT	; EXECUTE APPROPRIATE GAME STATE
------------	----------------------------------

.. however now that we have all the properties of the game in its current frame we can move on to updating how they are displayed:

JSR DISPLA	; EXECUTE CODE TO DISPLAY NEW SCREEN
------------	--------------------------------------

Again this will in most cases boil down to updating all of the elements one by one. Unlike when we were concerned with updating state, the order in which we do things is not particularly important here. The DISPLA routine wants to build a list of lines to paint on the screen so it will compute the lines for each type of object currently displayed in a more or less arbitrary sequence determined by the programmer. Notice that in the sequence below the display of each type of object follows a fixed pattern:

LDA I,BCCURS JSR SBCLOG JSR DSPCUR LDA I,BCCURS JSR SBCSWI LDA I,BCSHOT JSR SBCLOG JSR DSPCHG LDA I,BCSHOT JSR SBCSWI LDA I,BCINVVA JSR SBCLOG JSR DSPINV LDA I,BCINVVA	; DISPLAY CURSOR (PLAYER) ;DISPLAY CHARGES ;DISPLAY INVADERS
--	--

```

JSR SBCSWI
                ;DISPLAY EXPLOSIONS
LDA I,BCEXPL
JSR SBCLOG
JSR DSPEXP
LDA I,BCEXPL
JSR SBCSWI
                ;DISPLAY NYMPHS
LDA I,BCNYMP
JSR SBCLOG
JSR DSPNYM
LDA I,BCNYMP
JSR SBCSWI
                ;DISPLAY INFORMATION (SCORES, MSGS, ETC.)
LDA I,BCINFO
JSR SBCLOG
JSR INFO
LDA I,BCINFO
JSR SBCSWI
..
JSR DSPWEL      ;DISPLAY WELL

LDA I,BCENEL    ;DISPLAY ENEMY LINES
JSR SBCLOG
JSR DSPENL
LDA I,BCENEL
JSR SBCSWI

LDA I,BCSTAR    ;DISPLAY STAR FIELD
JSR SBCLOG
JSR DSTARF
LDA I,BCSTAR
JSR SBCSWI

```

In each case we prime the accumulator (A) with a value specific to that object, call the routine SBCLOG, call a routine specific to the object (e.g. DSPCUR), the close by calling the routine SBCWI. These top and tail routines are concerned with setting up and preparing the header and footer material for the list of lines that displays each object. The object-specific routine in the middle is concerned with collecting the lines themselves. We will get the opportunity to explore the detailed mechanics of how some of these objects are calculated and displayed in later sections.

mainloop

Unlike Tempest, the engine that runs Tempest 2000 is not contained within a mainline routine. Instead the donkey work is done during a vertical sync interrupt. Vertical sync is a point in time: specifically when the Atari Jaguar has finished writing pixels to the screen and has a short amount of time available before it starts writing pixels again from the top. An interrupt is a routine the system will call at a moment of the programmer's choosing. A vertical sync interrupt is when you combine the two, and in this case they are combined in a routine called `Frame`. It is this routine that manages the state of the game and prepares all the objects for display and the sound for output.

That said, Tempest 2000 does have a mainline routine but it is there to solve a problem encountered in development rather than as part of a grand design. As we shall see there are a number of ways of generating graphical data on the Atari Jaguar, and two of them have their own dedicated processors. These are the Graphics Processor (GPU), which specializes in the fast trigonometric operations required for 3D displays and the Blitter, which is suited for large copy and fill operations. It is worth being clear that neither of these processors or any of their operations actually write graphics to the screen. Instead this function is performed by a third dedicated processing unit, the Object Processor. It is the Object Processor that turns data into light. This unit takes a list of operations set up by the programmer for each frame and uses them to write pixels to the screen. These operations will usually include data prepared by the GPU and the Blitter in a long list of tasks for the Object Processor known as an Object List. It's the programmer's job to have a new Object List ready every time the Object Processor is about to paint the screen.

It was the initial intention that the `Frame` routine would be solely responsible for this task in Tempest 2000. But there was a snag: it kept crashing. We know this because

Jeff Minter left us one of his rare comments at the head of the `mainloop` routine:

```
; This loop runs the GPU/Blitter code. I found that if you
; started up the GPU/Blitter pair from inside the FRAME
; Interrupt, the system would fall over if they got really heavily
; loaded. MAINLOOP just waits for a sync from the FRAME routine,
; launches the GPU, then loops waiting for another sync.

mainloop:
    move #1, sync           ; Reset the sync
    move #1, screen_ready   ; Reset the screen ready.
    move pauen, _pauen      ; Reset the pause indicator.

main:   tst sync            ; loop waiting for another sync
        bne main             ; from the interrupt in 'Frame'

        move #1, sync          ; reset the sync so that we wait for a
new frame next time around.
        move.l dscreen, gpu_screen

        move.l mainloop_routine, a0 ; do the actual mainloop work,
mainloop_routine
        jsr (a0)                ; is usually a reference to the
                                ; draw_objects' routine.
```

So instead of doing all the necessary GPU and Blitter operations during a vertical sync interrupt, we do them here in this `mainloop`. Notice the busyloop in the second paragraph above where we wait for the value in `sync` to change to a zero. This forces the CPU to wait until the Frame routine resets it to zero. Once it has been reset we can go ahead and execute our `mainloop_routine`. This is actually a reference to the `draw_objects` routine - and it is this routine that does all the GPU and Blitter work required to calculate the pixels for display.

Here we see that when the game is initialized, `draw_objects` is selected as the destination for `mainloop_routine`, and then we enter the `mainloop`.

```
go_in:  move.l #rotate_web, routine
        move.l #0, warp_count
        move.l #draw_objects, mainloop_routine ; Make draw_objects the
mainloop_routine
        move pauen, _pauen
        clr db_on
        bra mainloop
```

We will investigate the mechanics of using the GPU and the Blitter in later sections, but for now let's get a sense of the kind of things the `draw_objects` routine uses the GPU and Blitter for. A quick glance through the top of the routine shows us drawing the starfield (`dostarf`).

```

draw_objects:
    tst h2h                      ; Are we in head to head mode?
    bne nodraw                   ; Yes, go to 'nodraw'.
    clr h2hor

stayhalt:
    tst drawhalt                ;
    bsr clearscreen
    move.b sysflags,d0
    and.l #$ff,d0
    move.l d0,_sysflags         ;pass sys flags to GPU
    tst sf_on                    ; Is the starfield active?
    bne dostarf                 ; If yes, go to 'dostarf'.
    bra gwb                     ; Otherwise do the web.

dostarf:
    move.l #3,gpu_mode          ; Prepare the starfield!
    move.l vp_x,in_buf+4        ; mode 3 is starfield1
    move.l vp_y,in_buf+8        ; Put x pos in the in_buf buffer.
    move.l vp_z,d0              ; Put y pos in the buffer.
    move.l vp_sf,d0             ; Get the current z pos.
    add.l vp_sf,d0              ; Increment it.
    move.l d0,in_buf+12          ; Add it to the buffer.
    move.l #field1,d0           ; Get the starfield data structure.
    move.l d0,in_buf+16          ; And put it in the buffer.
    move.l warp_count,in_buf+20 ; Add the warp count.
    move.l warp_add,in_buf+24   ; Add the warp increment.
    lea fastvector,a0            ; Get the GPU routine to use.
    jsr gpurun                  ; do gpu routine
    jsr gpuwait                 ; Wait until its finished.

```

As you can see 'doing the starfield' involves setting up a number of registers and variables before finally invoking the GPU (`jsr gpurun`) to do its magic and waiting for it to finish (`jsr gpuwait`). We will take a closer look at the mechanics of starfield generation in a later chapter, but the above begins to give us a flavour of the formula required to get a GPU routine up and running.

The next element we find is a routine for drawing the playfield of Tempest 2000: the web. As you can see there is a lot going on here and the truth is, this isn't even all of it. And not just that, there are a number of different GPU routines used for drawing webs scattered throughout the game depending on the mode we are playing or whether there is more than one player. Just looking briefly through the code below (and you should confine yourself to that for now) gives you a sense of the overhead involved in setting up a reasonably complex piece of GPU code. Note that at the end of the listing below we load a variable called `equine2` to the A0 register. This is an address to the actual GPU code that the Graphics Processor will run, using all the data set up in the previous lines. So there is much more complexity and detail underneath, even here. We will cover this in more detail in the chapter devoted to webs.

```

gwb:

```

```
move.l vp_x,d3
move.l vp_y,d4
move.l vp_z,d5

dscud:
solweb:
    move #120,d0
    add palfix2,d0
    and.l #$ff,d0
    move.l d0,ycent
    tst l_solidweb
    beq vweb
    cmp #1,webcol           ;our 'transparent' webs...
    beq vweb
    lea _web,a6             ;draw a solid poly Web
    tst 34(a6)
    beq n_wb
    lea in_buf,a0
    move.l 46(a6),d0
    move.l d0,(a0)
    move.l 4(a6),d0
    sub.l d3,d0
    move.l d0,4(a0)
    move.l 8(a6),d0
    sub.l d4,d0
    move.l d0,8(a0)
    move.l 12(a6),d0
    sub.l d5,d0
    bmi n_wb
    move.l d0,12(a0)
    move l_solidweb,d0      ; The web data structure.
    and.l #$ff,d0
    move.l d0,16(a0)
    move 28(a6),d0
    and.l #$ff,d0
    move.l d0,24(a0)
    move frames,d0
    and.l #$ff,d0
    move.l d0,28(a0)
    move.l #w16col,32(a0)
    move.l #0,gpu_mode
    lea equine2,a0          ; The GPU shader routine for webs.
    jsr gpurun
    jsr gpuwait
    jsr WaitBlit
```

You may note at the end that we call a routine called `WaitBlit`. This is because in addition to computing the polygons that make up the 3D web, the GPU also 'blits' or writes its results to a buffer that will ultimately be used by the Object Processor to write the pixels to the screen. It is this dual operation that caused Minter a headache when attempting it from within the vertical sync interrupt. (My own suspicion is that there just wasn't enough time in the interrupt to do everything he wanted on the GPU and Blitter -

so moving it here to the mainloop ensured that it would only be done opportunistically and at the risk of missing a frame every now and then.)

But there is more to Tempest 2000 than starfields and webs so there must be plenty of other things being drawn in here too. At first it is not easy to see where though. The next section of the `draw_objects` routine is cryptic at best but on close inspection does contain some clues. As elsewhere, I've added comments to assist understanding:

```

n_wb:    move.l activeobjects,a6 ; activeobjects is a list of things to draw!
          bsr d_obj
          bra odend

d_obj:   ; A loop for processing everything in 'activeobjects'.
          cmpa.l #-1,a6      ; Have we reached the end of activeobjects?
          beq oooend        ; If yes, skip to end.
          move 50(a6),d0    ; Is the object marked for deletion?
          beq no_unlink     ; If not, skip to no_unlink and draw it.

          ; This verbose section up until no_unlink is concerned entirely
          ; with deleting the dead object from the activeobjects list.
          move.l 56(a6),d1
          bmi tlink
          move.l d1,a5
          move.l 60(a6),60(a5) ; Make it invisible to the vsync interrupt.

tlink:   move #-1,50(a6)      ; mark it bad
          move.l 60(a6),-(a7)
          move d0,-(a7)
          move.l a6,a0
          move 32(a6),-(a7)   ;save player ownership tag
          move (a7)+,d1
          move (a7)+,d0
          lea uls,a1
          asl #2,d0
          move.l -4(a1,d0.w),a1
          jmp (a1)

uls:    dc.l afinc,ashinc,pshinc

pshinc: tst d1                ;player ownership of an unlinked bullet
         beq ulsh1
         add #1,shots+2
ulo:    move #1,locked
         bsr unlinkobject
         clr locked
         bra nxt_o
ulsh1:  add #1,shots
         bra ulo
ashinc: add #1,ashots
afinc:  add #1,afree
         bra ulo

```

```

; Actually draw the object.
; No need to remove the object, just draw it.

no_unlink:
    lea draw_vex,a0      ; Get our table of draw routines.
    move 34(a6),d0        ; Is this object smaller than a pixel?
    bpl notpxl            ; If not, go to notpxl.
    move.l #draw_pel,a0   ; Use draw_pex for pixel-size objects.
    bra apal              ; Jump to the draw call.

notpxl:  asl #2,d0       ; Multiply the val in d0 by 2.
        move.l 0(a0,d0.w),a0 ; Use it as an index into draw_vex.

apal:   move.l 60(a6),-(a7) ; Store the index of next object in a7.
        jsr (a0)             ; But first call the routine in draw_vex.
        jsr gpuwait           ; Wait for the GPU to finish.

nxt_o:  clr locked       ; Clear 'locked' just in case.

nxt_ob: move.l (a7)+,a6   ; Put the index of next object back in a6.
        bra d_obj             ; Go to the next object.

oooend: rts

```

This part of the `draw_objects` routine is concerned with processing a linked list called `activeobjects` using a loop that runs from `d_obj` to `bra d_obj` in the second last line from the end. This `activeobjects` list contains the detail for all objects that are active on the screen and require drawing by the GPU/Blitter. Some of the original (but sparse) code comments added by Minter suggest that this was initially conceived as a list for managing just the player and enemy bullets but it expanded over time.

Each object in the `activeobjects` list has the following structure:

Bytes	
0-4	Vector Object or Solid Object
4-8	X
8-12	Y
12-16	Z
16-20	Position on web.
20-24	Velocity
24-28	Acceleration & XY orientation
28-30	XZ Orientation & XZ orientation
30-32	Y Rotation
32-34	Z Rotation
34-36	Index into draw routine in <code>draw_vex</code> .
36-38	Start address of pixel data. & Delta Z
38-40	Colour change value
40-42	Colour
42-44	Scale factor
44-46	Mode to climb, descend or cross rail
46-48	Size of Pixel Data & Duration.
48-50	Fire Timer
50-52	Marked for deletion
52-54	Whether an enemy or not.
54-56	Object Type
56-60	Address of Previous Object
60-64	Address of Next Object

The structure of objects in `activeobjects`.

So in the 64 bytes of each object we cover quite a bit of ground. There is colour, co-

ordinate, motion, and orientation information. There's also detail that defines the behaviour of the object and most immediately relevant to what we are looking at here, an index into the draw routine to use for the object at bytes 34-36.

This index is a value that references the routine given at the appropriate position in the `draw_vex` array:

```
draw_vex:
    dc.l rrts,draw,draw_z,draw_vxc,draw_spike,draw_pixex,draw_mpixex,
    draw_oneyup,draw_pel,changex
    dc.l draw_pring,draw_prex,dxshot,drawsphere,draw_fw,dmpix,dsclaw,
    dsclaw2
```

The draw routine can vary depending on the type of object. There's a distinct treatment of vector object and objects that are solid polygons. Vector objects require the least work and if the header in bytes 0-4 indicates that it requires vector drawing only (for example the player's claw) then a vector draw in the GPU will suffice. The draw routine in the second item in `draw_vex` is the base routine for deciding whether a vector draw is sufficient or not, and most of the other routines make use of it in addition to the object-specific detail they manage:

```
draw:
    move.l a6,oopss    ; Stash the header.
    move.l (a6),d0      ; Is the header value greater than zero?
    bpl vector          ; If yes, then a vector draw will suffice.

    ; Otherwise we need to do add this object to the 'apriority'
    ; list so that it can be drawn as a solid polygon.

    ; The 'apriority' list stores objects in the descending order
    ; of their Z co-ordinate. This ensures that nearer objects are
    ; painted in front of objects that are further away or 'behind' them

    move.l fpriority,a0  ;get a free priority object
    move.l a6,(a0)
    move.l 12(a6),d0    ;get 'z'
    move.l d0,12(a0)    ;put z in prior object
    move.l apriority,a1
    move.l a1,a2

chklp:
    cmp.l #-1,a1        ;no objects active?
    bne prio1
    bra insertprior     ;we are at top of list then, if we are first a1=
    a2=-1

prio1:
    cmp.l 12(a1),d0    ;check against stored 'z'
    bge insertprior    ;behind, insert on to list
    move.l a1,a2
    move.l 8(a1),a1    ;get next object
```

```
bra chklp      ;loop until list end or next object in front of us
rts          ;return with object at right place in the list
```

Reading through the above we can see that if a vector draw won't do the job then the object gets added to a list called `apriority` and nothing else is done with it for now. So what we are doing in our `d_obj` loop is a first pass through the `activeobjects` list, with any items on the list that require attention to the order in which they are painted, passed off to the `apriority` list. Notice that we don't mutate or update the objects in any way, we simply pass them over so the structure of the objects we described above remains unchanged.

Once we have finished this first run through the `activeobjects` list our next order of business is to process the `apriority` list we've populated. This is done in `drawpolyos` which we call right after we've finished with our first pass of `activeobjects`.

```
; We've finished our first pass of activeobjects.
odend:
bsr showscore      ; Show the score.
; In Tempest Classic mode we don't need solid polygons.
tst blanka         ; Are we doing solid polygons?
beq odvec         ; If not, skip.
bsr drawpolyos    ; if we are, draw them.
```

Drawing our polygons happens below. We iterate through the '`apriority`' list, deleting items in it as we go, and calling the object-specific draw routine for each to render them to the screen.

```
solids:
dc.l rrts,cdraw_sflipper,draw_sfliptank,s_shot,draw_sfuseball
dc.l draw_spulsar,draw_sfusetank,ringbull,draw_spulstank
dc.l draw_pixex,draw_pup1,draw_gate,draw_h2hclaw,draw_mirr
dc.l draw_h2hshot,draw_h2hgen,dxshot,draw_pprex,draw_h2hball
dc.l draw_blueflip,ringbull,supf1,supf2,draw_beast,dr_beast3
dc.l dr_beast2,draw_adroid

; Routine for drawing all solid polygons.
; Process each object in the 'apriority' list. We remove each item after
; processing. The draw routine for each item is given by its index into
; the 'solids' array.
drawpolyos:
move.l #192,xcent   ; Set 192 as X centre.
move.l #120,d6       ; Set 120 as Y centre.
add palfix2,d6       ; Adjust for PAL if necessary.
move.l d6,ycent     ; Store it as Y centre.
move.l apriority,a0 ; Get our 'apriority' list.
dpoloop:
cmp.l #-1,a0
beq rrts            ; End of list was reached
move.l (a0),a6        ; Get the index to 'solids'
```

```

move.l (a6),d0      ; Store it in d0.
move.l a0,-(a7)    ; Stash our current position in the list.
bsr podraw          ; Go do object type draw
jsr gputwait        ; wait for gpu
move.l (a7)+,a0     ; Get our current position in the list
move.l 8(a0),-(a7)  ; Get the next position in the list
bsr unlinkprior    ; Delete the current object.
move.l (a7)+,a0     ; Move to the next position in the list.
bra dpoloop         ; Loop until all objects drawn and unlinked

podraw:
move.l #9,d4        ; Set X centre as 9.
move.l #9,d5        ; Set Y centre as 9.

soldraw:
neg d0
lea solids,a4       ; Get the 'solids' list.
lsl #2,d0           ; Multiply our index by 2.
move.l 0(a4,d0.w),a0 ; Get the draw routine address from 'solids'.
move.l 4(a6),d2     ; Get the X position from our object.
sub.l vp_x,d2       ; Subtract our X viewpoint.
move.l 8(a6),d3     ; Get the Y position from our object.
sub.l vp_y,d3       ; Subtract our Y viewpoint.
move.l 12(a6),d1    ; Get the Z position from our object.
sub.l vp_z,d1       ; Subtract our Z viewpoint.
bmi rrts            ; Skip if not visible.
move 28(a6),d0      ; Get orientation of object.
and.l #$ff,d0       ; Use only the least significant bytes.
jmp (a0)             ; Call the objects draw routine.
; The draw routine returns to 'dpoloop'.

```

The Frame Routine

We set up our frame interrupt handler near the very start of initialisation:

```

*****int setup
jsr scint                  ;set intmask according to controller prefs
move.l #Frame,$100
move.w n_vde,d0
or #1,d0
move d0,VI
move pit0,PITO
clr d0
move.b intmask,d0
move.w d0,INT1              ;enable frame int
move.w sr,d0
and.w #$f8ff,d0
move.w d0,sr                ;interrupts on

```

Frame:

tempest 2000

```
;      add #1,frames

;      move.l #$FFFFFFF,BORD1
;      movem.l d0-d5/a0-a2,-(a7)           ;simple thang to make

; vertical blank code goes here


fr:      move INT1,d0
        move d0,-(a7)
        btst #0,d0
        beq CheckTimer           ;go do Music thang

        movem.l d6-d7/a3-a6,-(a7)
        move.l blist,a0
        move.l dlist,a1
        move.l #$30,d0
xlist:   move.l (a0)+,(a1)+       ;copy built list to displayed list
        dbra d0,xlist

        bsr RunBeasties          ;build the next one


setdb:
;
; this code writes the proper screen address to double buffered objects in
the display list

;      lea 32(a0),a0           ;skip background object

        tst screen_ready         ;is GPU ready with a new screen
        beq no_new_screen        ;no
;
        tst pawsed
;
        bne no_new_screen
        tst sync
        beq no_new_screen
        move.l cscreen,d1
        move.l dscreen,cscreen
        move.l d1,dscreen         ;swap screens
        clr screen_ready
        clr sync
no_new_screen:

        move.l dlist,a0
        move db_on,d7             ;check for double buffer on (d7 holds # of
contiguous DB'ed screens)
        bmi no_db

stdb:    move.l cscreen,d6      ;get address of current displayed screen
        and.l #$fffffff8,d6      ;lose three LSB's
```

```

lsl.l #8,d6           ;move to correct bit position
move.l (a0),d1         ;get first word of the BMO
and.l #$7ff,d1        ;clear data pointer
or.l d6,d1             ;mask in new pointer
move.l d1,(a0)          ;replace in OL
lea 32(a0),a0          ;skip to nxt object
;      dbra d7,stdb       ;loop for all DB backgrounds

no_db:
    jsr dowf

    tst pal
    bne dtoon
    add #1,tuntime        ;do NTSC interrupts only 5/6 times
    cmp #4,tuntime
    ble dtoon
    cmp #6,tuntime
    bne nton
    clr tuntime
dtoon:   tst modstop
    bne nton
    jsr NT_VBL
nton:

;      bsr readpad          ;get joy values
    tst pawsed
    bne zial
    add #1,frames
    move.l fx,a0
    jsr (a0)                ;for plaette changes etc
zial:
    tst locked
    beq doframe
;      move #1,drawhalt
    bra loseframe
doframe:
    move.l routine,a0
    jsr (a0)                ;call do thangs routine
    clr drawhalt

loseframe:
    bsr checkpause         ;do pause and overriding reset command
    bsr domod               ;do music handling
    btst.b #0,sysflags
    bne chit                 ;check for no h.interlace
    move frames,d0
    and #$01,d0
    add #SIDE,d0
    sub palside,d0
    move d0,beasties        ;H. Interlace mode
    movem.l (a7)+,d6-d7/a3-a6
chit:

```

mainloop

```
CheckTimer: move (a7)+,d0           ;get back int status
            move d0,-(a7)
            btst #3,d0
            beq exxit

            tst roconon          ;Rotary Controller enabled?
            bne roco
            jsr dopad
            bra exxit             ;Yeah, interrupts at 8x normal speed, go do
            special stuff

roco:    move pitcount,d1
            and #$07,d1
            bne rotonly
            jsr dopad
rotonly: add #1,pitcount
            bsr readrotary

exxit:  move (a7)+,d0
            lsl #8,d0
            move.b intmask,d0
            move d0,INT1
            move d0,INT2
            movem.l (a7)+,d0-d5/a0-a2
;            move.l #0,BORD1
;            move.w #$101,INT1      ;do interrupt stuff
;            move.w #0,INT2
            rte
```

cry if i want to

Before the dawn of the three horsemen *png*, *gif* and *jpg*, computer people were free to invent their own schemes for storing images. In 1991 technology was still rudimentary enough that console and computer developers were largely pre-occupied with waging something resembling a *Color War*. So when Atari sat down to brew its own method of storing and representing image data one of its overriding concerns was *how many colours can we fit into this thing*.

The **CRY** file format is the result. 'Format' in this case is a slightly grandiose term for: here is a list of bytes in a file, each pair of bytes in the list represents a color. Since a single byte has 256 possible values, two bytes has 65,536 possible values. So welcome to our system that has a state-of-the-art 65,536 possible colours. Well, not quite. Let's see why.

claw says



Good luck figuring out what CRY is an acronym of. Maybe it refers to the three axes of the 'colour space': cyan, red, and yellow. Maybe it refers to the three components contained in each byte pair: color, radiance, and luminosity?

In your innocence dear reader, you may expect that this color scheme is as simple as assigning a color to each of the possible values and considering ourselves done for the day. But that is nowhere near complex enough to generate true job satisfaction. Instead, we must declare something called a 'colour space' and declare co-ordinates along an X and Y axis. Even better, since this is not the 1980s anymore, we ought to thinking in three dimensions and throw in a Z axis while we're at it.




Figure .1: This kind of thing. This is what we want.

With just two bytes (or 16 bits) to work with we have to come up with a way that is suitably complicated for representing all our colours along so many co-ordinate axes. So let's imagine an 8x8x8 colour cube with something resembling white at its top corner and all the colours of the spectrum radiating out of it.




Figure .2: This is not quite right, but you get the idea.

Now if we project this three dimensional object onto a flat surface, with our white top corner at the centre, we will have a 16x16 square with 256 different colours:




Figure .3: 3d, but in 2d: it's the 1990s.

Since 256 is exactly the number of permutations in a single byte we have lit upon a use-case for the first of our two bytes.

0	0	0	0	1	0	0	0	0	0	0	0	1	0	0	0
---	---	---	---	---	---	---	---	---	---	---	---	---	---	---	---

Figure .4: The first 8 bits now have a purpose. The first four are our X axis on the square, the second our Y axis.

This is all very nice, we have 256 colors going on and a whole other 8 bits to do something with. After thinking about it a bit, we don't have too many options. In fact we have one option: use our extra 8 bits to vary each of our 256 colors in 256 different ways. We can make this twiddle factor sound intelligent by calling it 'luminosity'. So depending on how much light we think is being cast on a surface with the color in question we can use our final 8 bits to dial its brightness up or down.

0	0	0	0	1	0	0	0	0	0	0	0	1	0	0	0
---	---	---	---	---	---	---	---	---	---	---	---	---	---	---	---

Figure .5: The 8 bits representing the luminosity/intensity of the color chosen by the first 8 bits.

To get an idea of the effect this has in practice we can visualize our little color square repeating across a much bigger square with ascending values of luminosity starting at 0 and ending at 255.




Figure .6: A big square with all our little squares in ascending luminosity.

That's a lot of dark. We better not wear shaded spectacles while playing games on this thing. Is there really 65,536 colors in there, it seems like we have an awful lot of black going on. Let's try something and see what we get if we ignore all color values that appear more than once.




Figure .7: Of course the 65,536 colors is a lie. I've counted them so you don't have to: there are 63,762 unique colours.

There are some strange little fractal shapes in the black stuff there. In the bottom right hand corner we can see that we eventually run out of luminosity, our last little square can only muster up 120 or so new colours.

the tempest files

Now that we know what the bytes in a CRY image mean let's take a peek at the 6 image files in the Tempest 2000 sources to see what some of these bytes look like in practice. (Don't worry we'll be looking at all of these files in detail under one pretext or another during the course of the rest of this book.) The seven image files are given slightly random, but very Minter-ish names: `beasty3.cry`, `beasty4.cry`, `beasty5.cry`, `beasty6.cry`, `beasty7.cry` and `beasty8.cry`.

claw says

Oh and let's not forget `beasty8-xtra.cry`. This is a slightly larger version of `beasty8.cry` included with the sources but for some reason is not used. There is also a story to tell about `beasty3.cry` but we will come to that later.

Remember there is nothing else in these files but the series of byte-pairs. There's no information about how big the image is or what its dimensions are. It really is just a list. To cope with this fact the game hard-codes its own expectation that the images are 320 pixels wide (i.e. the pixel width of the Jaguar's screen) and fixes the address that each image at will be loaded at. It does this early on in the declaration of constants at the top of the main file containing all the game's code (`yak.s`):

```
pic      EQU $820000      ; beasty3.cry
pic2    EQU pic+$1f400    ; beasty4.cry
pic3    EQU pic2+$1f400    ; beasty5.cry
pic4    EQU pic3+$25800    ; beasty6.cry
pic5    EQU pic4+(640*128)  ; beasty7.cry
pic6    EQU pic5+(640*200)  ; beasty8.cry
```

Each image has been loaded to a specific position in memory and the game is instructed exactly where each one is. While `beasty3.cry` is loaded to \$820000 (and assigned the variable name `pic`), `beasty4.cry` is loaded to the address 128,000 (\$1f400) bytes after that and assigned the name `pix2` because 128,000 bytes is how large `beasty3.cry` is. You get the idea hopefully.

As you can see in the above listing extract some of the images are larger than others so need more memory, but in all cases the assumption of treating every image as 320 pixels wide (640 bytes, since we have 2 bytes per pixel) holds.

Before passing on let's take a look at the contents of `beasty7.cry`.



Figure .8: beasty7.cry, packed with goodness.

Looks great, you have to agree. We're in for some fun.

what is blitting

Blitting is the process of writing pixels to the screen. A blitter is a *bit block processor*. When we *blit* an image to the screen we are copying a specific block of image data from storage to the screen where the user can see it.

This short extract from the routine that looks after the 'High Score' table in Tempest 200 gives you an idea of how easy using the blitter can be. All we have to do is point at the position in RAM containing our pixel data (in this case the pixels from `beasty7.cry` that we loaded to `pic5`), tell the blitter the x and y position we want to copy from, the width and height we want to copy, and the x and y position on the screen we want to copy to.

```
; Paint the 'Top Guns' graphic
move.l #pic5,a0      ; file images/beasty7.cry, our source data.
move.l #screen3,a1  ; the destination we're copying to.
move #1,d0           ; x position in pic5
move #1,d1           ; y position in pic5
move #223,d2         ; width of block to copy
move #79,d3          ; height of block to copy
move #70,d4          ; x pos of destination
move #10+8,d5        ; y pos of destination
jsr CopyBlock
```

So in the above we specify a block in the image data from `beasty7.cry` opposite from the top-left corner ($x:1, y:1$) with a width of 223 pixels and a height of 79 pixels. We then tell it that we want to write this to x position 70 and y position 18 on the screen, in other words somewhere just offset from the top left of the screen. This is the piece of the image we are selecting for copying:



Figure .1: The bit of `beasty7.cry` we've selected for copying.

If we look more closely at the snippet I extracted above we can see that we seem to be doing this 'specifying' by moving our values into things called `a0`, `a1`, `d0`, `d1`, and so on. These are 'registers'. An old computer such as the Motorola 68K CPU in the Atari Jaguar has only so many fingers and thumbs it can use for counting, and these are them. `a0`, `a1` and so on are 'address registers'. So if we have the address of a piece of code or data we want the CPU to use an 'address register' is where we store them. If we have some 'values' on the other hand, i.e. some numbers, then the 'data registers' `d0`, `d1` are where we put them.

As you read on you'll find it's a common pattern in Tempest 2000's source code to load up a bunch of registers with bits and pieces and then call a function (or 'routine' in assembler parlance) that expects those registers to contain the values it should work with. If this sounds familiar, and that we are essentially using these registers as parameters to a function, then you are correct: that's exactly what we're doing.

The simple operation we're performing here is to copy a portion of the `beasty7.cry` image file to the screen. The `CopyBlock` routine opposite just needs to know the position and dimensions of the portion we're copying and the position on the screen we want to place it. To achieve this the snippet of game code on the previous page sets up the registers as follows:

Register	Type	Value	Description
<code>a0</code>	Address	<code>pic5</code>	Source Screen
<code>a1</code>	Address	<code>screen3</code>	Destination Screen
<code>d0</code>	Data	1	X Position in Source Screen
<code>d1</code>	Data	1	Y Position in Source Screen
<code>d2</code>	Data	223	Width to Copy
<code>d3</code>	Data	79	Height to Copy
<code>d4</code>	Data	70	X Position in Destination Screen to Copy to
<code>d5</code>	Data	18	Y Position in Destination Screen to Copy to

Setting up the registers for `CopyBlock`.

```

CopyBlock:
;
; Copy from screen at a0 to screen at a1
; d0/d1=origin of sourceblock
; d2/d3=width and height of block to copy
; copy from blitter a1 to a2.
; d4/d5=destination XY
;
; This simple routine will assume both screens are the same width
; Using this blitter is a piece of piss.

move.l #PITCH1|PIXEL16|WID320|XADDINC,d7
move.l d7,A1_FLAGS ;a1 (Source) Gubbins

move.l #PITCH1|PIXEL16|WID384|XADDPIX|YADD1,d7
move.l d7,A2_FLAGS ;a2 (Dest) Gubbins

move d3,d7
swap d7
move d2,d7
move.l d7,B_COUNT ;set inner and outer loop counts

move d1,d7
swap d7
move d0,d7
move.l d7,A1_PIXEL ;origin of source

move d5,d7
swap d7
move d4,d7
move.l d7,A2_PIXEL ;origin of destination

move.l #0,A1_FPIXEL

move.l #$0001,A1_INC
move.l #$0,A1_FINC

move #1,d7
swap d7
move d2,d7
neg d7
move.l d7,A1_STEP
move.l d7,A2_STEP ;set loop steps

move.l a0,d7
move.l d7,A1_BASE
move.l a1,d7
move.l d7,A2_BASE ;set screen window bases

move.l #SRCEN|UPDA1|UPDA2|DSTA2|LFU_A|LFU_AN,d7
move.l d7,B_CMD
bra WaitBlit

```

claw says

The source and destination are described as 'screens' in the Tempest 2000 code. This is a useful way of thinking of our pixel data, which although it is just a list of bytes, always represents a screen of a certain height and width.

While in practice we are copying a section from an image to the screen here, we can think of it as copying from one screen to another - with the destination screen the one being shown to the player.

The mechanics of how the CopyBlock routine uses this information to actually copy a segment of our image to the screen are not obvious to the uninitiated. This is because it involves a lot of precise specification of many different instructions to the Blitter (Bit Block Processor) hardware. The amount of set up involved makes our preparation of the a0-a1 and d0-d5 registers look modest in comparison. There is a quite involved amount of bit-twiddling required to get the Blitter set up the way we want it. Using this routine may be a 'piece of piss', but writing it was fairly painstaking. For example:

```
move.l #PITCH1|PIXEL16|WID320|XADDINC,d7
move.l d7,A1_FLAGS ;a1 (Source) Gubbins
```

Here we are formulating some of the parameters we want the Blitter to use when it copies the pixels from our source. We're combining the parameters into a single 64-bit value, storing it in the d7 data register and then writing that to the Blitter's A1_FLAGS register. This is the register it will read to figure out some of the things we want to do with the copy operation.

Parameter	Bits	Hex	Description
PITCH1	00000000 00000000 00000000 00000000	00000000	Pixel data has no gaps
PIXEL16	00000000 00000000 00000000 00110000	00000020	Pixel Size 16 bits
WID320	00000000 00000000 01000010 00000000	00004200	Screen Width of 320
XADDINC	00000000 00000011 00000000 00000000	00030000	Add 1 to X at each pass
A1_FLAGS	00000000 00000011 01000010 00110000	00034220	Blitter's A1 Flags Register

Combining our parameters into the Blitter's A1 Flags Register.

There is a lot more of this kind of thing in there and it would definitely be too tedious to itemize each operation here in detail. But there is at least one other pattern worth pointing out in the routine, exemplified by the following snippet that copies the X (d0) and Y (d1) positions in *beasty7.cry* that we are telling the blitter to use as its origin for copying:


```
move d1,d7
swap d7
```

```
move d0,d7
move.l d7,A1_PIXEL ;origin of source
```


Here the A1_PIXEL register we're writing to (like all our address and data registers) is 32-bits (or 4 bytes) long. This means it can contain both our X and Y values which will be at most 2 bytes long each. So to get our d0 and d1 values in there we first write the d1 value to d7 giving:



Next, we swap its position:



Now we write d0 to d7:



The result is our X and Y values are both successfully stowed in d7, which we can now write in its totality to A1_PIXELS. This little exercise illustrates the granularity at which our code has to operate.

After setting up all our registers on the Blitter we finally get to tell it to set to work:

```
move.l #SRCEN|UPDA1|UPDA2|DSTA2|LFU_A|LFU_AN,d7
move.l d7,BLIT_CMD
```

As with everything else Blitter-related this is achieved above by writing a value to the Blitter command register. The register affords up to 30 different commands and here we just use a few of them by populating the register with six different values.

Parameter	Bits	Hex	Description
SRCCEN	00000000 00000000 00000000 00000001	00000001	Enable Source Read in Loop
UPDA1	00000000 00000000 00000001 00000000	00000100	Add the A1 value in the outer loop
UPDA2	00000000 00000000 00000010 00000000	00002000	Add the A2 value in the outer loop
DSTA2	00000000 00000000 00000100 00000000	00004000	Reverse role of A1 and A2 registers
LFU_A	00000001 00000000 00000000 00000000	01000000	AND the bits
LFU_AN	00000000 10000000 00000000 00000000	00800000	AND NOT the bits
BLIT_CMD	00000000 00000011 01000010 00110000	01804210	Blitter's Command Register

Combining our parameters into the Blitter's Command Register.

In a nutshell this command is instructing the blitter to read through our source data using an inner loop for the values to write along the X co-ordinates and an outer loop for the Y co-ordinates. The implementation details are obviously complex and we won't get into them here. Instead we can treat this as a recipe for the blitter when we have a chunk from a source image that we want to position on the screen as we do here.

One last feature is worth pointing out explicitly. The final step in the CopyBlock routine is to call the WaitBlit sub-routine. As the name suggests this piece of code busy-loops until the BLIT_CMD register indicates that the blitter has finished:

```
WaitBlit:  
    move.l BLIT_CMD,d7 ;get Blitter status regs  
    btst #0,d7          ;check if d7 is still zero  
    beq WaitBlit        ;loop if d7 is zero  
    rts
```

Before rounding up this piece let's take a look at another instance of CopyBlock in use. This time we're in the versionscreen routine which is responsible for painting the title screen that greets us when we launch Tempest 2000.



Figure .2: The title screen of Tempest 2000.

Here we find a slightly different order of business in setting up the blitting copy:

```
WaitBlit:  
    move #4,d0           ; x position in pic5  
    move #84,d1           ; y position in pic5  
    move #197,d2           ; width to copy  
    move #65,d3           ; height to copy  
    move #92,d4           ; x pos of write to  
    move #120-15,d5         ; y pos to write to
```

```

tst pal           ; are we PAL or NTSC?
beq mypal
add #10,d5       ; we're NTSC so offset the y pos by 10 pixels
mypal:
move.l #pic5,a0a ; file 'images/beasty7.cry' again
move.l #screen3,a1 ; destination we're copying to
jsr CopyBlock

```

Can you guess the piece of `beasty7.cry` we're copying here? It's this:




Figure .3: Extracting the '2000' chunk from `beasty7.cry`.

There is a little extra twiddling in our code to figure out whether the display we're on is PAL or NTSC. These were slightly conflicting standards for analogue television display through the 1960s to the 1990s. NTSC was the standard in the US and PAL the standard in Europe. Most video games of the time needed to account for the slightly different dimensions each standard offered. In the case of our title screen that meant adding an extra offset of 10 pixels to the Y position of the '2000' when painting it to the screen on an NTSC set in order to ensure that it appeared correctly positioned.

Now that we've got the general idea of how to get at least some pictures on to the screen, let's take a look at how we get some text up there.

character assassination

Writing some text to the screen, such as the copyright information on the title screen, is surprisingly convoluted in its detail, it will even introduce us to the machinations of the Jaguar's GPU. But although there is a surprising amount of code required from the developer to get a string of characters onto the display the general principle is one familiar from other platforms: you define a font, you specify the text you want to display and where you want to display it, and finally you call a routine that will take care of all the detail.

In our case the routine that takes care of all the detail will be our first taste of the implementation and use of what are known as 'shaders', free-standing programs that are written in a customized version of Motorola 68000 assembly language for the Atari's Graphics Processing Unit(GPU), capriciously dubbed 'Tom' by the hardware designers of Atari at the time.

Before we get to that we can coast through the relatively easy part of defining our font and specifying our string of text, as well as plugging in the values to the GPU's buffer that will be required by the shader when it takes care of actually putting pixels on screen.

When it comes to defining a font for the copyright information on Tempest 2000's title screen you may have already guessed where information is contained: in one of our beastly cry files, in this case `beasty4.cry`:



Figure .1: `beasty4.cry` with the copyright font at the bottom.

And as you may remember the data in `beasty4.cry` is pointed by the variable `pic2`:

```
pic      EQU $820000      ; beasty3.cry
pic2    EQU pic+$1f400    ; beasty4.cry
pic3    EQU pic2+$1f400   ; beasty5.cry
pic4    EQU pic3+$25800   ; beasty6.cry
pic5    EQU pic4+(640*128) ; beasty7.cry
pic6    EQU pic5+(640*200) ; beasty8.cry
```

So all we have to do is devise a scheme that maps from a given letter or number to its location (and dimensions) in `beasty.cry`. For example, if we want to display the letter 'A' we need to come up with a way of extracting this:



A.

It helps us that all of our font characters in the `cry` file are the same dimensions so we can make the height and width a consistent assumption for every glyph. Armed with the location of the pixel data and this dimension information all that remains is to create a look-up table that tells us the X and Y co-ordinates in `beasty4.cry` for each character.

The easiest form of look-up table we can make is an 'indexed' one. In other words, a procedure that converts each letter and digit in our character set into a number that we then use as an index into a list of the X and Y locations. For example, if 'A' is converted to the number 41 then we fetch its coordinates in `beasty4.cry` from the

42nd member of the list. If 'B' is converted to number 42 we fetch its co-ordinates from the 43rd member of the list, and so on. Fortunately we don't have to invent a method of our own for relating letters and digits to a number. The ASCII committee has done this for us already. So as long as we store the co-ordinates in our list in the same order as given by the ASCII committee we will simply be able to use the corresponding ASCII value for each letter and digit as our index.

With the main planks of our cunning plan settled in place we can iron out a few details. The best way of storing our co-ordinate information in this list is to cram the X and Y values into a single place. More specifically, we can assume our X and Y values will always be between 0 and 65,536 so each will fit in one half of a single 4 byte (64 bit value). Another way of describing a value of this length is as a 'long' and in 68K assembler the notation to denote such a value is dc.1.

So let's assume we have 'A' at Y co-ordinate 165 and X co-ordinate 2 in our `beasty4.cry` 'sprite sheet'. Since we're dealing with zero-based indices we subtract 1 from each of these values to give us 164 and 1 respectively. In hex 164 is \$A4 and 1 is \$01. We can fit this into our 4 byte entry as follows:

00	A4	00	01
----	----	----	----

Likewise with our fixed set of dimensions for each of the characters or glyphs. We get out our ruler, apply it to the screen, and determine they each have a height of 19 and a width of 16. We subtract 1 from each to give us 18 and 15. Convert to hex: \$12 and \$0F. Then plug them into our 4 byte value:

00	12	00	0F
----	----	----	----

With this information we can now isolate our 'A' glyph and extract it.



Figure .2: The 'A' glyph located in a red box by our dimension and co-ordinate information.

The mechanics of our look-up table are now fully worked out so we are ready to create it. Since it has over a hundred entries we store it in a separate file called `afont.s` and give it the very descriptive name `afont`. I think this is intended as an abbreviation of 'Atari Font', since it is the font used to write the Atari copyright information. An incomplete excerpt of the file follows below with ellipses (...) to indicate lines that have been omitted.

```
*  
*  
* Page 2 font, by Joby  
  
afont:  
    dc.l pic2      ; pixel data  
    dc.l $0012000f ; height ($0012) and width ($000f)  
  
    dc.l $b6011e   ;Space  
    dc.l $b600d2   ;!  
    dc.l $a40001   ;"  
    dc.l $a40001   ;#  
    dc.l $a40001   ;$  
    dc.l $a40001   ;%  
    dc.l $a40001   ;&  
    dc.l $a40001   ;'  
    dc.l $9100a5   ;(  
    dc.l $9100ba   ;)  
    dc.l $a40001   ;*  
    dc.l $a40001   ;+  
    dc.l $b600f8   ;,  
    dc.l $b6010b   ;-  
    dc.l $b600c2   ;.  
    dc.l $a40001   ;/  
...  
    dc.l $a40001   ;A <-- This is the one! Y: $00a4, X: $0001  
    dc.l $a40014   ;B  
    dc.l $a40027   ;C  
...
```

Listing .1: The opening block of the `afont` data structure in source file `afont.s`.

Notice that the order in which we have defined the location of each character corresponds to the order (or value) of each character according to the ASCII specification. This means that when we define the text we want to display all we have to do is convert each character in the text to its corresponding ASCII value 'n' and get that `n`th value from our `afont` list.

So let's define the text we want to display:

```
ataricop1: dc.b "COPYRIGHT 1981,",0  
ataricop2: dc.b "1994 ATARI CORP.",0
```

This is straightforward enough. Note that we have a zero at the end of each string. This makes them 'null-terminated' - a feature that will come in handy later when we need to identify that the end of the string when writing it to the screen.

Now that we have our text defined we can start the process of writing it to the screen. These two concise little paragraphs do this job for us. Given the amount of info encoded in our afont data structure we only need to set up one additional piece of info for the centext routine that will do our heavy lifting. This is the Y position that we want to write the text to on the screen (adjusted for whether we're on a PAL or NTSC screen).

```

lea afont,a1      ; Load the font data structure
lea ataricop1,a0 ; Load the first line of the copyright info
move #190-8,d0    ; Set the Y position for the text
add palfix2,d0    ; Adjust for PAL screens
jsr centext       ; Write the text

lea afont,a1      ; Load the font data structure
lea ataricop2,a0 ; Load the second line of the copyright info
move #207-5,d0    ; Set the Y position for the text
add palfix2,d0    ; Adjust for PAL screens
jsr centext       ; Write the text

```

The centext routine (short for 'center the text') we call into at the end of each paragraph above is where things start to get detailed. It is still not the tangled core of this particular beast but it is certainly a busier proposition than anything we've looked at so far. But we can settle our nerves for the moment because it is in fact doing something quite straightforward: it is setting up a buffer of data for the GPU to use. It stores this buffer at an address in the GPU RAM given by the inbuf variable. This buffer is a 36 byte array containing the 4-byte addresses of all the different bits of data the GPU will need to render the text. Here are all the bits and pieces in the order that they appear in inbuf once centext is finished:

GPU RAM Address	Name	Description
\$F0003F60	ataricop1	Address in RAM of the text to write
\$F0003F64	afont	Address in RAM of the font data structure
\$F0003F68	\$00000000	Drop Shadow Vector X axis
\$F0003F6C	\$00000000	Drop Shadow Vector Y axis
\$F0003F70	\$00010000	Text Scale X axis
\$F0003F74	\$00010000	Text Scale Y axis
\$F0003F78	\$00000000	Text Shear X axis
\$F0003F7C	\$00000000	Text Shear Y axis
\$F0003F84	\$00B600A7	Y & X Co-ordinates to Write Text To
\$F0003F80	\$00000000	Drop Shadow Mode Enabled

The 10 4-byte addresses stored beginning in inbuf by centext.

In addition to the items of data we've collected so far we can see that we plug in a few extras in the context routine. These are graphical effects such as the scaling to apply to each character, the dimensions of a drop shadow to apply on each, and finally the 'shear' to apply. 'Shear' distorts an object along the depth or 'Z' axis to make it appear like we are looking at it at an angle in 3 dimensions. The only one of these parameters we are using is the scaling, for the rest we just plug in zero. But here's what the copyright text would look like with some shear applied:



Figure .3: What the characters look like when we apply a shear value of 800 on both the X and Y axes.

And here's what they look like if we bump up the scale on the X and Y axis:



Figure .4: What the characters look like when we apply a scale value of 2000 instead of 1000. The pixelated, exploded appearance gives some hint that 'scale' here is not simply concerned with size but a specific 'explosion type' effect.

We'll encounter effective use of these features elsewhere in Tempest 2000 so won't go any further into them for now.

```
context:  
    move.l a0,d2          ; Store 'text' in d2 so we don't overwrite it.  
    lea in_buf,a0          ; Use 'in_buf' as the buffer to write to.  
    move.l d2,(a0)         ; Store 'text' in the first position.  
    move.l a1,4(a0)        ; Store 'afont' in the 2nd position.  
    move.l #0,8(a0)        ; Store 0 in the 3rd position.  
    move.l #0,12(a0)       ; Store 0 in the 4th position (Dropshadow vector).  
    move.l #$10000,16(a0)  ; Store 10000 in the 5th position (X Text Scale).  
    move.l #$10000,20(a0)  ; Store 10000 in the 6th position (Y Text Scale).  
    move.l #0,24(a0)        ; Store 0 in the 7th position (X Text Shear).  
    move.l #0,28(a0)        ; Store 0 in the 8th position (Y Text Shear).  
    move.l #0,36(a0)        ; Store 0 in the 10th position.  
  
    ; This paragraph is about figuring out the X pos we want to place  
    ; the text and then storing that along with the Y pos in the 10th  
    ; position in the buffer  
    bsr g_textlength ; Get the length of the text in d2 and store it in d7.
```

```

; This uses the text length to center it along the X axis and
; store the result in d0;
lsl #1,d7
neg d7
add #192,d7
; d0 contains the Y pos, so move it to the left hand side
; of the 4 byte word.
swap d0
move d7,d0          ; Move the X pos we've calculated into the right.
move.l d0,32(a0)    ; Move the X/Y pos in the 9th position.

; We're now ready to run the GPU.
lea texter,a0
jsr gpurun
jmp gpuwait

```

Listing .2: The `context` routine responsible for setting up the values in our GPU buffer.

`context`, given above, stuffs all these items into a buffer of GPU RAM called `in_buf` and when that is done loads and runs a separate little GPU module called a 'shader'. In this case the name of the shader program is `texter`, giving us a little clue to its domain of expertise. This shader is a free-standing piece of code written in a slightly different flavour of 68000 assembly language. Despite its name `texter` is contained in a source file called `stoat.gas`. (Nearly all of the shaders, like the main program source file, are named after beasts of one sort or another. There is no logic or convention that can be applied to the names. They're just a bit of fun.)

```

texter:
.incbin "bin/stoat.o"

```

`texter` is slightly more than just a 'writing strings to the screen' routine. In addition to deforming the text as we saw in the 'shear' example above, it can also 'explodes' the text using the 'scale' parameters. A comment near the top of `stoat.gas` (the source file for the `texter` routine) tells the possible inspiration for this effect:

```

; *****
; *
; * REX: Robotron explosion generator. Takes an image from the source screen
; *      and expands
; *      it in X and Y, then uses a1_n to draw the resultant matrix of single
; *      pixels.
; *
; *      Provide: dest screen in gpu_screen, in_buf: 0=source image address
; *      4=source image start pixel address, 8=x and y size of source, 12=X scale
; *              (16:16),
; *      16=Y scale (16:16), 20=X shear (16:16), 24=Y shear,
; *      28=Mode (0=Top edge, 1=Centered), 32=Dest X and Y
; *
; *****

```

The 'robotron explosion' is the effect visible below. As you gradually dial up the X and Y scale parameters and call this shader at each iteration it can appear as though the text is 'exploding' outwards. There is no real mystery to how the shader writes our characters to the screen. It uses the same blitting operation we covered previously but with the difference that the preparation of the blitting command and all the special calculations we might want to do if adding shear, scaling and drop shadows in advance of calling the blitter are done from within the special purpose processor of the GPU rather than in the CPU. The reason for doing it this way is because the GPU is designed to be fast at a small number of arithmetic operations such as adding and multiplying so if we're going to do a lot of them and don't want to slow the rest of the game down, the GPU is the man for our job. We'll next take a look at the internals of the `texter` shader, including the relatively trivial task of blitting our characters to the screen, in the next chapter.



Figure .5: The explosion effect in 'Robotron 2084'.

my first shader

The busy heart of the `texter` shader is its `textloop`. This is a simple, but not especially short, piece of code that will write an individual character to the screen, dilating, distorting and deforming it as required. The `textloop` will keep doing this until the string in `textptr` runs out of characters, i.e. hits a `$00` byte such as at the end of `ataricop1`:

```
ataricop1: dc.b "COPYRIGHT 1981,",0
```

This 'have we run out of characters to paint' logic is at the very start of the loop:

```
textloop:
    loadb (textptr),r0 ; Put our text in r0.
    movei #ntxt,r1      ; Put our cleanup-and-exit routine in r1.
    addq #1,textptr     ; Move to the next character in textptr
    cmpq #0,r0          ; Check if it's $00
    jump eq,(r1)        ; If it is, jump to our cleanup-and-exit routine.
```

In this instance `textptr` is pointing to `ataricop1`. As soon as we hit the 0 at the very end we will jump to `ntxt`, which still stop the shader and bail out - unless 'drop shadow' mode was selected: in which case it will paint the string a second time (slightly offset) to achieve the dropshadow effect:

```
ntxt:
    cmpq #1,mode       ; In drop-shadow mode?
    jr ne,StopGPU      ; No, bail.
    nop
    moveq #0,mode       ; Turn off drop-shadow flag so we bail next time.
    movei #textloop,r0  ; Put our textloop routine in r0.
    movefa r6,textptr   ; Rest textptr to the start of the string.
```

```
movefa r7,dstxy      ; Reset our destination position to its original.  
jump (r0)           ; Jump to r0 (textloop)
```

Listing .1: In fact there is no offset for drop-shadow mode implemented so the feature is unused. I suspect this is because the code is copy-pasted from elsewhere as we shall encounter a cousin of this routine later on.

Actually stopping the GPU involves clearing its command set to zero and loading that to the command register:

```
StopGPU:  
    movei #G_CTRL,r1    ; Get our GPU commands.  
    load (r1),r0          ; Load them to r0.  
    bclr #0,r0            ; Clear the GPU flags in r0 to make it stop.  
    store r0,(r1)          ; Actually stop the GPU by loading the cleared flags.  
stoploop:  
    jr stoploop          ; Spin until the GPU actually stops.
```

Now that we see how our loop through the text starts and exits, we will obviously be curious to know how the 'writing to the screen' part is effected. This is nearly identical to the method we stepped through for the 'Blitter'. The moment that pixels are written to the screen is in these statements at the end of `textloop`:

```
movei # (SRCEN|CLIP_A1|UPDA1F|UPDA1|UPDA2|LFU_A|LFU_AN|DCOMPEN) ,r0  
...  
store r0,(blit)       ; draw the sprite
```

Here `blit` is the equivalent of the `B_CMD` register we encountered in blitting. This is defined earlier on in the routine:

```
movei #B_CMD,blit     ; Use blit as an alias for B_CMD
```

By writing the contents of the `r0` register to `blit` we are initiating a paint to the screen using all of the data we have setup earlier in the loop. This leads naturally to the question: what data? Well, not unlike 'Blitting' we set up a bunch of parameters and data points defining our 'source' (`A1`) and our 'destination'(`A2`). For both, these parameters and data points are concerned with things like positon, dimensions, screen width, and so on, not to mention where we are going to get the pixels for drawing from (in our case we already know that this is going to be from `pic`, the variable point to the contents of `beasty4.cry`).

We'll see how this data is calculated and derived shortly but first we'll take a look at the parameters and data points that are fed to the GPU and how. Below we have the GPU blitter registers that control the way our pixel data is written to the screen, in the order in which our `texter` routine writes them:

Common Name	T2K Name	Address in GPU RAM	Sample Value	Description
A1_PIXEL	a1_n+_pixel	\$F0220C	00000001	Current X Position in the Pixel Data
A1_FPIXEL	a1_n+_fpxipel	\$F02218	00000001	Current Y Position in the Pixel Data
A1_STEP	a1_n+_step	\$F02210	00000001	Current Y Position in the Pixel Data
A1_FSTEP	a1_n+_fstep	\$F02214	00000001	Current Y Position in the Pixel Data
A1_INC	a1_n+_inc	\$F0221C	00000001	Current Y Position in the Pixel Data
A1_FINC	a1_n+_finc	\$F02220	00000001	Current Y Position in the Pixel Data
A1_FLAGS	a1_n+_flags	\$F02204	00000001	Current Y Position in the Pixel Data
A1_BASE	a1_n	\$F02200	00000001	Current Y Position in the Pixel Data
A1_CLIP	a1_n+_clip	\$F02208	00000001	Current Y Position in the Pixel Data

The GPU blitter registers in the order in which `texter` populates them in the listing below.

And here is the relevant code in the `texter` populating the registers:

```

blit      REGEQU r13
a1_n      REGEQU r14
a2_n      REGEQU r15
...
rex:
...
movei #A1_BASE,a1_n                         ; Make a1_n A1_BASE
...
store xx,(a1_n+_pixel)                      ; Write to A1_PIXEL
store yy,(a1_n+_fpxipel)                    ; Write to A1_FPIXEL
store r0,(a1_n+_step)                       ; Write to A1_STEP
store r1,(a1_n+_fstep)                      ; Write to A1_FSTEP
store xinc,(a1_n+_inc)                      ; Write to A1_INC
store yinc,(a1_n+_finc)                      ; Write to A1_FINC

movei #gpu_screen,r0           ; Prep gpu_screen as our destination
movei #(PITCH1|PIXEL16|WID384|XADDINC),r1   ; Prep r1 for A1_FLAGS
load (r0),r31                           ; Store gpu_screen in r31

store r1,(a1_n+_flags)                  ; Write flags to A1_FLAGS
movei #$1180180,r1                     ; Prep clip value.
store r31,(a1_n)                      ; Write gpu_screen to A1_BASE
store r1,(a1_n+_clip)                 ; Write clip value to A1_CLIP

```

This takes care of how we want the pixels to be written. On the other side of the equation we have to define for the GPU how want the source data in `beasty7.cry` to be read. Here is the order in which `texter` populates this other side of the scale:

Common Name	T2K Name	Address in GPU RAM	Value	Description
A2_BASE	a2_n	\$F02224	_base	Address of the source data
A2_PIXEL	a2_n+_pixel	\$F02230	spixel	Address of the pixel data in <code>pic5/beasty7.cry</code>
A2_STEP	a2_n+_step	\$F02234	Derived from ssize	The X and Y values for stepping
A2_FLAGS	a2_n+_flags	\$F02228	Calculated	Current Y Position in the Pixel Data

The GPU blitter registers in the order in which `texter` populates them in the listing below.

And here we are doing the populating:

```
store _bass,(a2_n)           ; Write source address to A2_BASE
store spixel,(a2_n+_pixel)   ; Write pixel data address to A2_PIXEL
move ssize,r0                 ; Calculate the step size
and lomask,r0
neg r0
and lomask,r0
bset #16,r0                  ; Set the width of the character
movei #(PITCH1|PIXEL16|WID320|XADDPPIX),r1 ; Prep flags
store r0,(a2_n+_step)         ; Write step size to A1_STEP
store r1,(a2_n+_flags)        ; Write flags to A1_FLAGS
```

So far we've worked backwards from writing to the screen and got to the point where we completed the necessary admin of stuffing our source (A2) and destination (A1) registers with the necessary odds and ends the GPU needs to write a character of our string to the screen. We've even received some clue as to what some of this data may actually mean.

The value in `_bass` that we write to `A2_BASE` is the value at the very start of our `afont` data structure, in other words `pic2`, the contents of `beasty7.cry` where our font pixels live. This is the blob of data the GPU will index into for its pixels to actually paint:

```
afont:
dc.l pic2      ; pixel data
```

We loaded this from the `in_buf` passed to `texter` by first storing it in `fontbase` and then passing it into `_bass`:

```
movei #in_buf,r0    ;load the parameters into registers
load (r0),textptr    ;point at start of textstring
addq #4,r0
load (r0),fontbase   ;point to start of font datastructure
...
load (fontbase),_bass
```

The value `spixel` we write to `A2_PIXEL` is the X and Y position of the character we want to paint in `beasty7.cry/pic2`. We get this value by indexing the ASCII value of our current character into our `afont` data structure.

```
textloop:
loadb (textptr),r0 ; Put our text in r0.
movei #ntxt,r1      ; Put our cleanup-and-exit routine in r1.
addq #1,textptr     ; Move to the next character in textptr
cmpq #0,r0          ; Check if it's $00
jump eq,(r1)         ; If it is, jump to our cleanup-and-exit routine.
nop                ; It's no zero so we can get the x/y value for it from
      afont.
```

```

subq #32,r0           ; ASCII 0-31 do NOT print
shlq #2,r0
add fontbase,r0       ; Add the fontbase address to our offset so we get the
                      ; actual address afont
load (r0),spixel     ; Write the actual address to spixel.

```

In the above, if our character is A, for example, this will result in us pulling out the following entry in afont

```

afont:
...
dc.l $a40001 ;A

```

As we saw earlier this contains the Y (a4) and X (01) position of our A character in the beasty7.cry image:



Figure 1: The 'A' glyph located in a red box by our dimension and co-ordinate information.

unused stars

Tempest 2000 is full of starfields, from the title screen to the game itself - there's always a field of stars up to something in the background.



Figure .1: Example of starfield effects. The middle one is not used by the game.

We'll start in totally the wrong place by looking at piece of code that actually goes unused in the game. A routine called `initstarfield` that populates a random space of stars. This was probably a first iteration at drawing a star-spangled background and dropped later when fancier alternatives were developed. Before we look at the fancy, and necessarily more complex alternatives, let's take a look at what populating a random-ish starfield looks like.

What we have to do first is populate a data structure for our starfield. This is a list (or array) of elements with each element containing an X, Y, and Z co-ordinate plus a color value for the star. There are 127 such elements in total, and we store the number of elements as a 'header' or first entry in the array. We'll store this array in an address called `field1` and do the work of populating it in a routine called `initstarfield`.

The values we come up with for X and Y are totally random. All we do is come up with a number between 0 and 127 for the X and Y co-ordinates and a number between 0 and 512 for our Z co-ordinate. We then use the X and Y co-ordinates to seed a color value.

Here is the routine. As we said before, it ended up unused but has the merit of being relatively simple to read and understand:

```
initstarfield:  
;  
; initialise a starfield data structure for the GPU to display  
  
    lea field1,a0      ; field1 is where the data structure is stored  
    move #127,d7       ; The number of times to loop through 'isf' below.  
    move.l #128,(a0)+ ; Store the number of stars at the start of field1.  
; Create 128 stars and store them as an array in field1.  
; Each star is: X,Y,Z,cry_index  
isf:  
    bsr rannum        ; Get a random number between 0 and 255 and store in d0  
    move d0,d2          ; Stow d0 in d2 for use later  
    sub #$80,d0         ; Get rid of the high bit so the num is between 0 and 128  
    swap d0             ; Turn e.g. 00000032 into 00320000  
    move.l d0,(a0)+     ; Store our rand num as our X co-ordinate.  
  
    bsr rannum        ; Get a random number between 0 and 255 and store in d0  
    move d0,d3          ; Stow d0 in d3 for use later  
    sub #$80,d0         ; Get rid of the high bit so the num is between 0 and 128  
    swap d0             ; Turn e.g. 00000032 into 00320000  
    move.l d0,(a0)+     ; Store our rand num as our Y co-ordinate.  
  
    bsr rannum        ; Get a random number between 0 and 255 and store in d0  
    asl #1,d0           ; Multiply the number by 2 so that its between 0 and 512  
    swap d0             ; Turn e.g. 00000032 into 00320000  
    move.l d0,(a0)+     ; Store our rand num as our Z co-ordinate.  
  
; Use our X and Y co-ordinates to come up with a random color for the star  
.  
; So if X(d2) is 00000032 and Y(d3) is 00000088:  
and #$f0,d2 ; 00000032 -> 00000030  
lsl #4,d3   ; 00000088 -> 80000008  
and #$0f,d3 ; 80000008 -> 00000008  
or d2,d3   ; 00000030 or 00000008 -> 00000038  
lsl #8,d3   ; 00000038 -> 000003800  
move d3,(a0) ; Store our color value.  
  
    lea 20(a0),a0 ; Move a0 20 bytes ahead ready for the next element.  
    dbra d7,isf    ; Loop until d7 is 0.  
    rts
```

Listing .1: Populating an unused starfield data structure. This is a fuzzier version of the ring starfield used in the credits screen.

This is our data structure (or at least the first 2 elements in it) in a table:

Value	Decimal	Description
0000000E	127	Number of Elements
00064000	127	Element 1: X co-ordinate
00064000	127	Element 1: Y co-ordinate
00064000	127	Element 1: Z co-ordinate
00064000	127	Element 1: Color Value
00064000	127	Element 2: X co-ordinate
00064000	127	Element 2: Y co-ordinate
00064000	127	Element 2: Z co-ordinate
00064000	127	Element 2: Color Value

First 2 elements of the data structure created by `initstarfield`.

And this is what our data structure looks like when its animated by our shader `fastvector` (more of which later).




Figure .2: Frames of the unused starfield during animation.

The actual starfield used in the credits screen makes use of a sine table to accurately calculate the co-ordinates of pixels in an expanding ring structure.

```
sines:
dc.w $0003,$0609,$0C0F,$1215,$181B,$1E21,$2427,$2A2D
dc.w $3033,$3639,$3B3E,$4144,$4649,$4B4E,$5053,$5557
dc.w $595C,$5E60,$6264,$6667,$696B,$6D6E,$7071,$7274
dc.w $7576,$7778,$797A,$7B7B,$7C7D,$7D7E,$7E7E,$7E7E
dc.w $7E7E,$7E7E,$7E7E,$7D7D,$7C7B,$7B7A,$7978,$7776
dc.w $7573,$7271,$6F6E,$6C6B,$6967,$6563,$615F,$5D5B
dc.w $5957,$5552,$504D,$4B48,$4643,$403E,$3B38,$3533
dc.w $302D,$2A27,$2421,$1E1B,$1815,$120F,$0C08,$0502
```

```

dc.w $00FD,$FAF7,$F4F0,$EDEA,$E7E4,$E1DE,$DBD8,$D5D2
dc.w $CFC7,$CAC7,$C4C1,$BFBC,$B9B7,$B4B2,$B0AD,$ABA9
dc.w $A6A4,$A2A0,$9E9C,$9A98,$9795,$9392,$908F,$8D8C
dc.w $8B8A,$8988,$8786,$8584,$8483,$8382,$8282,$8282
dc.w $8282,$8282,$8283,$8383,$8485,$8586,$8788,$898A
dc.w $8B8D,$8E8F,$9192,$9496,$9799,$9B9D,$9FA1,$A3A5
dc.w $A7AA,$ACAE,$B1B3,$B6B8,$BBBD,$C0C3,$C5C8,$CBCE
dc.w $D1D4,$D7D9,$DCDF,$E2E6,$E9EC,$EFF2,$F5F8,$FBFE
dc.w $0000,$0192,$0323,$04B5,$0645,$07D5,$0963,$0AFO
dc.w $0C7C,$0E05,$0F8C,$1111,$1293,$1413,$158F,$1708
dc.w $187D,$19EF,$1B5C,$1CC5,$1E2A,$1F8B,$20E6,$223C
dc.w $238D,$24D9,$261F,$275F,$2899,$29CC,$2AFA,$2C20
dc.w $2D40,$2E59,$2F6B,$3075,$3178,$3273,$3366,$3452
dc.w $3535,$3611,$36E4,$37AE,$3870,$3929,$39DA,$3A81
dc.w $3B1F,$3BB5,$3C41,$3CC4,$3D3D,$3DAD,$3E14,$3E70
dc.w $3EC4,$3F0D,$3F4D,$3F83,$3FB0,$3FD2,$3FEB,$3FFA
dc.w $3FFF,$3FFA,$3FEB,$3FD2,$3FB0,$3F83,$3F4D,$3F0D
dc.w $3EC4,$3E70,$3E14,$3DAD,$3D3D,$3CC4,$3C41,$3BB5
dc.w $3B1F,$3A81,$39DA,$3929,$3870,$37AE,$36E4,$3611
dc.w $3535,$3452,$3366,$3273,$3178,$3075,$2F6B,$2E59
dc.w $2D40,$2C20,$2AFA,$29CC,$2899,$275F,$261F,$24D9
dc.w $238D,$223C,$20E6,$1F8B,$1E2A,$1CC5,$1B5C,$19EF
dc.w $187D,$1708,$158F,$1413,$1293,$1111,$0F8C,$0E05
dc.w $0C7C,$0AFO,$0963,$07D5,$0645,$04B5,$0323,$0192 ; <- sine ($0C7C)
dc.w $0000,$FF6E,$FDDD,$FC4B,$FABB,$F92B,$F79D,$F610
dc.w $F484,$F2FB,$F174,$EFEF,$EE6D,$ECED,$EB71,$E9F8
dc.w $E883,$E711,$E5A4,$E43B,$E2D6,$E175,$E01A,$DEC4
dc.w $DD73,$DC27,$DAE1,$D9A1,$D867,$D734,$D606,$D4E0
dc.w $D3C0,$D2A7,$D195,$D08B,$CF88,$CE8D,$CD9A,$Ccae
dc.w $CBCB,$CAEF,$CA1C,$C952,$C890,$C7D7,$C726,$C67F
dc.w $C5E1,$C54B,$C4BF,$C43C,$C3C3,$C353,$C2EC,$C290
dc.w $C23C,$C1F3,$C1B3,$C17D,$C150,$C12E,$C115,$C106 ; <- cosine ($C23C)
dc.w $C101,$C106,$C115,$C12E,$C150,$C17D,$C1B3,$C1F3
dc.w $C23C,$C290,$C2EC,$C353,$C3C3,$C43C,$C4BF,$C54B
dc.w $C5E1,$C67F,$C726,$C7D7,$C890,$C952,$CA1C,$CAEF
dc.w $CBCB,$CCAЕ,$CD9A,$CE8D,$CF88,$D08B,$D195,$D2A7
dc.w $D3C0,$D4E0,$D606,$D734,$D867,$D9A1,$DAE1,$DC27
dc.w $DD73,$DEC4,$E01A,$E175,$E2D6,$E43B,$E5A4,$E711
dc.w $E883,$E9F8,$EB71,$ECED,$EE6D,$EFEF,$F174,$F2FB
dc.w $F484,$F610,$F79D,$F92B,$FABB,$FC4B,$FDDD,$FF6E

```

A table like this allows us to 'hard code' the points of an arbitrarily sized circle by providing the sine and cosine values for each point. To get the X and Y co-ordinates of a specific point in the circle you pull out the corresponding sine and cosine values from the table, multiply each by your desired radius, and that gives you your X and Y value to draw your point on the screen. You do that for each point (say 32 in total) and you have a ring of dots that you can connect to form a circle.

$$x = \cos(\theta) * radius \quad (1)$$

$$y = \sin(\theta) * radius \quad (2)$$

The `sines` table seems inscrutable at first glance but the specific use of it made by the `ringstars` routine is straightforward. The routine plots a circle of 32 points and for each point it uses the current point number (e.g. 31) to pull out a sine and cosine value from the first entry in the corresponding row on the table. If we imagine we're plotting the 31st point this means for the sine value, we will take the first value from row 31, and for the cosine value it will take the first value 8 rows after that. As you can see in the above table this corresponds to \$0C7C for the sine and \$C23C for the cosine.

With this in hand, and a radius of 200 pixels, we can calculate our X and Y co-ordinates. The steps are the same for both the X and Y values, except for the use of `sine` and `cosine`:

Step No.	Description	Cosine	Sine
1	Value taken from <code>sine</code> table	\$0C7C	\$C23C
2	Use last byte only	\$007C	\$003C
3	Multiply by radius (200)	\$60E0	\$2EE0
4	Shift Left by 7 bits	\$00307000	\$00177000
5	Final value treated as 16:16 fraction	X:48.28672	Y:23.28672

Steps to calculate the X and Y value for Point 31.

The whole process is admittedly convoluted but the last step may seem especially mysterious. In order to treat the X and Y values as fractions rather than whole numbers, the blitter will split our final value in half and treat the left-hand side as a whole number and the right-hand side as a fraction. So \$00307000, for example, becomes \$0030 and \$7000, which are 48 and 28672 in decimal respectively, giving us a decimal fractional value of 48.28672.

There is another complication we have glossed over in Step 2 above. Since sine and cosine values can be positive or negative we have to know how +/- values are indicated here. The answer is that any value of \$80 or above is treated as negative: so \$FE is -1, \$FD is -2, all the way down to \$81 which is -127 and \$8000 which is -128. In our example both \$7C and \$3C are less than \$80 so both are treated as positive.

Here then is the first half of the `ringstars` routine where the process we've outlined is implemented:

```
ringstars:
;
; 'initialise a starfield of 8 rings of 64 stars each': it says this,
; but its actually 8 rings of 32 stars each.
```

```

move #200,d5          ; the radius we will use for the ring
rst:
  lea field1,a0        ; field1 is where we'll store it
  lea sines,a1          ; our sine table (see later) in a1
  lea p_sines,a2        ; our positive-only sine table in a2
  move #7,d7            ; d7 will track our 8 rings
  move.l #256,(a0)+     ; header of our structure: no. of stars (256)

  move #$0000,d4        ; d4 will contain the star color
ring1:
  move #32,d6            ; d6 will track no. of stars per ring (32)
ring2:
  move d6,d0              ; Put no. of current star in d0
  asl #3,d0                ; Multiply by 8 to get the row in our sine table
  move.b 0(a1,d0.w),d1    ; Get the sine from position d0 in our sine table
  add.b #$40,d0            ; Add offset to index the cos in our sine table
  move.b 0(a1,d0.w),d2    ; Get the cos from position d0 in our sine table.
  ext d1                  ; Chop off everything but the last byte for sine.
  ext d2                  ; Chop off everything but the last byte for cos.
  muls d5,d1                ; Get Y by multiplying sin * radius
  muls d5,d2                ; Get X by multiplying cos * radius
  asl.l #7,d1                ; Shift left 7 bits to create an X value
  asl.l #7,d2                ; Shift left 7 bits to create a Y value
  move.l d1,(a0)+           ; Move X into our star data structure.
  move.l d2,(a0)+           ; Move Y into our star data structure.

```

Listing .2: Populating the data structure for the starfield used in the Tempest 2000 credits screen.

And here is what plotting these points ourselves looks like. Iterating above for a full 32 points, pulling values from our sines table as we go, gives us a simple circle as expected:




Figure .3: Each of the 32 points as plotted using the X/Y values calculated using the procedure above.

The remainder of the routine calculates a Z value for our point in 3D space and this one is even more intricate.

```

; Calculate the Z value.
lsl #2,d0
and #$ff,d0
move.b 0(a1,d0.w),d1
move.b 0(a2,d0.w),d2
and #$f0,d2
lsl #4,d2
ext d1
bpl sposss
neg d1
sposss:
swap d1
clr d1
asr.l #1,d1
clr.l d0
move d7,d0
swap d0
lsl.l #6,d0      ;Z position according to ring no.
add.l d1,d0
move.l d0,(a0)+ ; Move Z into our star data structure

move d4,d0
add d2,d0
move d0,(a0)      ;colour
lea 20(a0),a0
dbra d6,ring2
add #$2000,d4
dbra d7,ring1
rts

```

Listing .3: Calculating the Z and color value for the starfield data structure used in the Tempest 2000 credits screen.

flipper

The data structure as given in the source code is:

```
s_flipper:
    dc.l 4          ; 4 faces in this object, a shaded solid Flipper

    dc.w $f0        ; Face colour - Red.
    dc.w 3,$8000   ; vertex index, intensity.
    dc.w 1,$c000   ; vertex index, intensity.
    dc.w 0,$ff00   ; vertex index, intensity.
    dc.w 0

    dc.w $f4        ; Face colour - Orange.
    dc.w 4,$4000   ; vertex index, intensity.
    dc.w 2,$8000   ; vertex index, intensity.
    dc.w 0,$c000   ; vertex index, intensity.
    dc.w 0

    dc.w $f4        ; Face colour - Red.
    dc.w 5,$8000   ; vertex index, intensity.
    dc.w 1,$c000   ; vertex index, intensity.
    dc.w 0,$ff00   ; vertex index, intensity.
    dc.w 0

    dc.w $f0        ; Face colour - Orange.
    dc.w 6,$4000   ; vertex index, intensity.
    dc.w 2,$8000   ; vertex index, intensity.
    dc.w 0,$c000   ; vertex index, intensity.
    dc.w 0

fverts: dc.w 9,9,5,9,13,9,1,0,17,0,1,18,17,18
```

A little more explanation, taking the first entry in the structure:

```
dc.w $f0      ; Face colour - Red.  
dc.w 3,$8000 ; Index into fverts, followed by intensity value.  
dc.w 1,$c000 ; Index into fverts, followed by intensity value.  
dc.w 0,$ff00 ; Index into fverts, followed by intensity value.  
dc.w 0          ; Padding that rounds up the entry to 16 bytes
```

We can see that the middle three entries start with an index into the array `fverts` we listed above. The index is pair-wise, so for example an index of 3 must be multiplied by 2, and then counting into `fverts` starting from 0 we end up at the 7th element and the pair: 1,0. We can make this more explicit by breaking out the presentation of `fverts` to match the way it's used:


```
fverts: dc.w 9,9    ; Index 0  
        dc.w 5,9    ; Index 1  
        dc.w 13,9   ; Index 2  
        dc.w 1,0     ; Index 3  
        dc.w 17,0   ; Index 4  
        dc.w 1,18   ; Index 5  
        dc.w 17,18  ; Index 6
```

What are these pairs of numbers? They are (x,y) co-ordinates of course. It turns out `vertex` is just a fancy term for a point in two-dimensional space. So we can now begin to discern what this data structure is: it contains 4 sets of 3 vertices each. If you have 3 points in space, you have a triangle. So what we have here is a description of 4 triangles and each triangle is being referred to as a 'face'.

Let's use the indices in the first entry to see what the first face gives us. We already saw how to get from an index of 3 to a vertex of 1,0 above. When we do the other two the three pairs of co-ordinates turn out as: to be as follows:

Index	X	Y	Description
3	1	0	First vertex.
1	5	9	Second Vertex.
0	9	9	Third Vertex.

With these, and our specified colour of red (\$f0), we can make a triangle:



```
dc.w $f0      ; Colour - Red.
dc.w 3,$8000  ; index, intensity.
dc.w 1,$c000  ; index, intensity.
dc.w 0,$ff00  ; index, intensity.
dc.w 0
```

```
dc.w 9,9    ; Index 0
dc.w 5,9    ; Index 1
dc.w 1,0    ; Index 3
```

Figure .1: Our three vertices joined together in a triangle. Notice that we've scaled up our co-ordinates by 10. So (9,9), for example, is given as (90,90).

One triangle is all very nice, but we have a flipper to build. Should we try a second triangle? Let's see how that goes.

The instructions for our new triangle are in the second paragraph of the `s_flipper` data structure:

```
dc.w $f4      ; Face colour - Orange.
dc.w 4,$4000  ; vertex index, intensity.
dc.w 2,$8000  ; vertex index, intensity.
dc.w 0,$c000  ; vertex index, intensity.
dc.w 0
```

When we translate this to co-ordinates from `fverts` we get:

Index	X	Y	Description
4	17	0	First vertex.
2	13	9	Second Vertex.
0	9	9	Third Vertex.

Not wildly different than our previous effort. Putting our vertices for the two faces together in a single table we get:

Face	Vertex 1	Vertex 2	Vertex 3
1	1,0	5,9	9,9
2	17,0	13,9	9,9

Let's see what we get when we add it in:





Figure .2: Adding in our second triangle

OK, things are starting to take shape. Here are the vertices for all four faces together:

Face	Vertex 1	Vertex 2	Vertex 3
1	1,0	5,9	9,9
2	17,0	13,9	9,9
3	1,18	5,9	9,9
4	17,18	13,9	9,9

Let's see what it looks like when we add in the third triangle:




```
dc.w $f4      ; Colour - Red.
dc.w 5,$8000  ; index, intensity.
dc.w 1,$c000  ; index, intensity.
dc.w 0,$ff00  ; index, intensity.
dc.w 0
```

```
dc.w 9,9    ; Index 0
dc.w 5,9    ; Index 1
dc.w 1,18   ; Index 5
```

Figure .3: Adding in our third triangle

And the fourth and final triangle:



```
dc.w $f0      ; Colour - Orange.
dc.w 6,$4000  ; index, intensity.
dc.w 2,$8000  ; index, intensity.
dc.w 0,$c000  ; index, intensity.
dc.w 0
```

```
dc.w 9,9    ; Index 0
dc.w 13,9   ; Index 2
dc.w 17,18  ; Index 6
```

Figure .4: Adding in our final triangle

```
draw_sflipper:
  lea s_flipper,a1
  bra drawsolidxy
```

```
drawsolidxy:
  lea in_buf,a0
```

flipper

```
move.l a1,(a0)+ ; pointer to our s_flipper data structure
move.l d2,(a0)+ ; x position to draw flipper
move.l d3,(a0)+ ; y position to draw flipper
move.l d1,(a0)+ ; z position to draw flipper
move.l d4,(a0)+ ; x position of flipper's origin
move.l d5,(a0)+ ; y position of flipper's origin
move.l d0,(a0)  ; angle

move.l #0,gpu_mode
lea equine,a0
jsr gpurun      ;do clear screen
jsr gpuwait
rts
```

wells

Wells are fun and should be easy to draw. To start with, we can describe a 2d shape of our choice using a bunch of x/y co-ordinates as points.




Figure .1: Make our dots.

Next we can join these together to give us a 2d shape.




Figure .2: Start with a 2d shape..

To make this thing three dimensional all we have to do is project our shape onto some chosen distant position along the Z plane.




Figure .3: ..then make it three dimensional.

Seems simple enough. With this as our objective the data structure below gives us everything we need to create the web. In addition to listing the 16 vertices that make up the web's shape in two dimensions.

```
; Data structure for the 'HEART' well.
; X Co-ordinates
    .BYTE ODA,0A4, 87, 80, 79, 5C, 26, 10      ;HEART
    .BYTE 10, 20, 48, 80,0B8,0E0,0F0,0F0
; Y Co-ordinates
    .BYTE 0E4,0E8,0B7, 80,0B7,0E8,0E4,0B2      ;HEART
    .BYTE 7A, 47, 20, 10, 20, 47, 7A,0B2
```

To help you relate this table to the co-ordinates in our graphs let's list it in decimal instead:

```
; Data structure for the 'HEART' well.
; X Co-ordinates
    .BYTE 218,164,135,128,121, 92, 38, 16      ; HEART
    .BYTE 16, 32, 72,128,184,224,240,240
; Y Co-ordinates
    .BYTE 228,232,183,128,183,232,228,178      ; HEART
    .BYTE 122, 71, 32, 16, 32, 71,122,178
```

```
.SBTTL UTILITY-DRAW WELL SHAPE
DSPHOL:
    JSR LVLWEL          ;SET UP WELL INDEX & ID
    STA SAVEX           ;WELL INDEX
    STX SAVEX           ;CYCLE
    LDA I,0
    STA VGBRIT
    LDA I,5              ;MAKE WELL REALLY SMALL
    JSR VGSCA1
    LDA SAVEX           ;GET CYCLE (TIMES THRU ALL WELLS
    AND I,7
    TAX
    LDY X,SPWEKO        ;GET SPECIAL WELL COLOR FOR CYCLE
    STY COLOR
    LDA I,MZCLOL
    JSR VGSTAT          ;SET WELL COLOR
    LDX WELLID
    LDA SAVEX
    LDY X,HOLRAP
    IFEQ               ;PLANAR?
    SEC                ;NO. START BEAM AT FIRST POINT
    SBC I,OF             ;IN TABLE (FOR CLOSED WELLS)
    ENDIF
    TAY
    LDA Y,NEWLIZ
    STA PYL
    EOR I,80             ;ADJUST Z SIGN
    TAX
```

```
LDA Y,NEWLIX           ;SAVE COORDS OF 1ST PT
STA PXL
EOR I,80               ;ADJUST X SIGN
JSR VGVTR1             ;POSITION BEAM AT 1ST PT ON WELL
LDA I,OC0               ;TURN BEAM ON
STA VGBRIT
LDX I,NLINES-1
STX INDEX2
BEGIN                 ;LOOP FOR EACH PT ON EDGE
LDY SAVEY
LDA Y,NEWLIX           ;
TAX
SEC
SBC PXL               ;DELTA X
PHA                   ;
STX PXL               ;CURRENT X OLD X
LDA Y,NEWLIZ
TAY
SEC
SBC PYL               ;DELTA Z
TAX
STY PYL               ;CURRENT Z>OLD Z
PLA
JSR VGVTR1             ;DRAW VECTOR TO NEXT PT.
DEC SAVEY
DEC INDEX2
MEND
LDA I,1                ;NORMAL SIZE AGAIN
JMP VGSCA1
```

webs

Webs are fun and should be easy to draw. To start with, we can describe a 2d shape of our choice using a bunch of x/y co-ordinates as points.




Figure .1: Make our dots.

Next we can join these together to give us a 2d shape.




Figure .2: Start with a 2d shape..

To make this thing three dimensional all we have to do is project our shape onto some chosen distant position along the Z plane.




Figure .3: ..then make it three dimensional.

Seems simple enough. With this as our objective the data structure below gives us everything we need to create the web. In addition to listing the 16 vertices that make up the web's shape in two dimensions, we also include some information on whether the web should be open (like the one pictured above) or closed (like a circle or square). Since a web of 16 vertices will give us 14 lanes which the player and enemies can occupy we also add a 'Rotation Table' to tell us how players and enemies should be oriented when on that lane.

```
; Data structure for the 'Sine Wave' web.
web13:
dc.w 14      ; Number of lanes in the web.
dc.w 6       ; The lane the player starts on.

; The x/y pairs of all vertices in the web. So for example
; -2/14 indicates the X and Y co-ordinates of the first vertex.
; There are always 16 in total.
dc.w -2,14,-2,12,-1,10, 1, 8
dc.w  3, 7, 5, 7, 7, 8, 9,10
dc.w 11,12,13,13,15,13,17,12
dc.w 19,10,20, 8,20, 6,-2,14

dc.w 0      ; 0 = Open Web, -1 = Closed Web

; Rotation Table (angle of an object within a particular lane)
; The length of this list is specified by the '14' above.
dc.w -64,-48,-32,-16, 0,16,32,32,16, 0,-16,-32,-48,-64
```

The values given in the Rotation Table are angles of roll to apply to the claw (and to the claw's enemies) when placing them in the lane. Negative values rotate the object to the right, while positive values rotate it to the left.




Figure .4: The rotation table applied to the claw in each lane of the web.

This is the routine that converts the web data structure into a list of vertices and lines between vertices that will make up the web itself:

```
extrude:  
;  
; extrude a web from a list of 16 pairs of XY coordinates addressed by (a1)  
;  
; a0 = vector ram space; a2.l = z depth to extrude to; d0-d7 as above  
  
move.l vadd,a0  
movem.l d0-d7/a0/a2,-(a7)      ;save so routine can return address  
clr connect  
move.l a2,-(a7)      ;save z depth  
bsr initvo          ;make header, do standard vector object init  
move.l a3,a4        ;save first vertex  
move.l (a7)+,d7      ;retrieve z-depth  
move d7,d0  
asr #1,d0  
move d0,web_z       ;Current Web z centering  
clr.l d0  
clr.l d1  
clr d5              ;to catch highest X point  
move (a1)+,d6          ; No of lines in the web.  
move d6,web_max        ; Keep it in web_max  
move (a1)+,web_firstseg ; first position on web  
move.l a1,web_ptab     ;position table  
move.l a3,(a5)+        ;first vertex to lanes list
```

```
; Read in the x/y pairs  
xweb:  
; Get the current x and y pair  
move (a1)+,d0  
move (a1)+,d1      ;get X and Y  
ext.l d0  
ext.l d1  
  
; Check if this is the large X value so far  
cmp d5,d1          ; Compare x with the largest so far, stored in d5.  
blt xweb2          ; If it's less, skip to xweb2 below.  
move d1,d5          ; It's bigger, so save it in d5.  
  
xweb2:  
; Store the x,y,z value for the near point in the web  
move.l d0,(a2)+ ; x value  
move.l d1,(a2)+ ; y value  
clr.l (a2)+      ; z value for near point (always 0)  
  
; Store the x,y,z value for the far point in the web  
move.l d0,(a2)+ ; x value  
move.l d1,(a2)+ ; y value  
move.l d7,(a2)+ ; z value for far point (calculated by initvo).  
  
move d3,(a3)+      ;vertex ID to conn list
```

```
tst d6
beq lastpoint ;special case for last point!

; Connect the vertices
move d3,d4 ;copy vertex #
addq #1,d4
move d4,(a3)+ ;connect to n+1
addq #1,d4
move d4,(a3)+ ;connect to n+2
move #0,(a3)+ ;end vertex
subq #1,d4 ;point to n+1
move d4,(a3)+ ;n+3
addq #2,d4 ;connect
move d4,(a3)+ ;delimit
move.l a3,(a5)+ ;to v.conn list
add #2,d3 ;move 2 vertices

; Get the next pair
dbra d6,xweb
```




Figure .5: Adding in our second triangle

object list

About 60 times a second the Jaguar will want to write something to the screen. A module in the Jaguar system called the 'Object Processor' will take whatever the current list of things to draw happens to be and draw them. It's Tempest 2000's job to keep this beast well fed with new stuff. The way to do this is to keep shovelling fresh imagery into a structure called an 'Object List' for the Object Processor to chew on. As the name suggests this is a list of objects, but since everything is an object these days we might need to be more specific about what these objects are and what they contain.

There are a few different types of objects, but the only one of real interest is the one that contains something that can be displayed: an image. This image takes the form of a section of data similar to the cry pixel data we looked at in 'cry if i want to'. There can be a number of different flavours of this data that make the Object Processor's job easier in different circumstances but in Tempest 2000 we only ever detain ourselves with the rich cry data, the good stuff.

Below we set out the raw contents of an actual Object List used for a single frame in the 'Demo' sequence. The interesting objects in the list are the 'Bit Mapped Objects'. You can tell they're interesting because we have to specify so much about them that we need two 'phrases' (i.e. two sets of 8 bytes each) to encapsulate it all.

Data	Object Type
13,48,00,1D,6E,45,C1,60	Bit Mapped Object (First Phrase)
00,00,80,06,01,80,CF,F8	Bit Mapped Object (Second Phrase)
00,00,00,00,00,00,00,04	Stop Object
00,00,00,00,00,00,00,04	Stop Object
05,00,00,1D,72,0C,01,E0	Bit Mapped Object (First Phrase)
00,00,80,06,01,80,CF,F8	Bit Mapped Object (Second Phrase)

Figure .1: Object List

It is kind of amazing how much you can pack into 16 bytes if you try. Below we parse out the contents of the first 'Bit Mapped Object' in our Object List and show the screen of pixel data it references in the first 8-byte phrase.

Raw Data	Parsed Data	Image at DATA: 0x134800
13,48,00,1D,6E,45,C1,60	TYPE: 0 LINK: 0xeb70 YPOS: 44 DATA: 0x134800 HEIGHT: 279	
00,00,80,06,01,80,CF,F8	XPOS: -7 REFLECT: 0 DEPTH: 4 RMW: 0 PITCH: 1 TRANS: 1 DWIDTH: 96 RELEASE: 0 IWIDTH: 96 FIRSTPIX: 0 INDEX: 0	

Figure .2: First Bit Mapped Object in the Object List

Here is what a selection of the parameters we supply above are a way of saying:

- DATA: 0x134800 → Paint the data found at address 0x134800 in RAM.
- DEPTH: 4 → Treat the data as using 16 bits for each pixel.
- YPOS: 44 → Start painting this image at line 44.
- XPOS: -7 → Start painting this image at X pos -7.
- DWIDTH: 96 → Treat the data as having a line width of 380 (96*4).
- IWIDTH: 96 → Treat the image as having a width of 380 (96*4).
- HEIGHT: 279 → Treat the data as having a height of 280 lines.

You'll notice we have what looks like a lovely bit of glitch at the bottom of our image data.





This handsome scramble is the left-over data from previously drawn frames in the RAM. It is not visible because as we noted above we are painting this image at Y position 49, so although we are painting the full 280 lines of data only the first 231 are visible, and our glorious glitch sits off screen.

These vacant 49 lines at the top of the screen are occupied by our second 'Bit Mapped Object'. This is the player's score and their remaining lives:

Raw Data	Parsed Data	Image in Data 0x134800
05,00,00,1D,72,0C,01,E0	TYPE: 0 LINK: 0xeb90 YPOS: 60 DATA: 0x50000 HEIGHT: 48	
00,00,80,06,01,80,CF,F8	XPOS: -7 DEPTH: 4 PITCH: 1 DWIDTH: 96 INDEX: 0 REFLECT: 0 RMW: 0 TRANS: 1 IWIDTH: 96 FIRSTPIX: 0	

Figure .3: Second Bit Mapped Object in the Object List

We can be flexible about how many Bit Mapped Objects we have and even what we put in them. Here are the two objects we use for the high score screen. The first is the rotating web in the background:

Raw Data	Parsed Data	Image at DATA: 0x100000
10,00,00,1D,6E,45,C1,60	TYPE: 0 LINK: 0xeb70 YPOS: 44 DATA: 0x100000 HEIGHT: 279	
00,00,80,06,01,80,CF,F8	XPOS: -7 DEPTH: 4 PITCH: 1 DWIDTH: 96 INDEX: 0 REFLECT: 0 RMW: 0 TRANS: 1 IWIDTH: 96 FIRSTPIX: 0	

The second is the high-score screen itself:

object list

Raw Data	Parsed Data	Image at DATA: 0x5000
05,00,00,1D,72,45,C1,60	TYPE: 0 LINK: 0xeb90 YPOS: 44 DATA: 0x5000 HEIGHT: 279	 A screenshot from the game TOP GUNS. At the top, it says "TOP GUNS" in large yellow letters. Below that is a red eye icon. The menu displays "1: BEAST (JAK) 50001T LVL 1" and a list of 10 characters with their names and levels: <ul style="list-style-type: none">1: EUE LVL 12: CDE LVL 13: EOU LVL 14: EOX LVL 15: CLK LVL 16: DDC LVL 17: TUE LVL 18: RUE LVL 19: FUR LVL 110: FUR LVL 1 The background is black with a faint grid pattern.
00,00,80,06,01,80,CF,F8	XPOS: -7 REFLECT: 0 DEPTH: 4 RMW: 0 PITCH: 1 TRANS: 1 DWIDTH: 96 RELEASE: 0 IWIDTH: 96 FIRSTPIX: 0 INDEX: 0	

Notice that unlike our previous case we are superimposing these two images (or compositing them) so that the high score appears above the rotating web.

CURSORS

The Tempest source code refers to the player's ship as a CURSOR. It does this because it is the 1980s and that is what you call something on the screen that you can control.

The graphics in Tempest are generated using Atari's 'Quadrascans' vector technology. The images are all defined using a series of vectors. A vector in this case is a (X,Y) value pair that moves the beam in an X,Y direction to a new point on the screen. A series of vectors moving from point to point around the screen will eventually form a complete image.

When we take a look at the data structure defining a relatively upright version of the player's ship we find this. I'm going to speculate that NCSR4S is short for 'New Cursor 4 Start' and NCSR4E for 'New Cursor 4 End':

```
NCSR4S:  
    VEC 3,-2  
    VEC 5,2  
    VEC -3,1  
    VEC 2,-1  
    VEC -4,-1  
    VEC -2,1  
    VEC 2,1  
    VEC -3,-1  
  
NCSR4E:
```

The number pairs in the above listing are the vectors we're talking about, and we're going to see how we can use them to build up an image of the ship assuming an origin point of zero. Our first step, therefore, is to draw a line from (0,0) to (3,-2) as follows:




Figure .1: Draw a line from $(0,0)$ to $(3,-2)$.

Our next step reveals a bit more about the actual nature of the operation we are performing. To draw a line from our new position using a vector of $5,2$ we add it to our current position of $(3,-2)$ to draw a line to $8,0$:





Figure .2: $(3,-2) + (5,2) \rightarrow (8,0)$


With this as our method, we can start building up our complete image, adding each vector in our array to the previous result to define a new line to draw.



$$(8,0) + (-3,1). \longrightarrow (5,1) \quad (5,1) + (2,-1). \longrightarrow (7,0) \quad (7,0) + (-4,-1). \longrightarrow (3,-1)$$

Figure .3: Starting to take shape.


Adding our last three vectors completes the picture, and we have a claw.




$$(3,-1) + (-2,1). \longrightarrow (1,0) \quad (1,0) + (2,1). \longrightarrow (3,1) \quad (3,1) + (-3,-1). \longrightarrow (0,0)$$

Figure .4: Putting the final pieces in place.


Finally, here are all of the CURSOR elements along with their data structures.




NCRS1S: VEC 0,-2 VEC 2,-1 VEC 3,4 VEC -3,-3 VEC -1,0 VEC 0,2 VEC 2,1 VEC -3,-1 NCRS1E:




NCRS2S:
VEC 1,-2
VEC 7,2
VEC -3,1
VEC 2,-1
VEC -6,-1
VEC 0,1
VEC 2,1
VEC -3,-1
NCRS2E:




NCRS3S:
VEC 2,-2
VEC 6,2
VEC -3,1
VEC 2,-1
VEC -5,-1
VEC -1,1
VEC 2,1
VEC -3,-1
NCRS3E:




NCRS4S:
VEC 3,-2
VEC 5,2
VEC -3,1
VEC 2,-1
VEC -4,-1
VEC -2,1
VEC 2,1
VEC -3,-1
NCRS4E:




NCRS5S:
VEC 5,-2
VEC 3,2
VEC -3,1
VEC 2,-1
VEC -2,-1
VEC -4,1
VEC 2,1
VEC -3,-1
NCRS5E:



NCRS6S:
VEC 6,-2
VEC 2,2
VEC -3,1
VEC 2,-1
VEC -1,-1
VEC -5,1
VEC 2,1
VEC -3,-1
NCRS6E:



NCRS7S:
VEC 7,-2
VEC 1,2
VEC -3,1
VEC 2,-1
VEC 0,-1
VEC -6,1
VEC 2,1
VEC -3,-1
NCRS7E:



NCRS8S:
VEC 3,1,0
VEC 3,-4
VEC 2,1
VEC 0,2
VEC -3,1
VEC 2,-1
VEC 0,-2
VEC -1,0
VEC -3,3
NCRS8E:

Claws


In Tempest 2000, for no good reason, the 'classic' tempest claws use trios of bytes to give co-ordinates in 3D space. I say 'for no good reason' because the claw is still flat, so the Z co-ordinate is always a 1. We define our claws using a list of X,Y,Z co-ordinates, and each co-ordinate is paired with a list of points that it connects to. It is this list of connecting points that allows us to draw lines:

```
claw3: dc.b 4,5,1           ; Co-ordinate 1
       dc.b 2,8,0           ; Draw line to co-ords 2 and 8.
       dc.b 4,11,1           ; Co-ordinate 2
       dc.b 3,0             ; Draw line to co-ord 3.
       dc.b 11,12,1          ; Co-ordinate 3
       dc.b 4,0             ; Draw line to co-ord 4.
       dc.b 15,8,1           ; Co-ordinate 4
       dc.b 5,0             ; Draw line to co-ord 5.
       dc.b 12,4,1           ; Co-ordinate 5
       dc.b 6,0             ; Draw line to co-ord 6.
       dc.b 17,8,1           ; Co-ordinate 6
       dc.b 7,0             ; Draw line to co-ord 7.
       dc.b 11,15,1          ; Co-ordinate 7
       dc.b 8,0             ; Draw line to co-ord 8.
       dc.b 2,12,1           ; Co-ordinate 8
       dc.b 0,0             ; Don't draw a line.
```

This is what it looks like when we consume the first co-ordinate and its paired points:


```
claw3: dc.b 4,5,1           ; Co-ordinate 1
       dc.b 2,8,0           ; Draw line to co-ords 2 and 8.
```

This is telling us to connect our first trio 4,5,1 to co-ordinates '2' (4,11,1)..




Draw a line from (4,5,1) to (4,11,1).

.. and '8' (2,12,1).




Draw a line from (4,5,1) to (2,12,1).


With this as our method, we can start building up our complete image, adding each point in our array to the previous result to define a new line to draw.



```
dc.b 4,11,1 ; Co-ordinate 2  
dc.b 3,0 ; Draw line to co-ord 3.  
dc.b 11,12,1 ; Co-ordinate 3
```




```
dc.b 11,12,1 ; Co-ordinate 3  
dc.b 4,0      ; Draw line to co-ord 4.  
dc.b 15,8,1   ; Co-ordinate 4
```




```
dc.b 15,8,1 ; Co-ordinate 4  
dc.b 5,0 ; Draw line to co-ord 5.  
dc.b 12,4,1 ; Co-ordinate 5
```


claws



```
dc.b 12,4,1 ; Co-ordinate 5  
dc.b 6,0 ; Draw line to co-ord 6.  
dc.b 17,8,1 ; Co-ordinate 6
```




```
dc.b 17,8,1 ; Co-ordinate 6  
dc.b 7,0 ; Draw line to co-ord 7.  
dc.b 11,15,1 ; Co-ordinate 7
```



```
dc.b 11,15,1 ; Co-ordinate 7  
dc.b 8,0 ; Draw line to co-ord 8.  
dc.b 2,12,1 ; Co-ordinate 8
```

more claws



The claws for Tempest 2000.

The Tempest 2000 claws are built using the same form of data structure we saw for '[flippers](#)'. So although we don't need to rehearse the details in full again here. It is still worth seeing how a claw is built up from its component polygons.

The data structure for an upright claw in Tempest 2000 mode is as follows:

```
sclaw6:
dc.l 6           ; 4 faces in this object, a shaded solid Flipper

dc.w $ff          ; Face colour - Yellow.
dc.w 0,$ff00      ; vertex index, intensity.
dc.w 1,$ff00      ; vertex index, intensity.
dc.w 2,$c000      ; vertex index, intensity.
dc.w 0

dc.w $fe          ; Face colour - Yellow.
dc.w 1,$ff00      ; vertex index, intensity.
dc.w 2,$c000      ; vertex index, intensity.
dc.w 4,$ff00      ; vertex index, intensity.
dc.w 0

dc.w $fd          ; Face colour - Yellow.
dc.w 2,$c000      ; vertex index, intensity.
dc.w 3,$8000      ; vertex index, intensity.
dc.w 4,$ff00      ; vertex index, intensity.
dc.w 0

dc.w $fd          ; Face colour - Yellow.
dc.w 3,$8000      ; vertex index, intensity.
dc.w 5,$c000      ; vertex index, intensity.
dc.w 4,$ff00      ; vertex index, intensity.
dc.w 0


dc.w $fe          ; Face colour - Yellow.
dc.w 4,$ff00      ; vertex index, intensity.
dc.w 5,$c000      ; vertex index, intensity.
dc.w 6,$ff00      ; vertex index, intensity.
dc.w 0

dc.w $ff          ; Face colour - Yellow.
dc.w 5,$c000      ; vertex index, intensity.
dc.w 6,$ff00      ; vertex index, intensity.
dc.w 7,$ff00      ; vertex index, intensity.
dc.w 0

verts:
dc.w 3,8          ; Index 0
dc.w 4,14         ; Index 1
dc.w 6,13         ; Index 2
dc.w 12,13        ; Index 3
dc.w 13,16        ; Index 4
dc.w 15,7          ; Index 5
dc.w 17,6          ; Index 6
dc.w 11,3          ; Index 7
```


As before we are constructing our claw with a series of triangles. Each claw is made up 6 triangles, and for each triangle we specify 3 vertices.

With these, and our specified colour of yellow (\$ff), we can make a triangle:




```
dc.w $ff      ; Face colour - Yellow.  
dc.w 0,$ff00 ; vertex index, intensity.  
dc.w 1,$ff00 ; vertex index, intensity.  
dc.w 2,$c000 ; vertex index, intensity.  
dc.w 0
```

```
dc.w 3,8      ; Index 0  
dc.w 4,14     ; Index 1  
dc.w 6,13     ; Index 2
```




```
dc.w $fe      ; Face colour - Yellow.  
dc.w 1,$ff00 ; vertex index, intensity.  
dc.w 2,$c000 ; vertex index, intensity.  
dc.w 4,$ff00 ; vertex index, intensity.  
dc.w 0
```

```
dc.w 4,14     ; Index 1  
dc.w 6,13     ; Index 2  
dc.w 13,16    ; Index 4
```




```
dc.w $fd      ; Face colour - Yellow.  
dc.w 2,$c000  ; vertex index, intensity.  
dc.w 3,$8000  ; vertex index, intensity.  
dc.w 4,$ff00  ; vertex index, intensity.  
dc.w 0
```

```
dc.w 6,13  ; Index 2  
dc.w 12,13 ; Index 3  
dc.w 13,16 ; Index 4
```




```
dc.w $fd      ; Face colour - Yellow.  
dc.w 3,$8000  ; vertex index, intensity.  
dc.w 5,$c000  ; vertex index, intensity.  
dc.w 4,$ff00  ; vertex index, intensity.  
dc.w 0
```

```
dc.w 12,13  ; Index 3  
dc.w 15,7   ; Index 5  
dc.w 13,16  ; Index 4
```



```
dc.w $fe      ; Face colour - Yellow.  
dc.w 4,$ff00  ; vertex index, intensity.  
dc.w 5,$c000  ; vertex index, intensity.  
dc.w 6,$ff00  ; vertex index, intensity.  
dc.w 0
```

```
dc.w 13,16 ; Index 4  
dc.w 15,7  ; Index 5  
dc.w 17,6  ; Index 6
```



```
dc.w $ff      ; Face colour - Yellow.  
dc.w 5,$c000  ; vertex index, intensity.  
dc.w 6,$ff00  ; vertex index, intensity.  
dc.w 7,$ff00  ; vertex index, intensity.  
dc.w 0
```

```
dc.w 15,7 ; Index 5  
dc.w 17,6 ; Index 6  
dc.w 11,3 ; Index 7
```


meltovision

'melt-o-vision' was the name given to a neat interstitial feature in Tempest 2000 where the screen blurs, rotates and fades before transitioning to the next screen.



Figure .1: Melt-o-vision in action.

```
; ****
; fade ; Do a meltovision transition.
; This is meltovision.
; https://www.trademarkia.com/meltovision-74534584
; Go into FADE after merging screen3 to current screen and
; turning off BEASTIES+64
; ****
fade:   tst beasties+76           ; Are we ready for meltovision?
        bmi ofade             ; If so, do a meltovision transition.
        ; If not do a merge first.
        move.l #screen3,a0      ; Point a0 at screen3.
        move.l gpu_screen,a1    ; Point a1 at gpu_screen.
        moveq #0,d0              ; X position in source screen
        moveq #0,d1              ; Y position in source screen
        move #384,d2            ; Width
        move #240,d3            ; Height
        add palfix1,d3          ; Adjust for PAL
        moveq #0,d4              ; X position in destination screen.
        moveq #0,d5              ; Y position in destination screen.
        tst mfudj               ; Fudging height required?
        beq pmf2                ; If not, skip.
        sub #8,d3                ; Otherwise fudge source height by 8.




pmfade:  tst mfudj             ; Fudging height required?
        beq pmf2                ; If not, skip.
        add #8,d5                ; Otherwise fudge dest height by 8.
        clr mfudj               ; Clear the fudge factor.

        ; Now we can merge the two screens together.
pmf2:   jsr MergeBlock         ; Merge screen3 and gpu_screen
        move #-1,beasties+76     ; Signal we can do a fade now.

        ; Now we can set up melt-o-vision (tm).
ofade:  clr _pauen            ; Can't pause in fade
        move.l #ffade,demo_routine ; Set meltovision as the demo_routine
        move.l #failcount,routine  ; Set the counter routine.
        move #150,pongx           ; Set the x counter to 150
        ; Get a random value for pongz.
        jsr rannum               ; Random number between 0 and 255 in d0.
        and #$7,d0                ; Just the last 3 bits please.
        sub #$03,d0                ; Subtract 3.
        and #$ff,d0                ; Keep the result between 0 and 255
        move d0,pongz              ; Set as pongz (our z counter).
        move.l #$f80000,delta_i    ; Set our delta intensity.
        move z,-(a7)               ; Stash z in the stack.
        clr z                     ; Clear hard reset signal.
        bsr gogame               ; Run the meltovision transition.
        move.l #screen3,a0          ; Point a0 at screen3.
        jsr clrscreen             ; Clear the screen in a0.
        move (a7)+,z                ; Restore z from the stack.
        move #1,sync                ; Signal a sync is required.
        rts
```

This innovation was so important Atari decided to trademark it. They later decided it wasn't important at all and abandoned the trademark. That's how business works I guess.

In order to scale, rotate and zoom the current screen we need to merge the two layers it's composed of. This means taking `gpu_screen`, which is the 'background', and `screen3`, which is our 'foreground', and compositing them so we have one region of data we can manipulate.

Raw Data	Parsed Data	Image at <code>gpu_screen</code> (0x100000)
10,00,00,1D,6E,45,C1,60	TYPE: 0 LINK: 0xeb70 YPOS: 44 DATA: 0x100000 HEIGHT: 279	
00,00,80,06,01,80,CF,F8	XPOS: -7 REFLECT: 0 DEPTH: 4 RMW: 0 PITCH: 1 TRANS: 1 DWIDTH: 96 RELEASE: 0 IWIDHT: 96 FIRSTPIX: 0 INDEX: 0	
Raw Data	Parsed Data	Image at <code>screen3</code> (0x50000)
05,00,00,1D,72,45,C1,60	TYPE: 0 LINK: 0xeb90 YPOS: 44 DATA: 0x500000 HEIGHT: 279	
00,00,80,06,01,80,CF,F8	XPOS: -7 REFLECT: 0 DEPTH: 4 RMW: 0 PITCH: 1 TRANS: 1 DWIDTH: 96 RELEASE: 0 IWIDHT: 96 FIRSTPIX: 0 INDEX: 0	

We do all of this in the fade routine on the opposite page. Once we have our source and destination co-ordinates ready we can call `MergeBlock` to do the compositing:

```
; Now we can merge the two screens together.  
pmf2: jsr MergeBlock ; Merge screen3 and gpu_screen
```

`MergeBlock`, which we reproduce on the next page, consists of a call to the Jaguar's 'Blitter' with a simple command to draw the contents of `screen3` onto `gpu_screen`.

```

MergeBlock:
    move.l #0,B_PATD      ; Set transparency colour.
    move.l #0,B_PATD+4   ; Set transparency colour

    ; Set up the blitter command.
    move.l #PITCH1|PIXEL16|WID384|XADDINC,d7
    move.l d7,A1_FLAGS   ; Set the Source screen flags.

    move.l #PITCH1|PIXEL16|WID384|XADDPIX|YADD1,d7
    move.l d7,A2_FLAGS   ; Set the Dest screen flags.

    ; d3 is the height, d2 is the width.
    move d3,d7           ; Put d3 (height) in lower word of d7.
    swap d7               ; Swap position of first 2 bytes with last 2.
    move d2,d7           ; Put d2 (width) in lower word of d7.
    move.l d7,B_COUNT    ; Store as inner and outer loop counts

    ; Set up the origin of source screen (x:d0, y:d1).
    move d1,d7           ; Put d1 (Y pos) in lower word of d7.
    swap d7               ; Swap position of first 2 bytes with last 2.
    move d0,d7           ; Put d0 (X pos) in lower word of d7.
    move.l d7,A1_PIXEL   ; Store as X and Y origin of source screen.

    ; Set up the origin of dest screen (x:d4, y:d5).
    move d5,d7           ; Put d5 (Y pos) in lower word of d7.
    swap d7               ; Swap position of first 2 bytes with last 2.
    move d4,d7           ; Put d4 (X pos) in lower word of d7.
    move.l d7,A2_PIXEL   ; Store as X and Y origin of dest screen.

    move.l #0,A1_FPIXEL
    move.l #$0001,A1_INC
    move.l #$0,A1_FINC

    ; Set our width (d2) as the number of loop steps.
    move #1,d7           ; Store 1 in d7
    swap d7               ; Swap position of first 2 bytes with last 2.
    move d2,d7           ; Put d2 (width) in lower word of d7.
    neg d7               ; Make it negative.
    move.l d7,A1_STEP    ; Store as the number of loops for source.
    move.l d7,A2_STEP    ; Store as the number of loops for dest.

    ; Copy the screen data for source and dest.
    move.l a0,d7          ; Copy the source screen (a0) to d7.
    move.l d7,A1_BASE     ; Set it as our source screen.
    move.l a1,d7          ; Copy the destination (a1) to d7.
    move.l d7,A2_BASE     ; Set it as our destination screen.

    ; Run the blitter command to draw one block onto the other.
    move.l #SRCEN|UPDA1|UPDA2|DSTA2|LFU_A|LFU_AN|DCOMPEN,d7
    move.l d7,B_CMD        ; Run the blitter command.
    bra WaitBlit          ; Wait for the blitter to finish.

```

The result is we go from having an object list with two screens, to just one, containing the gpu_screen. Here is what the object list contains with the two screens merged and the melt-o-vision transition from the high score screen just getting started.

Raw Data	Parsed Data	Image at gpu_screen (0x100000)
10,00,00,1D,6E,45,C1,60	TYPE: 0 LINK: 0xeb70 YPOS: 44 DATA: 0x100000 HEIGHT: 279	
00,00,80,06,01,80,CF,F8	XPOS: 4088 REFLECT: 0 DEPTH: 4 RMW: 0 PITCH: 1 TRANS: 1 DWIDTH: 96 RELEASE: 0 IWIDTH: 96 FIRSTPIX: 0 INDEX: 0	

Going back to the fade routine we can see that the next step after merging the two screens is to set up the meltovision transition itself. This consists of setting our GPU manager routine to ffade and a co-routine failcount that will track our progress along the X axis:

```
move.l #fffade,demo_routine ; Set meltovision as the demo_routine
move.l #failcount,routine    ; Co-routine to bail when pongx reaches 0.
```

And determining how much movement along the X axis (pongx) and how much rotation around the Z axis (pongz) the transition will consist of:

```
move #150,pongx           ; Set the x counter to 150
; Get a random value for pongz.
jsr rannum                ; Random number between 0 and 255 in d0.
and #$07,d0                ; Just the last 3 bits please.
sub #$03,d0                ; Subtract 3.
and #$ff,d0                ; Keep the result between 0 and 255
move d0,pongz              ; Set as pongz (our z counter).
```

Notice that we keep our rotation step (which is really an angle value) between 0 and 7, i.e. just the last 3 bits. With this done we can start running the meltovision transition itself:

```
bsr gogame                 ; Run the meltovision transition.
```

gogame, which we won't print here, simply runs in a loop and invokes ffade until failcount has reduced pongx to zero.

```

ffade:
    ; This section ensures we only rotate a little, i.e. 7
    ; individual rotations of our value in pongz.
    add #1,pong      ; Increment pongy.
    move pongy,d0    ; Move it to d0.
    and #$03,d0     ; Select the lowest 3 bits only.
    bne failfade    ; If non-zero, do a step in meltovision.
    tst.b pongz+1   ; Check pongz (out angle of rotation)..
    beq failfade    ; If it's zero, do a meltovision step.
    bmi ffinc       ; If it's negative, increment it.
    sub.b #2,pongz+1; Otherwise decrement it by 2.
ffinc: add.b #1,pongz+1 ; Increment pongz.

    ; Run a step in the meltovision transition.
failfade:
    move.l #(PITCH1|PIXEL16|WID384),d0 ; Set d0 with our flags.
    move.l d0,source_flags ; Set as the source flags.
    move.l d0,dest_flags ; And set as the dest flags.
    lea in_buf,a0        ; Point our GPU RAM input buffer at a0.
    move pongz,d0        ; Put angle of rotation in d0.
    and.l #$ff,d0       ; Keep it between 0 and 255
    move.l cscreen,(a0)  ; Put source screen in GPU buffer.
    move.l #384,4(a0)   ; Store full-screen width in GPU buffer.
    move.l #240,d1       ; Store full-screen height in d1.
    add palfix1,d1      ; Adjust for PAL screens if required.
    move.l d1,8(a0)     ; Store height in buffer.
    move.l #$1f4,12(a0) ; Put X scale in GPU buffer.
    move.l #$1f4,16(a0) ; Put Y scale in GPU buffer.
    move.l d0,20(a0)    ; Put initial angle in brads in GPU buffer.
    move.l #$c00000,24(a0); source x centre in 16:16
    move.l #$780000,d0   ; Set initial y centre position.
    add.l palfix3,d0    ; Adjust for PAL screens if required.
    move.l d0,28(a0)   ; Put y centre in GPU buffer.
    move.l #$0,32(a0)   ; offset of dest rectangle
    move.l delta_i,36(a0); change of i per increment
    ; Select 'scar' in antelope.gas.
    ; Op 2 of this module is Scale and Rotate
    move.l #2,gpu_mode   ; Select 'scar' in antelope.gas
    move.l #demons,a0    ; Load the demons module (antelope.gas).
    jsr gpurun          ; Run it.
    jmp gpuwait         ; Wait until GPU finished.

```

```

failcount:
    sub #1,pongx        ; Subtract 1 from pongx.
    bpl rrrts           ; If still positive we're not done yet.
    move #-1,x_end      ; Signal we're done
    move #1,screen_ready ; Signal screen ready for display.
    rts

```

We can play with a few parameters to alter the effect, and this will help us understand how it works.

For example, if we change our angle of rotation for each step to -16, by changing the following line in `fade` from:

```
move d0,pongz ; Store d0 in pongz.
```

to:

```
move #$f0,pongz ; Store -16 in pongz.
```

And if we increase the number of allowed rotations from 7 (3 bits) in `ffade` (opposite) from:

```
and #$03,d0 ; Select the lowest 3 bits only.
```

to 256 (8 bits):

```
and #$ff,d0 ; Select the whole byte instead.
```

We get an entertainingly rapid rotation of the screen!



Figure .2: Melt-o-vision with a faster rotation.

We can also get rid of the rotation completely and alter the scaling factor to see what happens. Let's do that:

```

move pongz,d0          ; Initial angle.

...
move.l #$1f4,12(a0)    ; X scale.
move.l #$1f4,16(a0)    ; Y scale.

→

move #$00,d0          ; No rotation.
nop
...
move.l #$208,12(a0)    ; X scale.
move.l #$208,16(a0)    ; Y scale.

```

We've increased the scaling factor from \$1f4 to \$208. The results are slightly strange.



Figure .3: Melt-o-vision with a higher scale factor and no rotation.

We get no obvious scaling of the image anymore, except across the central X/Y axis.

If we play with the centre Y position, by increasing it, we observe a different effect.

```

move pongz,d0          ; Initial angle.

...
move.l #$780000,d0      ; Initial Y centre
.

→

move #$00,d0          ; No rotation.
nop
...
move.l #$7F0000,d0      ; Initial Y centre
.

```

With a 'center' north of the centre of the screen the scaling ascends out of the top of the screen.



Figure .4: Melt-o-vision with a higher center factor and no rotation.

bullet

The player's bullet in Tempest 2000 is something of an outlier compared to all the other graphics in the game. For some reason, it is not a polygon constructed from a data structure like the flipper or the player's claw. Instead it is pinched directly from a spritesheet, 'beasty3.cry'. Take a look at the sheet and see if you can spot the bullet:



Figure .1: beasty3.cry, with a bullet in there somewhere - can you find it?

```

; ****
; draw_pixex
; Draws a pixel based explosion.
; A member of the draw_vex and solids list.
; Called during the draw_objects sequence as a member of the draw_vex list.
; Called during the draw_objects sequence as a member of the solids list.
; ****
draw_pixex:
    move.l #4,gpu_mode      ; Use the 'rex' routine in camel.gas.
    lea in_buf,a0            ; Point our GPU RAM input buffer at a0.
    move.l #pic,(a0)+        ; Use 'pic' (beasty3.cry) as our sprite sheet.
    move.l 36(a6),(a0)+      ; Y/X position within the 'pic' sprite sheet.
    move.l 46(a6),(a0)+      ; Width and height to pluck from the sheet.
    move.l 42(a6),d0          ; Put the scale in d0.
    asr.l #2,d0              ; Divide by 4.
    move.l d0,(a0)+          ; Add the x-scale to our GPU RAM input buffer.
    move.l d0,(a0)+          ; Add the y-scale to our GPU RAM input buffer.
    move.l #0,(a0)+          ; Add no shear to the input buffer.
    move.l #0,(a0)+          ; Add Mode 1 = Centered to input buffer.
    move.l 4(a6),d0          ; Get the source X position.
    sub.l vp_x,d0            ; Subtract the player/camera viewpoint X position.
    move.l d0,(a0)+          ; Add it to our GPU RAM input buffer.
    move.l 8(a6),d0          ; Get the object's Y position.
    sub.l vp_y,d0            ; Subtract the player/camera viewpoint Y position.
    move.l d0,(a0)+          ; Add it to our GPU RAM input buffer.
    move.l 12(a6),d0          ; Get the object's Z position.
    sub.l vp_z,d0            ; Subtract the player/camera viewpoint Z position.
    move.l d0,(a0)+          ; Add it to our GPU input buffer.
    lea parrot,a0            ; Load the GPU module in camel.gas.
    jsr gpurun               ; do clear screen
    jmp gpuwait              ; Wait for the GPU to finish.

```

claw says



Above, the routine `draw_pixex` is responsible for drawing the current state of a bullet. Despite its name, it's not always drawing an 'explosion'. Instead it is drawing a segment selected from the sprite sheet 'pic' (beasty3.cry). The routine specifies the co-ordinates and dimensions of the segment to pluck from 'pic' and also details the expected scaling and shearing to apply. As you adjust this scale and shear over time you can achieve all sorts of interesting effects.

Look closely below, in the bottom left hand of the sheet and you can see it in a red box:



Figure .2: The 'bullet' glyph located in the lower left in a red box.

The delightfully named routine '`frab`' is responsible for firing a bullet. In here we can see the co-ordinates for plucking our bullet from the spritesheet being set. We define the X and Y position within the sheet and the width and height too.

```
frab:
; The bullet is taken from the sprite sheet in 'beasty3.cry'.
; This is an exception to the rule for nearly all other objects,
; which are usually constructed polygons.
...
move.l #$00b6000a,d0 ; Y (00b6) and X (000a) in sprite sheet 'pic'.
move.l #$00070007,d1 ; Width (07) and height (07) for the bullet.
...
move.l d0,36(a0)      ; Set the X/Y position in the sprite sheet.
move.l d1,46(a0)      ; Set the width/height in the sprite sheet.
```

Notice how the Y and X values are encoded in a single 4-byte value:

```
move.l #$00b6000a,d0 ; Y (00b6) and X (000a) in sprite sheet 'pic'.
```




Figure .3: Y of 182 (b6) and X of 10 (0a) encoded in a 4 byte long.

Likewise the width and height:

```
move.l #$00070007,d1 ; Width (07) and height (07) for the bullet.
```




Figure .4: Width of 7 (07) and height of 07 (07) encoded in a 4 byte long.

With this setup complete, the routine `draw_pixex` on the opposite page will write these values to the buffer used by the GPU routine 'rex'. With everything we have configured the result will be to simply draw our selected segment from 'beasty3.cry' (i.e. our bullet) to the desired position on the screen.

I cheated earlier by shortening the `frab` listing above. It actually contains an additional detail where we can select a different type of bullet:

```
frab:
; The bullet is taken from the sprite sheet in 'beasty3.cry'.
; This is an exception to the rule for nearly all other objects,
; which are usually constructed polygons.
...
move.l #$00b6000a,d0 ; Y (00b6) and X (000a) in sprite sheet 'pic'.
move.l #$00070007,d1 ; Width (07) and height (07) for the bullet.
tst 32(a0)           ; What type of bullet are we using?
beq konk             ; If default, skip to konk.
; We're using the other bullet!
move.l #$b60000,d0   ; Y (00b6) and X (0000) in sprite sheet 'pic'.
move.l #$090009,d1   ; Width (09) and height (09) for the bullet.
konk:
move.l d0,36(a0)     ; Set the X/Y position in the sprite sheet.
move.l d1,46(a0)     ; Set the width/height in the sprite sheet.
```

For this alternate bullet, we select the sprite to the left of our previous bullet. Notice how the co-ordinates and width/height are subtly altered for this version of the bullet.

```
move.l #$b60000,d0 ; Y (00b6) and X (0000) in sprite sheet 'pic'.
```



Figure .5: Y of 182 (b6) and X of 00 (00) encoded in a 4 byte long.

```
move.l #$090009,d1 ; Width (09) and height (09) for the bullet.
```




Figure .6: Width of 9 (09) and height of 09 (09) encoded in a 4 byte long.

The result is that we instead select the bullet sprite to the left of our original bullet:




Figure .7: The alternate 'bullet' glyph located in the lower left in a red box.

activeobjects

`activeobjects` is a master list of every in-game object on the screen at a given time, with the exception of the player's claw. It makes sense to keep everything in a list like this and have each entry tell us what the game should do with the object, where it should draw it, what type of object it is, and so on. The best way of understanding the use of the `activobjects` list is to see it in action. To do that, we're going to take a look at a snapshot from a game and examine its full `activeobjects` list.

Before we do that, let's take a quick look at what a typical entry in the `activeobjects` list looks like. Each entry in the `activeobjects` list is 64 bytes long. Depending on how many bytes we want to use for each property in an object we could define anything from 16 to 64 properties. In practice, Tempest 2000 has a total of 21 properties, some are 4 bytes (a long), others are 2 bytes (a word).

Below we show a typical entry for a player's bullet.

Bullet	Object Data
	<pre>Draw Routine: draw_pixex YX in Sprite Sheet: 00b6000a X: -22 Colour: 0078 Y: -12 Scale factor: 0001 Z: 47 0 = climb rail, 1 = cross rail, 2 = blowaway: 0000 Web Lane: 2 Width/Height in Sprite Sheet: 00070007 Velocity: 1 Marked for deletion: 0000 Acceleration: 0000ca7e Enemy or Not: not an enemy Roll: 40 run_vex routine: player_shot Pitch: 0 Address of Previous Object: ffffffff Yaw: 0 Address of Next Object: 0000cc38 draw_vex routine: draw_pixex Current Address: 0xcc78</pre>

As you can imagine there are usually a few of these. If you take a look at the components included in the 64 bytes of data representing the object you will see there are a few things whose utility is obvious: we give the X, Y, and Z co-ordinates of the object; which lane it is on, its velocity, and its colour. Some others might seem a bit more recherche: roll, pitch, and yaw, for example, define the angle and shear we apply to the object when drawing it in 3D space. Another way of describing this is the degree of tilt we apply in the X,Y, and Z directions when drawing the object. The most obscure-looking of all are the various 'routines' associated with the object. In the example given above we can see that the routines `draw_pixex` and `player_shot` are defined as properties. These are routines that the game will run when either drawing or updating the state of the object. Lastly we can see that the object contains addresses for the next and previous objects in the list. This is a useful feature when reading the `activeobjects` list - they tell us that when we're finishing processing this particular `activeobject`, which object we should jump to next.

So let's take a look at a random game screen and enumerate the `activeobject` list associated with it. We've chosen a rather busy screen:




Figure .1: Our random screen.


Let's enumerate the entire `activeobjects` list for this screen to give us an idea of how much is being managed at once. We give the objects in the order in which they appear in the list.

Enemy Bullet Object Data




Draw Routine: s_shot Y/X in Sprite Sheet: 00b6000a
 X: -12 Colour: 0018
 Y: -26 Scale factor: 0000
 Z: 100 0 = climb rail, 1 = cross rail, 2 = blowaway: 0000
 Web Lane: 0 Width/Height in Sprite Sheet: 00070007
 Velocity: 0 Marked for deletion: 0000
 Acceleration: 0000ca82 Enemy or Not: enemy
 Roll: 124 run_vex routine: run_ashot
 Pitch: 0 Address of Previous Object: ffffffff
 Yaw: 0 Address of Next Object: 0000d038
 draw_vex routine: draw Current Address: 0xcbf8

Player Bullet Object Data




Draw Routine: draw_pixex Y/X in Sprite Sheet: 00b6000a
 X: 22 Colour: 0058
 Y: -16 Scale factor: 0001
 Z: 194 0 = climb rail, 1 = cross rail, 2 = blowaway: 0000
 Web Lane: 10 Width/Height in Sprite Sheet: 00070007
 Velocity: 1 Marked for deletion: ffff
 Acceleration: 0000ca6e Enemy or Not: not an enemy
 Roll: 376 run_vex routine: player_shot
 Pitch: 0 Address of Previous Object: 0000cdf8
 Yaw: 0 Address of Next Object: 0000cd38
 draw_vex routine: draw_pixex Current Address: 0xcc38

Player Bullet Object Data




Draw Routine: draw_pixex Y/X in Sprite Sheet: 00b6000a
 X: 26 Colour: 0028
 Y: -24 Scale factor: 0001
 Z: 191 0 = climb rail, 1 = cross rail, 2 = blowaway: 0000
 Web Lane: 11 Width/Height in Sprite Sheet: 00070007
 Velocity: 1 Marked for deletion: ffff
 Acceleration: 0000ca72 Enemy or Not: not an enemy
 Roll: 368 run_vex routine: player_shot
 Pitch: 0 Address of Previous Object: 0000ccf8
 Yaw: 0 Address of Next Object: 0000d0b8
 draw_vex routine: draw_pixex Current Address: 0xcc78

Player Bullet Object Data




Draw Routine:	draw_pixex	Y/X in Sprite Sheet:	00b6000a
X:	26	Colour:	0028
Y:	-24	Scale factor:	0001
Z:	191	0 = climb rail, 1 = cross rail, 2 = blowaway:	0000
Web Lane:	11	Width/Height in Sprite Sheet:	00070007
Velocity:	1	Marked for deletion:	ffff
Acceleration:	0000ca76	Enemy or Not:	not an enemy
Roll:	368	run_vex routine:	player_shot
Pitch:	0	Address of Previous Object:	0000d078
Yaw:	0	Address of Next Object:	0000ccf8
draw_vex routine:	draw_pixex	Current Address:	0xccb8

Player Bullet Object Data



Draw Routine:	draw_pixex	Y/X in Sprite Sheet:	00b6000a
X:	26	Colour:	0028
Y:	-24	Scale factor:	0001
Z:	191	0 = climb rail, 1 = cross rail, 2 = blowaway:	0000
Web Lane:	11	Width/Height in Sprite Sheet:	00070007
Velocity:	1	Marked for deletion:	ffff
Acceleration:	0000ca7a	Enemy or Not:	not an enemy
Roll:	368	run_vex routine:	player_shot
Pitch:	0	Address of Previous Object:	0000ccb8
Yaw:	0	Address of Next Object:	0000cc78
draw_vex routine:	draw_pixex	Current Address:	0xccf8

Player Bullet Object Data



Draw Routine:	draw_pixex	Y/X in Sprite Sheet:	00b6000a
X:	22	Colour:	0058
Y:	-16	Scale factor:	0001
Z:	194	0 = climb rail, 1 = cross rail, 2 = blowaway:	0000
Web Lane:	10	Width/Height in Sprite Sheet:	00070007
Velocity:	1	Marked for deletion:	ffff
Acceleration:	0000ca7e	Enemy or Not:	not an enemy
Roll:	376	run_vex routine:	player_shot
Pitch:	0	Address of Previous Object:	0000cc38
Yaw:	0	Address of Next Object:	0000cb78
draw_vex routine:	draw_pixex	Current Address:	0xcd38

Flipper Object Data



Draw Routine:	cdraw_sflipper	Start address of pixel data.:	fff70006
X:	4	Colour:	00f0
Y:	8	Scale factor:	0001
Z:	30	0 = climb rail, 1 = cross rail, 2 = blowaway:	0000
Position/lane on web.:	7	Size of Pixel Data:	c9ce0007
Velocity:	4	Marked for deletion:	0000
Acceleration/Flipper Mode:	00010808	Enemy or Not:	enemy
XZ Orientation (Roll):	96	run_vex routine:	run_flipper
Y Rotation (Pitch):	0	Address of Previous Object:	0000cdb8
Z Rotation (Yaw):	0	Address of Next Object:	0000cb38
draw_vex routine:	draw_vxc	Current Address:	0xcd78



Flipper	Object Data
	<p>Draw Routine: cdraw_sflipper Start address of pixel data.: fff70008 X: 16 Colour: 00f0 Y: -4 Scale factor: 0001 Z: 30 0 = climb rail, 1 = cross rail, 2 = blowaway: 0000 Position/lane on web.: 9 Size of Pixel Data: c9ca0101 Velocity: 4 Marked for deletion: 0000 Acceleration/Flipper Mode: 00010808 Enemy or Not: enemy XZ Orientation (Roll): 36 run_vex routine: run_flipper Y Rotation (Pitch): 0 Address of Previous Object: 0000ce38 Z Rotation (Yaw): 0 Address of Next Object: 0000cd78 draw_vex routine: draw_vxc Current Address: 0xcdcb8</p>


Enemy Bullet	Object Data
	<p>Draw Routine: s_shot Y/X in Sprite Sheet: 009e0000 X: -16 Colour: 00d8 Y: 2 Scale factor: 0000 Z: 30 0 = climb rail, 1 = cross rail, 2 = blowaway: 0000 Web Lane: 4 Width/Height in Sprite Sheet: 00a00001 Velocity: 0 Marked for deletion: ffff Acceleration: 0000ca76 Enemy or Not: enemy Roll: 204 run_vex routine: run_ashot Pitch: 13 Address of Previous Object: 0000cf78 Yaw: 15 Address of Next Object: 0000cc38 draw_vex routine: draw Current Address: 0xcdcf8</p>

Flipper	Object Data
	<p>Draw Routine: cdraw_sflipper Start address of pixel data.: 00000000 X: -12 Colour: 00f0 Y: -26 Scale factor: 0001 Z: 71 0 = climb rail, 1 = cross rail, 2 = blowaway: 0000 Position/lane on web.: 0 Size of Pixel Data: c9d20007 Velocity: 0 Marked for deletion: 0000 Acceleration/Flipper Mode: 00000808 Enemy or Not: enemy XZ Orientation (Roll): 112 run_vex routine: run_flipper Y Rotation (Pitch): 0 Address of Previous Object: 0000ce78 Z Rotation (Yaw): 0 Address of Next Object: 0000cdb8 draw_vex routine: draw_vxc Current Address: 0xce38</p>

Flipper	Object Data
	<p>Draw Routine: cdraw_sflipper Start address of pixel data.: 00000000 X: -12 Colour: 00f0 Y: -26 Scale factor: 0001 Z: 125 0 = climb rail, 1 = cross rail, 2 = blowaway: 0000 Position/lane on web.: 0 Size of Pixel Data: c9ce0000 Velocity: 1 Marked for deletion: 0000 Acceleration/Flipper Mode: 00000808 Enemy or Not: enemy XZ Orientation (Roll): 112 run_vex routine: run_flipper Y Rotation (Pitch): 0 Address of Previous Object: 0000cef8 Z Rotation (Yaw): 0 Address of Next Object: 0000ce38 draw_vex routine: draw_vxc Current Address: 0xce78</p>

activeobjects

Spiker	Object Data
	<p>Draw Routine: 0000b6c4 X: 8 Y: 6 Z: 187 Position/lane on web.: 7 Velocity: 1 Acceleration/Flipper Mode: 0000cff8 XZ Orientation (Roll): 592 Y Rotation (Pitch): 0 Z Rotation (Yaw): 0 draw_vex routine: draw</p> <p>Start address of pixel data.: fffff00a Colour: 008f Scale factor: 0001 0 = climb rail, 1 = cross rail, 2 = blowaway: 0002 Size of Pixel Data: 00070007 Marked for deletion: 0000 Enemy or Not: enemy run_vex routine: run_spiker Address of Previous Object: 0000cff8 Address of Next Object: 0000cfb8 Current Address: 0xceb8</p>
Fuseball	Object Data
	<p>Draw Routine: draw_sfuseball X: -16 Y: -25 Z: 163 Position/lane on web.: 0 Velocity: -1 Acceleration/Flipper Mode: 00004000 XZ Orientation (Roll): 526 Y Rotation (Pitch): 0 Z Rotation (Yaw): 0 draw_vex routine: draw</p> <p>Start address of pixel data.: 00000001 Colour: 000f Scale factor: 0001 0 = climb rail, 1 = cross rail, 2 = blowaway: 0001 Size of Pixel Data: 30300101 Marked for deletion: 0000 Enemy or Not: enemy run_vex routine: run_fuseball Address of Previous Object: 0000cf38 Address of Next Object: 0000ce78 Current Address: 0xcef8</p>
Power-Up	Object Data
	<p>Draw Routine: draw_pup1 X: 14 Y: 0 Z: 29 Position/lane on web.: 8 Velocity: 0 Acceleration/Flipper Mode: 00010808 XZ Orientation (Roll): -48 Y Rotation (Pitch): 0 Z Rotation (Yaw): 0 draw_vex routine: draw_pupx</p> <p>Start address of pixel data.: fff70007 Colour: 0058 Scale factor: 0000 0 = climb rail, 1 = cross rail, 2 = blowaway: 0023 Size of Pixel Data: 0010fff8 Marked for deletion: 0000 Enemy or Not: not an enemy run_vex routine: run_pupx Address of Previous Object: 0000cbb8 Address of Next Object: 0000cef8 Current Address: 0xcf38</p>
Player Bullet	Object Data
	<p>Draw Routine: draw_pixex X: 22 Y: -16 Z: 194 Web Lane: 10 Velocity: 1 Acceleration: 0000ca72 Roll: 376 Pitch: 0 Yaw: 0 draw_vex routine: draw_pixex</p> <p>Y/X in Sprite Sheet: 00b6000a Colour: 0058 Scale factor: 0001 0 = climb rail, 1 = cross rail, 2 = blowaway: 0000 Width/Height in Sprite Sheet: 00070007 Marked for deletion: ffff Enemy or Not: not an enemy run_vex routine: player_shot Address of Previous Object: ffffffff Address of Next Object: 0000cdf8 Current Address: 0xcf78</p>

Object Type	Object Data
Flipper	<p>Draw Routine: cdraw_sflipper X: -16 Y: 2 Z: 55 Position/lane on web.: 4 Velocity: -4 Acceleration/Flipper Mode: 00000808 XZ Orientation (Roll): 16 Y Rotation (Pitch): 0 Z Rotation (Yaw): 0 draw_vex routine: draw_vxc</p>  <p>Start address of pixel data.: 0000ffff Colour: 00f0 Scale factor: 0001 0 = climb rail, 1 = cross rail, 2 = blowaway: 0000 Size of Pixel Data: 00070007 Marked for deletion: 0000 Enemy or Not: enemy run_vex routine: run_flipper Address of Previous Object: 0000ceb8 Address of Next Object: 0000ccb8 Current Address: 0xfcdb8</p>
Spiker	<p>Draw Routine: 0000b694 X: 8 Y: 6 Z: 190 Position/lane on web.: 7 Velocity: 0 Acceleration/Flipper Mode: 00000000 XZ Orientation (Roll): -16 Y Rotation (Pitch): 0 Z Rotation (Yaw): 0 draw_vex routine: draw_spike</p>  <p>Start address of pixel data.: 00576000 Colour: 008f Scale factor: ffff 0 = climb rail, 1 = cross rail, 2 = blowaway: 0000 Size of Pixel Data: c9ca0000 Marked for deletion: 0000 Enemy or Not: not an enemy run_vex routine: run_spike Address of Previous Object: 0000d038 Address of Next Object: 0000ceb8 Current Address: 0xcff8</p>
Flipper	<p>Draw Routine: cdraw_sflipper X: 0 Y: 8 Z: 330 Position/lane on web.: 6 Velocity: 0 Acceleration/Flipper Mode: 00000808 XZ Orientation (Roll): 0 Y Rotation (Pitch): 0 Z Rotation (Yaw): 0 draw_vex routine: fffd</p>  <p>Start address of pixel data.: 00000000 Colour: 00f0 Scale factor: 0001 0 = climb rail, 1 = cross rail, 2 = blowaway: 0000 Size of Pixel Data: 00100000 Marked for deletion: 0000 Enemy or Not: enemy run_vex routine: run_flipper Address of Previous Object: 0000cbf8 Address of Next Object: 0000cff8 Current Address: 0xd038</p>
Player Bullet	<p>Draw Routine: draw_pixex X: 26 Y: -24 Z: 191 Web Lane: 11 Velocity: 1 Acceleration: 0000ca86 Roll: 368 Pitch: 0 Yaw: 0 draw_vex routine: draw_pixex</p>  <p>Y/X in Sprite Sheet: 00b6000a Colour: 0028 Scale factor: 0001 0 = climb rail, 1 = cross rail, 2 = blowaway: 0000 Width/Height in Sprite Sheet: 00070007 Marked for deletion: ffff Enemy or Not: not an enemy run_vex routine: player_shot Address of Previous Object: 0000cb78 Address of Next Object: 0000ccb8 Current Address: 0xd078</p>

rotary club

Sprinkled throughout yak.s, the principal source code file for *Tempest 2000*, we find references to a never-produced 'rotary controller'. This was a device that would emulate the controller present in the coin-operated original of Tempest in 1981.

Towards the end of the Frame interrupt routine we find the main logic used to detect and, if a rotary controller is present, detect its input:

```
; Check for joypad input. The logic for rotary controllers
; was added to support the potential release of a rotary
; controller for the Jaguar, but this never happened.
; So all rotary controller logic in Tempest 2000 is unused.
tst roconon          ; Is Rotary Controller enabled?
bne roco             ; Yes, do the rotary controller.
jsr dopad            ; No, check for normal joypad input.
bra exxit            ; We're done - so exit interrupt.

; Logic for reading input from the never-built and never-
; released rotary controller. We only look for button input
; every 8 interrupts.
roco:   move pitcount,d1      ; Get the rotary input interval counter.
       and #$07,d1           ; Do a modulus 7 on the counter.
       bne rotonly           ; If non-zero, read the rotary control only.
       jsr dopad              ; Otherwise, look for button presses too.
rotonly:
       add #1,pitcount        ; Increment the interval counter.
       bsr readrotary         ; Read input from the rotary controller.
```

You might wonder how it was possible to test a rotary controller that doesn't exist. The answer to this is that you make one of your own. In this case, Yak adapted a an Atari

drive controller marketed for the antiquated Atari 2600 game console and used it as a substitute for the rotary input promised by the Jaguar hardware team.



The humble 2600 Atari drive controller.

Once we've decided we're reading from this hacked-up rotary controller our first task is to hide it behind some secret button presses on the options screen.

```
selector:  
    cmp.l #option2,the_option ; Are we on the 'Options' screen?  
    bne selector2             ; If not, skip.  
    btst.b #6,sysflags        ; Are controllers enabled?  
    bne sopt2                 ; If not, skip.  
    btst.b #4,pad_now         ; Pause pressed by Player 1?  
    beq selector2             ; If not, skip.  
    btst.b #4,pad_now+4       ; Pause pressed by Player 2?  
    beq selector2             ; If not, skip.  
    bset.b #6,sysflags        ; Enable controllers.  
    jsr sayex                ; Say 'Excellent!'.
```

If we have 2 controllers and both press the pause button it enables us to select the controller type and enable the rotary controller:



Selecting and enabling the rotary controller.

Next we have to figure out what it's telling us. For this, and for the ordinary joypad controller too, we inspect a 4-byte phrase (JOYIN) that tells us what buttons or dials have been pressed most recently. These 4 bytes are a bitmap: if a bit is set it tells us a specific button has been pressed.

Button	Hex	Bitmap
abutton	20000000	00100000000000000000000000000000
pausebutton	10000000	00010000000000000000000000000000
bbutton	02000000	00000001000000000000000000000000
right	00800000	00000000100000000000000000000000
left	00400000	00000000001000000000000000000000
down	00200000	00000000000100000000000000000000
up	00100000	00000000000010000000000000000000
seven	00080000	00000000000001000000000000000000
four	00040000	00000000000000100000000000000000
one	00020000	00000000000000010000000000000000
asterisk	00010000	00000000000000010000000000000000
cbutton	00002000	00000000000000001000000000000000
optionbutton	00000200	00000000000000000000000010000000
two	00000080	00000000000000000000000000000000
five	00000040	00000000000000000000000000000000
eight	00000020	00000000000000000000000000000000
zerobutton	00000010	00000000000000000000000000000000
three	00000008	00000000000000000000000000000000
six	00000004	00000000000000000000000000000000
nine	00000002	00000000000000000000000000000000
hash	00000001	00000000000000000000000000000000

All controller buttons and the bit they set in the JOYIN bitmap.

readrotary does this job. It is only interested in a subset of the buttons. These are:

Button	Hex	Bitmap
right	00800000	00000000100000000000000000000000
left	00400000	00000000001000000000000000000000

But the rotary controller has everything in a slightly different position, so when we read JOYIN, we'll find the right and left movement bits 4 bits to the left like so:

Button	Hex	Bitmap
right	08000000	00001000000000000000000000000000
left	04000000	00000100000000000000000000000000

Thus readrotary's first task is to extract the most recently pressed buttons and get

them into the register d0. It must then move everything 4 bits to the right. It does this as follows.

```

readrotary:
    btst.b #3,sysflags      ; Rotary controller enabled for Player 1?
    beq op2                 ; If not, skip to checking Player 2.
    move.l #$f0fffffc,d1   ; d1 = Joypad data mask (Player 1)
    moveq.l #$fffffff,d2   ; d2 = Cumulative joypad reading
    move.w #$81fe,JOYOUT   ; Ask for the most recent button presses.
    move.l JOYIN,d0         ; Read joypad, pause button, A button
    or.l   d1,d0             ; Mask off unused bits
    ; Rotate d0 4 bits to the right so that the bits are where we
    ; expect them to be.
    ror.l #4,d0              ; Shift everything 4 bits to the right.
    and.l d0,d2              ; d2 = xxAPxxxx RLDUxxxx xxxxxxxx xxxxxxxx
    swap   d2                ; Swap position of the first 2 with last 2 bytes.
    move   d2,d0              ; Store result in d0.

    lea lstcon,a1            ; Point a1 at lstcon (last read from controller).
    lea roconsens,a2        ; Point a2 at roconsens.
    bsr rroco                ; Calculate the input movement and store in d0.
    add.l d0,rot_cum          ; Add result to the cumulative movement.

op2:
    ; We omit the duplicated logic for player 2.
    rts

```

Once we have a value to work with, the question is how we decide if it's telling us to move the claw right or left. What we will do is compare the current input with the last input reported by the controller and depending on whether it is before or after the input value's position in the `conseq` array move it right or left, respectively.

```
conseq: dc.b 0,1,3,2
```

We can capture some of these potential outcomes as follows.

	Previous Value	Current Value	Result
1	0		Move Right
1	3		Move Left
3	1		Move Right
3	2		Move Left
2	3		Move Right
2	0		Move Left
0	2		Move Right
0	1		Move Left

It's not exactly intuitive why this is the case.

But anyway, here is the code that does the job.

```

rroco:
    rol.b #2,d0          ; Phase Bits to bottom of word
    and #$03,d0          ; Get juicy bits
    move (a1),d4          ; Put last value read (lstcon) in d4.
    cmp d4,d0            ; Compare to most recent read.
    beq decsens          ; Did not move, so exit early.

    move d0,(a1)          ; Store current read in lstcon.
    lea consequ,a0        ; Point to sequence values
    clr d5                ; Set d5 to 0.


    ; Search sequence values (conseq) for a match to the last value
    ; we read (lstcon).
slocate:
    cmp.b 0(a0,d5.w),d4 ; Get val using d5 as index into consequ.
    beq slocated          ; If it equals last value read (d4), exit loop.
    addq #1,d5            ; Otherwise, increment d5..
    bra slocate           ; .. and loop.

    ; d5 now points to the position in consequ matching the previous
    ; value read from the rotary controller. To determine whether
    ; to move left or right we now see if the currently read value
    ; is before (right) or after (left) the most recent value read
    ; from the controller.
slocated:
    subq #1,d5            ; Subtract 1 from d5.
    and #3,d5              ; Clamp to 0 - 7.
    cmp.b 0(a0,d5.w),d0 ; Use it as index into consequ.
    beq rclaw_right        ; If val is equal to d0(curr read), move right.
    addq #2,d5            ; Add two to d5
    and #3,d5              ; Clamp to 0 - 7.
    cmp.b 0(a0,d5.w),d0 ; Use it as index into consequ.
    beq rclaw_left         ; If val is equal to d0 (curr read), move left.
    clrl d0                ; Clear d0.
    rts                   ; Return

```


things hidden

These appear to be enemy attack ships of a previously unknown configuration defined in the source file ALVROM.MAC. An assembly flag excluded these objects from the final release of the game but their full specification is available to us in a series of vector commands. This has enabled the us to painstakingly reconstruct the artefacts using modern equipment. We present them here to the reading public for the first time.



ENER21 to ENER24 in ALVROM.MAC.

The best of our finds is a clear predecessor to the iconic 'claw' ship. Each is defined using an array of X/Y co-ordinates that a macro by the name of CALVEC encodes into

a list of vector commands. For example, the first image in Figure 2 above is given as follows in [lines 1483-1512](#) in ALVROM.MAC:¹

```

ENER21:
  ICALVE          ; X:0   Y:0
  CALVEC -1,-3    ; X:-1  Y:-3
  .BRITE=VARBRT   ; Set brightness to 1
  CALVEC -4,24.    ; X:-4   Y:36
  CALVEC -24.,0    ; X:-36  Y:0
  CALVEC -12.,-40. ; X:-18  Y:-64
  CALVEC -20.,-2   ; X:-32  Y:-2
  CALVEC -15.,3    ; X:-21  Y:3
  CALVEC -10.,-12. ; X:-16  Y:-18
  CALVEC -12.,6    ; X:-18  Y:6
  CALVEC -5,13.    ; X:-5   Y:19
  CALVEC -3,-24.   ; X:-3   Y:-36
  CALVEC -1,-3     ; X:-1   Y:-3
  .BRITE=0         ; Set brightness to 0
  CALVEC 1,-3      ; X:1   Y:-3
  .BRITE=VARBRT   ; Set brightness to 1
  CALVEC 3,-24.    ; X:3   Y:-36
  CALVEC 5,13.     ; X:5   Y:19
  CALVEC 12.,6     ; X:18  Y:6
  CALVEC 10.,-12.   ; X:16  Y:-18
  CALVEC 15.,3     ; X:21  Y:3
  CALVEC 20.,-2    ; X:32  Y:-2
  CALVEC 12.,-40.  ; X:18  Y:-64
  CALVEC 24.,0     ; X:36  Y:0
  CALVEC 4,24.     ; X:4   Y:36
  CALVEC 1,-3      ; X:1   Y:-3
  .BRITE=0         ; Set brightness to 0
  CALVEC NXE,0     ; X: 0   Y:0
  RTSL

```

The listing gives X and Y co-ordinates in hex, which we can readily plot as vertices on a graph. During assembly these values were converted to '[relative draw](#)'² vector commands for use by the Atari Analogue Vector Generator (AVG). These encode X and Y vectors, along with an intensity value I as follows:

Vector Command Bits											
X	Y	I	000Y	YYYY	YYYY	YYYY	IIIIX	XXXX	XXXX	XXXX	
FF	FD	00	0001	1111	1111	1100	0001	1111	1111	1111	

The values chosen above are not arbitrary: X is -1 (FF) and Y is -3 (FD). Together with an assumed intensity value of 0 these form the first entry in ENER21: CALVEC -1,-3,

¹<https://github.com/historicalsource/tempest/blob/6c783bee488ed736fc3fdc3a81fdc412c3bec386/ALVROM.MAC#L1483>

²https://arcarc.xmission.com/Tech/neilw_xy.txt

which gets encoded in [one's complement](#)³ for the thirteen bits of each value: 1FFD 1FFF.

Let's take that again, step by step. Our command is:

```
CALVEC -1,-3 ; X:-1 Y:-3
```

This calls the macro CALVEC with the values -1 (FE) as X and -3 (FD) as Y. The end result we expect from this operation is a vector command that will draw a line from the current point on the screen to the new co-ordinates. To do this CALVEC manipulates the X and Y values we pass relative to the previous position of X and Y, and offloads the rest of the work to another macro, VCTR:

```
.MACRO CALVEC NEWX,NEWZ
.XN =NEWX-OLDX ; Subtract the previous value of X from the current.
.ZN =NEWZ-OLDZ ; Subtract the previous value of Y from the current.
VCTR .XN,.ZN,.BRITE ; Call the VCTR macro with our new X and Y.
OLDX =NEWX ; Store the current X as our previous value.
OLDZ =NEWZ ; Store the current Y as our previous value.
.ENDM
```

It is VCTR that will transform our humble parameters -1 and 3 into a fully formed 4 byte vector command. This will be 1FFD1FFF, let's see how that translation happens:

```
; VCTR - DRAW VECTOR INSTRUCTION
; THIS INSTRUCTION DRAWS A VECTOR ON THE DISPLAY AREA
; RELATIVE TO THE PREVIOUS BEAM POSITION BEFORE THE
; INSTRUCTION IS EXECUTED. DX IS THE CHANGE IN BEAM X
; POSITION; DY IS THE CHANGE IN BEAM Y POSITION; ZZ
; SPECIFIES THE BEAM INTENSITY (0 THROUGH 7., 0 IS
; NO INTENSITY, 7. IS BRIGHTEST INTENSITY).
;
.SBTTL VCTR
.MACRO VCTR DX,DY,ZZ
...1=DX ; STORE DX AS '1'
...2=DY ; STORE DY AS '2'
.IF LT,...1 ; IF NEGATIVE X
...1=-DX ; MAKE IT POSITIVE
.ENDC ; END IF
.IF LT,...2 ; IF NEGATIVE Y
...2=-DY ; MAKE IT POSITIVE
.ENDC ; END IF
.WORD DY&^H1FFF,<ZZ*>^H2000>+<DX&^H1FFF> ; Turn into a vector
command
.ENDM
```

The crux is the penultimate line:

³https://en.wikipedia.org/wiki/Ones%27_complement

.WORD	DY&^H1FFF ,<ZZ*^H2000>+<DX&^H1FFF>	; Turn into a vector command
-------	------------------------------------	------------------------------

The macro assembler computes this into a single line of assembly:

.WORD	1FFD,1FFF
-------	-----------

Let's see the steps it follows to do this. First we convert each of -1 (FF) and -3 (FD) into 'positive' values:

Command	Left Operand	Operation	Result
...1=-DX	FF	-	00
...1=-DY	FD	-	02

Now we can run the bit operations that will composite everything into a sequence of 4 bytes:


Command	Left Operand	Operation	Right Operand	Result	Cumulative Result
DY&^H1FFF	02	XOR	1FFF	1FFD	1FFD0000
ZZ*^H2000	00	AND	2000	0000	1FFD0000
DX&^H1FFF	00	XOR	1FFF	0000	1FFD1FFF

The successive operations give us our desired result, a vector command that draws a beam line from the current position to a new position along the vector -1,-3:


Y Value				Intensity/X Value			
1	F	F	D	1	F	F	F
000Y YYY YYY YYY IIIIX XXXX XXXX XXXX							
0001 1111 1111 1100 0001 1111 1111 1111							

There are twelve other finds of interest given below. Unlike the set in Figure 2 above, none of these resemble early iterations of the player's 'claw'. All our finds appear in an area of the source code described as ENEMY PICTURES and are more likely to be just that: a set of alien enemies for a very early iteration of Tempest that according to programmer David Theurer was a 'First Person Space Invaders'. ⁴


⁴<https://arcadeblogger.com/2018/01/19/atari-tempest-dave-theurers-masterpiece/>



ENER11 to ENER14 in ALVROM.MAC.



ENER41 to ENER44 in ALVROM.MAC.



SAU to SA4 in ALVROM.MAC.

space game reenactment society

It was called *ALIEN SPACE GAME*. It didn't test well. It was meant to be a kind of three-dimensional space invaders knock-off, if we can imagine what that might look like. And we definitely shouldn't.

According to Theuerer: "The initial gameplay was to be simply a 'First Person Space Invaders'. I got it up and running, but it wasn't much fun. Too much like Space Invaders (surprise surprise). So after the review they said do you want to kill it? Or have you got any other ideas?"


He did have another idea, it was *Tempest*, so he made that instead. But as we saw in '[things hidden](#)' there is just enough wreckage of that initial concept in the source code for us to imagine what that first attempt at a game might have looked like.

Scattered here and there we find traces of the crime.

```
SPACG=0      ;SUPPRESS SPACE GAME CODE
```


Well I'm sorry but the SPACE GAME refuses to be fully suppressed. We have already seen some of the intriguing enemies defined in ALVROM.MAC.

But there are other remnants to be found in there. We find the player's bullets:




DSHTBL, DS2TBL, DS3TBL and DS4TBL in ALVROM.MAC.

And the 'spears' fired by the player's enemies:



ESHTBL, ES2TBL, ES3TBL and ES4TBL in ALVROM.MAC.

There is also an 'asteroid', which appears to have been some kind of bomb to be deployed when necessary.




```
ASTTBL: .BYTE ENDAST-ASTTBL  
.BYTE -6,0,0  
.BYTE 3,5,4  
.BYTE 3,5,-4  
.BYTE 0,-5,0
```

ASTTBL in ALVROM.MAC.


We find the definition of a primitive-looking 'fort', a simple pyramidal shape that perhaps served as an obstacle that the player could hide behind to avoid enemy 'spears' and which was perhaps eventually degraded by enemy attack over time. To draw the forts a macro called FORTMAC was used to draw the same pyramidal shape with the initial dimension specified as a parameter FS.

```
.MACRO FORTMAC .FS
  VCTR .FS,-<.FS/3>,0
  VCTR -.FS,.FS,6
  VCTR -.FS,-.FS,6
  VCTR 2*.FS,0,6
  VCTR -.FS,<.FS/3>,0
  RTSL
.ENDM
```


It appears this degradation was implemented by gradually reducing the size of the fort
- as you can see by the values on the axis on the images below.




FORTMAC 15 in ALVROM.MAC.



FORTMAC 10 in ALVROM.MAC.




FORTMAC OE in ALVROM.MAC.



FORTMAC OC in ALVROM.MAC.

Finally we find the remains of the player's ship. Unfortunately it is something of a disappointment - so clearly little more than a placeholder.



GUNPIC in ALVROM.MAC.

```
GUNPIC: VCTR -8,0,6  
VCTR -8,-10,6  
VCTR -10,0,6  
VCTR -8,-18,6  
VCTR 50,0,0  
VCTR -8,18,6  
VCTR -10,0,6  
VCTR -8,10,6  
VCTR -8,0,6  
RTSL
```

There are other clues. We find the remnants of what must have been the game's playfield, a grid-like structure. which despite its name contain 10 vertical lines rather than just 5.

```
.SBTTL DRAW TABLES:GRID LINES  
VRT4DRW:  
VRT5DRW: ;DRAW 5 VERT LINES  
TLABS 0  
SBRITE 040  
TVCTR 1  
SBRITE 0  
TVCTR 3  
SBRITE 040  
TVCTR 2  
SBRITE 0  
TVCTR 4  
SBRITE 040  
TVCTR 5  
SBRITE 0  
TVCTR 7  
SBRITE 040  
TVCTR 6  
SBRITE 0  
TVCTR 8  
SBRITE 040  
TVCTR 9  
OBJEND
```

If we put all these elements together we can make a wild and unfounded speculation as to how *ALIEN SPACE GAME* might have looked.



I probably shouldn't have even tried.

This is my deeply wonky attempt at visualising such a thing. Note that I've assumed that 'first person' space invaders must involve some kind of version of the classic space invaders playfield but with a three-dimensional depth of field: the baddies come at us slowly from the back of the screen, getting larger as they approach. Our player sits at the near edge of the field shooting along the grid that the enemies advance across. The 'forts' sit in front of the player and presumably can shield him from enemy attack. Meanwhile the asteroid is hanging out there in space in search of a purpose. Aren't we all, darling.

sexy yes

The sound effects in Tempest 2000 are stored in the game as Pulse Code Modulation (PCM) data. This is the most common format for storing raw audio and flavors of it are used in everything from CDs to telephony. PCM data is as simple as a stream of numbers. Each of the numbers is between a lower and upper limit; for example between -128 and 128. If you plot enough of these numbers along a graph you get what looks like a wave. This is pretty much the sound wave, the analog version of our digital values. If you put these values through the right piece of hardware, such as a Digital Analog Converter (DAC) you can get a sound out the other side.




Figure .1: A sample of 80 bytes plotted as a wave graph.

The PCM data in Tempest 2000's sound samples is nothing more than this series of bytes, each one representing a value between -128 and 128, which can be converted

into sound. A single series of bytes will give us one 'channel' of sound, i.e. a mono sound sample. What we actually have are stereo sound samples so each sample contains two channels. The simplest way of incorporating two separate channels in the sample (one for the left speaker and one for the right) is to interleave them, in alternating bytes. We can see that this is the way the Tempest samples are encoded if we split out an arbitrary sample into separate streams of alternate bytes and graph them.




Figure .2: Plotted sound waves for the left and right channels respectively.

Since the sounds we want to play are just blobs of data, we need a way of knowing where these blobs are, and perhaps some information on what we have to do with them. This is provided by `samtab`, a block of data at address \$9ac800 that tells us everything we need to know:

samtab	EQU \$9ac800
--------	--------------

Just by itself, the `samtab` table reveals some interesting potential for ways that we can use the raw PCM bytes. Here it is in its raw form. With the bytes in `samtab` exactly as they are.

Name	Prio-	Priority	Period	Start	Length	Repeat Start	Repeat Length		
<hr/>									
'Engine Noise 1	',	\$0001,	\$01ac,	\$009acd00,	\$0011a0,	\$00,	\$009acd02,	\$00119e,	\$00
'Player Shot Normal 2'	',	\$0002,	\$01ac,	\$009adea4,	\$0008e8,	\$00,	\$009adea4,	\$000000,	\$00
'Engine Noise	',	\$0003,	\$01ac,	\$009ae290,	\$003378,	\$00,	\$009aee9e,	\$002594,	\$00
'Player Death	',	\$0004,	\$00d6,	\$000000000,	\$00549a,	\$00,	\$00000000,	\$000000,	\$00
'Player Death 2	',	\$0005,	\$01ac,	\$000000000,	\$002458,	\$00,	\$00000000,	\$000000,	\$00
'Player Shot Normal	',	\$0006,	\$01ac,	\$009b160c,	\$0007a4,	\$00,	\$009b160c,	\$000000,	\$00
'Player Jump	',	\$0007,	\$01ac,	\$009b1db4,	\$0018de,	\$00,	\$009b1db4,	\$000000,	\$00
'Crackle	',	\$0008,	\$00d6,	\$009b3696,	\$004594,	\$00,	\$009b3696,	\$000000,	\$00
'Cleared Level	',	\$0009,	\$01ac,	\$009b7c2e,	\$0037a2,	\$00,	\$009b7c2e,	\$000000,	\$00
'Warp	',	\$000a,	\$0238,	\$009bb3d4,	\$006ec8,	\$00,	\$009bb3d4,	\$000000,	\$00
'Large Explosion	',	\$000b,	\$01ac,	\$009c22a0,	\$0050c2,	\$00,	\$009c22a0,	\$000000,	\$00
'Powered Up Shot	',	\$000c,	\$01ac,	\$009c7366,	\$001976,	\$00,	\$009c7366,	\$000000,	\$00
'Get Power Up	',	\$000d,	\$01ac,	\$009c8ce0,	\$001aea,	\$00,	\$009c8ce0,	\$000000,	\$00
'Tink For Spike	',	\$000e,	\$00fe,	\$009ca7ce,	\$00040e,	\$00,	\$009ca7ce,	\$000000,	\$00
'NME At Top Of Web	',	\$000f,	\$01ac,	\$009cabef,	\$00001e,	\$00,	\$009cabef,	\$000000,	\$00
'Pulse For Pulsar	',	\$0010,	\$0358,	\$009cac02,	\$0019fe,	\$00,	\$009cac02,	\$000000,	\$00
'Normal Explosion	',	\$0011,	\$00d6,	\$009cc604,	\$002ab6,	\$00,	\$009cc604,	\$000000,	\$00


```
'Extra Explosion      ', $0012, $0358, $009cf0be, $0018ca, $00, $009cf0be, $000000, $00
'Static or Pulsar   ', $0013, $011c, $009d098c, $003fe4, $00, $009d098c, $000000, $00
'Pulsar Pulse       ', $0014, $0358, $009d4974, $000f0c, $00, $009d4974, $000000, $00
'Off Shielded NME  ', $0015, $00aa, $009d5884, $0027ca, $00, $009d5884, $000000, $00
'Excellent          ', $0016, $0200, $009d8052, $005976, $00, $009d8052, $000000, $00
'Superzapper Recharge', $0016, $0200, $009dd9cc, $00a958, $00, $009dd9cc, $000000, $00
'yes                ', $0018, $0200, $009e8328, $005a6c, $00, $009e832a, $005a6a, $00
'oneup              ', $0019, $0200, $009edd98, $0043ae, $00, $009edd98, $000000, $00
'screeeam            ', $001a, $0200, $009f214a, $004568, $00, $009f214a, $000000, $00
'sexy yes 1          ', $001b, $0200, $009f66b6, $002c54, $00, $009f66b6, $000000, $00
'sexy yes 2          ', $001c, $0200, $009f9362, $003236, $00, $009f9362, $000000, $00
'tink                ', $001e, $0200, $009fc59c, $0005ce, $00, $009fc59c, $000000, $00
'zero                ', $001f, $0200, $009fc6e, $000008, $00, $009fc6e, $000000, $00
'dummy               ', $0020, $0200, $009fc7a, $00a1d8, $00, $009fc7a, $000000, $00
```

Listing .1: The contents of samtab at \$9AC800.

And here is a 'normalized' version of the table, with appropriate values converted to decimal for convenience.

Name	Priority	Period	Start	Length	Repeat Start	Repeat Length
'Engine Noise 1'	1	428	\$009acd00	4512	\$009acd02	4510
'Player Shot Normal 2'	2	428	\$009adea4	2280	\$009adea4	0
'Engine Noise'	3	428	\$009ae290	13176	\$009aee9e	9620
'Player Death'	4	214	\$00000000	21658	\$00000000	0
'Player Death 2'	5	428	\$00000000	9304	\$00000000	0
'Player Shot Normal'	6	428	\$009b160c	1956	\$009b160c	0
'Player Jump'	7	428	\$009b1db4	6366	\$009b1db4	0
'Crackle'	8	214	\$009b3696	17812	\$009b3696	0
'Cleared Level'	9	428	\$009b7c2e	14242	\$009b7c2e	0
'Warp'	10	568	\$009bb3d4	28360	\$009bb3d4	0
'Large Explosion'	11	428	\$009c22a0	20674	\$009c22a0	0
'Powered Up Shot'	12	428	\$009c7366	6518	\$009c7366	0
'Get Power Up'	13	428	\$009c8ce0	6890	\$009c8ce0	0
'Tink For Spike'	14	254	\$009ca7ce	1038	\$009ca7ce	0
'NME At Top Of Web'	15	428	\$009cabef	30	\$009cabef	0
'Pulse For Pulsar'	16	856	\$009cac02	6654	\$009cac02	0
'Normal Explosion'	17	214	\$009cc604	10934	\$009cc604	0
'Extra Explosion'	18	856	\$009cf0be	6346	\$009cf0be	0
'Static or Pulsar'	19	284	\$009d098c	16356	\$009d098c	0
'Pulsar Pulse'	20	856	\$009d4974	3852	\$009d4974	0
'Off Shielded NME'	21	170	\$009d5884	10186	\$009d5884	0
'Excellent'	22	512	\$009d8052	22902	\$009d8052	0
'Superzapper Recharge'	22	512	\$009dd9cc	43352	\$009dd9cc	0
'yes'	24	512	\$009e8328	23148	\$009e832a	23146
'oneup'	25	512	\$009edd98	17326	\$009edd98	0
'screeeam'	26	512	\$009f214a	17768	\$009f214a	0
'sexy yes 1'	27	512	\$009f66b6	11348	\$009f66b6	0
'sexy yes 2'	28	512	\$009f9362	12854	\$009f9362	0
'tink'	30	512	\$009fc59c	1486	\$009fc59c	0
'zero'	31	512	\$009fc6e	8	\$009fc6e	0
'dummy'	32	512	\$009fc7a	41432	\$009fc7a	0

Turning the raw bytes of samtab into something we can read.



All 28 sound samples and their spectrograms. If your reader supports it you can play the samples.

Of course the best way of seeing how this works is to follow one through in practice. Here is the little routine that shouts 'Excellent!' when you navigate the options menu. We can see that we do a little set up such as selecting the sample from the table above ('Excellent' is at position 21 (\$16 in hex)), setting its pitch and then calling another routine called `fox` that inches nearer the actual work of playing the sample:

```
; ****
; Say 'Excellent', e.g. when the user naviates to an option.
; ****
sayex:
move #21,sfx           ; Select effect $16 (21) from samtab, 'Excellent'.
move #101,sfx_pri      ; Set the priority.
move.l #$160,sfx_pitch ; Set the pitch.
jsr fox                ; Play the effect (see below).
move #101,sfx_pri      ; Do it again.
move.l #$162,sfx_pitch ; This time with a different pitch.
jmp fox                ; Play it.
```

`fox` (why waste an opportunity to pun on something in the animal kingdom) sets up our data for playing the sample. Its main preoccupation is turning our index into the `samtab` table to a pointer to the data there.

```
; ****
; Play a selected sound sample.
; ****
fox:
movem.l d0-d3/a0,-(a7) ; Save d0,d1,d2,d3,a0 to the a7 register.
                        ; Because we're going to clobber them here.
move sfx,d0             ; Store sfx in d0, this is '$16' for 'Excellent'.

; Use '21' to point to the address of sample data, i.e. directly after
; the 'Excellent' in this entry in samtab:
; 'Excellent', $0016, $0200, $009d8052, $005976, $00,

; Since each entry in samtab is 40 bytes long, multiplying our index by
; 40 will bring us to the start of the 'Excellent' entry. The steps below
; are a fast way of doing this multiplication by 40.
lsl #3,d0               ; Turn $16(21) into $a8 by left-shifting.
move d0,d1               ; Store in d1.
lsl #2,d0               ; Turn $a8 into $2a0 by left-shifting.
add d1,d0               ; d0 is now $348 (840), i.e. 21 * 40.

; Now that we have an offset for our sample information in samtab, we'll
; add 20 to it (so that it's after the sample name) and store it in a0.
lea samtab,a0            ; Store samtab address in a0.
lea 20(a0,d0.w),a0        ; Add 840 + 20 to it.

; Now we can set up the last few items our sound synth needs.
move sfx_pri,d1
move sfx_vol,d2
move.l sfx_pitch,d3
```

```

; Invoke the sound synth to play our sample.
jsr PLAYFX2

; Clean up.
move d0,handl
clr sfx_pri
clr sfx_vol
clr.l sfx_pitch

movem.l (a7)+,d0-d3/a0 ; Restore the old unclobbered values we saved off
                        ; at the start.

rts

```

To recap, at this point we have set up our registers as follows:

Register	Value	Description
a0	\$9acb5c	Address of the 'Excellent' metadata in samtab.
d1	101	sfx_pri.
d2	\$ff	sfx_vol: the volume to play at.
d3	\$162	sfx_pitch: the pitch to play at.

These are the garnish for a choice piece of mystery meat, PLAYFX2. PLAYFX2 is a library routine provided to Atari by a crowd called 'Imagitec Designs'. It is not something we have the source code for, and nor did Minter when he was writing Tempest 2000. Instead, he was given a binary blob called syn6.o and a list of routine names to use when he wanted to play one of his samples. This list is given at the start of the main yak.s source file:

INIT_SOUND	EQU	\$4040	;jump table for the SFX/Tunes module
NT_VBL	EQU	\$4046	
PT_MOD_INIT	EQU	\$404c	
START_MOD	EQU	\$4052	
STOP_MOD	EQU	\$4058	
PLAYFX2	EQU	\$405e	
CHANGE_VOLUME	EQU	\$4064	
SET_VOLUME	EQU	\$406a	
NOFADE	EQU	\$4070	
FADEUP	EQU	\$4076	
FADEDOWN	EQU	\$407c	
ENABLE_FX	EQU	\$4082	
DISABLE_FX	EQU	\$4088	
CHANGEFX	EQU	\$409a	;new in syn6
HALT_DSP	EQU	\$408e	
RESUME_DSP	EQU	\$4094	
intmask	EQU	\$40a0	

These are all entry points in the game binary for each routine. So for PLAYFX2 the entry

point is \$405e. In the source code this routine exists as raw machine code only. Along with all the other Imagitec sound routines it lives in a file called `moomoo.dat`:

```
.include "moomoo.dat"
```

This is a slightly modified version of the `syn6.o` sound synthesizer binary provided by Imagitec Designs. With modern tooling, it is possible for us to recreate `moomoo.dat` by partially disassembling it, then reassembling and relinking it. This will more or less re-enact the build steps Minter would have followed to generate it.

The initial bytes of `moomoo.dat` can be disassembled as follows into a file we'll call `moomoo_header.s`.

```
; Header for moomoo.s
;.org $4040
.include "jaguar.inc"

init_sound      EQU $000041ba
nt_vbl          EQU $0000425e
pt_mod_init     EQU $000041d6
start_mod       EQU $000045e2
stop_mod        EQU $00004606
playfx2         EQU $00004370
change_volume   EQU $000044ea
set_volume      EQU $00004546
nofade          EQU $00004598
fadeup          EQU $000045b2
faidedown       EQU $000045ca
enable_fx       EQU $0000464e
disable_fx      EQU $00004660
resume_dsp      EQU $00004470

INIT_SOUND:     jmp    (init_sound).l
NT_VBL:         jmp    (nt_vbl).l
PT_MOD_INIT:    jmp    (pt_mod_init).l
START_MOD:      jmp    (start_mod).l
STOP_MOD:       jmp    (stop_mod).l
PLAYFX2:        jmp    (playfx2).l
CHANGE_VOLUME:  jmp    (change_volume).l
SET_VOLUME:     jmp    (set_volume).l
NOFADE:         jmp    (nofade).l
FADEUP:         jmp    (fadeup).l
FAIDEDOWN:      jmp    (faidedown).l
ENABLE_FX:       jmp    (enable_fx).l
DISABLE_FX:     jmp    (disable_fx).l
CHANGEFX:       jmp    (disable_fx).l
HALT_DSP:        jmp    (disable_fx).l
RESUME_DSP:     jmp    (resume_dsp).l

intmask:        dc.b 0
; Address 0x40A2
```

```
return_early:  
    rts  
  
; Address 0x40A4  
update_interrupt:  
    move.w d0,-(sp)  
    bset #3,(intmask).l  
    clr.w d0  
    move.b (intmask).l,d0  
    move.w d0,(INT1).l ; RW CPU Interrupt Control Register  
    move.w (sp)+,d0  
    rts
```

Listing .2: The contents of `moomoo_header.s`.

It turns out that the address we're using to invoke PLAYFX2 simply jumps to another adress in the binary, one that contains the actual playfx2 routine:

```
PLAYFX2:    jmp     (playfx2).l
```

So to recreate the steps used to generate `moomoo.dat` we first assemble this file:

```
rmac -fa -i moomoo_header.s -o moomoo_header.o
```

Then we link `moomoo_header.o` with the `syn6.o` file provided by Imagitec:

```
rln -n -z -u -v -a 4040 xd xd -e moomoo_header.o syn6.o -o moomoo.dat
```

Voila, we have our `moomoo.dat`. If you'd like to follow these steps yourself, [you can use this Python notebook](#).

story of a beep

Whenever the cursor is moved left or right, *Tempest* emits a beep. This is the story of that beep. To tell the story we will start at the end rather than the beginning.

To magic up this beep, we have to emit a series of commands to the Atari cabinet's sound chip, known as the POKEY. These commands are as follows:

Register	Description	Value	Bits	Meaning
AUDC1	Channel 1 Control	A2	10110010	Rectangular curve, volume 2
AUDF1	Channel 1 Frequency	0F	00001111	High pitched sound

Then after a short interval, we follow up with the commands to stop the sound:

Register	Description	Value	Bits	Meaning
AUDC1	Channel 1 Control	00	00000000	Stop
AUDF1	Channel 1 Frequency	00	00000000	No sound

It's as simple as that. But since we will want to generate more complex sounds later, we have to develop some machinery that's capable of emitting more than just a single beep. So it will be instructive, with this minimally simple example, to see how this machinery generates the above commands, which consist of little more than telling the sound chip to play a beep and then stop. We find our sound definitions and the machinery underlying them in `ALSOUN.MAC`.

So below is the data driving our 'cursor crossing a line' beep. There are two sequences of bytes. The first defines the values and the timing for writing those values to the AUDF1 register. Writing something to AUDF1 is like playing a note. The value we write determines the note that will be played.

The second sequence defines the way that note is played. We can manipulate the performance of the note by writing to the AUDC1 register. Depending on the value we write we can alter the volume and the distortion to the note we selected.

```
L05F: .BYTE 0F,4,0,1 ; CURSOR CROSSED A LINE
      .BYTE 0,0
L05A: .BYTE 0A2,4,40,1
      .BYTE 0,0
```


It will be easier to understand the above sequence definition if we comment the purpose of each byte in the sequence:

```
L05F: .BYTE 0F ; VALUE TO WRITE TO AUDF1
      .BYTE 4 ; BEATS TO WAIT BEFORE NEXT CHANGE
      .BYTE 0 ; AMOUNT TO CHANGE BY
      .BYTE 1 ; NUMBER OF TIMES TO CHANGE (1 MEANS 0)
      .BYTE 0 ; RESTART POSITION (0 MEANS NO RESTART)
      .BYTE 0 ; STOP

L05A: .BYTE 0A2 ; VALUE TO WRITE TO AUDC1
      .BYTE 4 ; BEATS TO WAIT BEFORE NEXT CHANGE
      .BYTE 40 ; AMOUNT TO CHANGE BY
      .BYTE 1 ; NUMBER OF TIMES TO CHANGE (1 MEANS 0)
      .BYTE 0 ; RESTART POSITION (0 MEANS NO RESTART)
      .BYTE 0 ; STOP
```

Close inspection of the above example shows that the commands we set out on the previous page are the result of following the instructions of each of these byte sequences. For the L05F sequence, which controls what we write to the AUDF1 register, we write the following commands to turn on the note, wait 4 frames, and then turn it off again:

Register	Description	Value	Bits	Meaning
AUDF1	Channel 1 Frequency	0F	00001111	High pitched sound
AUDF1	Channel 1 Frequency	00	00000000	No sound



All 28 sound samples and their spectrograms. If your reader supports it you can play the samples.

Simultaneously the L05A sequence instructs us to set the volume and the sound type by writing to the AUDC1 register and then, after 4 frames, silence it again:

Register	Description	Value	Bits	Meaning
AUDC1	Channel 1 Control	A2	101,10010	Rectangular curve, volume 2
AUDC1	Channel 1 Control	00	00000000	Stop

We can look at a slightly more involved example to get an idea of the manipulation that this scheme affords us. Here are the two sequences that define an 'Enemy Shot' (ESLSON):

```
; ENEMY SHOT
ES8F: .BYTE 0      ; VALUE TO WRITE TO AUDF1
      .BYTE 3      ; BEATS TO WAIT BEFORE NEXT CHANGE
      .BYTE 2      ; AMOUNT TO CHANGE BY
      .BYTE 9      ; NUMBER OF TIMES TO CHANGE (1 MEANS 0)
      .BYTE 0      ; RESTART POSITION (0 MEANS NO RESTART)
      .BYTE 0      ; STOP

ES8A:  .BYTE 8      ; VALUE TO WRITE TO AUDC1
      .BYTE 3      ; BEATS TO WAIT BEFORE NEXT CHANGE
      .BYTE OFF    ; AMOUNT TO CHANGE BY (OFF means -1)
```

story of a beep

```
.BYTE 9 ; NUMBER OF TIMES TO CHANGE (1 MEANS 0)
.BYTE 0 ; RESTART POSITION (0 MEANS NO RESTART)
.BYTE 0 ; STOP
```

Notice that the commands are to be run simultaneously: every 3 frames. You can also notice that we are to run each command 8 times in total (9 - 1). We also note that OFF is another way of saying -1. So equipped with these provisions we can execute the commands as follows:

Register	Value	Meaning	Register	Value	Meaning
AUDC1	08	Volume 8	AUDF1	00	No Sound
AUDC1	07	Volume 7	AUDF1	02	Very Very Low Pitch
AUDC1	06	Volume 6	AUDF1	04	Very Low Pitch
AUDC1	05	Volume 5	AUDF1	06	Still Low Pitch
AUDC1	04	Volume 4	AUDF1	08	Low Pitch
AUDC1	03	Volume 3	AUDF1	0A	Getting Higher
AUDC1	02	Volume 2	AUDF1	0C	Higher Still
AUDC1	01	Volume 1	AUDF1	0F	Higher again

unused explosions

The *Tempest* source in `ALSOUN.MAC` contains a couple of unused sound definitions. The comment next to them suggests that they were originally intended as explosion effects.


```
;  
;EXPLOSION SOUND  
;  
T51F:    .BYTE 0C0,8,4,10  
        .BYTE 0,0  
T51A:    .BYTE 0A6,20,0F8,4  
        .BYTE 0,0  
T52F:    .BYTE 40,8,4,10  
        .BYTE 0,0  
T52A:    .BYTE 0A6,20,0FE,4  
        .BYTE 0,0
```

When you listen to them they do sound like that alright. Give them a try below.

unused explosions



T51F



T52F

The unused explosion sounds in `ALSOUN.MAC`.

play us a tune pal

Music in *Tempest 2000* comes in the form of `.mod` files. The mechanics of processing these files are hidden from us in a blob of bytes provided by Imagitec Ltd and included in the game binary. This blob of bytes is a sound synthesiser that transforms the contents of a `.mod` file into a tune that you can hear with your ears. What we can do instead is dig into the contents and structure of a `.mod` file to understand how it stores a tune. The obvious candidate for this research is the music played over the title screen.

I'm not sure if this music ever had a proper title, however the source code knows it fondly as `tune13.mod`. This file, along with all the other tunes included in the game, are stored in a reference table called `modbase`. This allows us to pick a tune for playing by referencing it using an index into the table.

```
modbase      EQU  $8d6800
modtable:
.dcl modbase + $100 ; tune13.mod
.dcl modbase + $18d2c ; tune7.mod
.dcl modbase + $37a04 ; tune1.mod
.dcl modbase + $5bd7c ; tune3.mod
.dcl modbase + $80b22 ; rave4.mod
.dcl modbase + $99d72 ; tune5.mod
.dcl modbase + $b0a50 ; tune12.mod
.dcl 0
```

In the little routine that plays a selected tune we can see this happening in practice. The index value for the selected tune is passed in the `d0` register and used as an index into `modbase` to retrieve the address in memory of the tune we've picked to play.


```


playtune:
    jsr STOP_MOD           ; Stop any current tune.
    lsl #2,d0               ; Multiply index (d0) by 4
    lea modbase,a0          ; Point a0 at modbase.
    move.l 0(a0,d0.w),a0    ; Get tune base using index.
    jsr PT_MOD_INIT         ; Point the MOD player at a0.
    move.b vols,d0           ; Store the volume in d0.
    and.l #$ff,d0            ; Keep it between 0 and 255
    clr d1                  ; Clear d1
    jsr SET_VOLUME           ; Actually set the volume.
    move.l d0,vset            ; Store the selected volume.
    jsr NOFADE                ; Turn off 'fading'.
    jmp START_MOD             ; Start playing the tune.

```


So to play the title screen tune, which is the first entry in `modtable`, we point the sound synthesiser at the region in memory containing `tune13.mod` by calling its black-box routine `PT_MOD_INIT` and then once we've set the volume, instruct it to start playing the darn thing (`jmp START_MOD`).

To understand the contents of `tune13.mod` and how it gives us a tune, it is best to start with the most basic building blocks it contains. These are 11 raw sound samples of Pulse Code Modulation (PCM) data like the ones we discussed in '[sexy yes](#)'. They are stored at the very end of the file, one after another from byte 9276 to byte 101416. This means they constitute the bulk of the `tune13.mod` tune file - which is not surprising since they are the part of the file that carries all the sounds. If you want to hear what each sample sounds like, your reader might support clicking on each sample to play it.






Sample 11



Sample 12



Sample 13

All 11 sound samples and their spectrograms. If your reader supports it you can play the samples.

If you try playing the files you'll notice that the majority of them are single notes on an instrument. This gives us a clue as to what the rest of the `tune13.mod` file might contain: instructions on how to play these notes in the correct order. More than this, the instructions might even tell us what pitch to play each note at, for example, whether a C sharp or a B flat.

We find these instructions at byte 1084 in the file and they occupy all the space in the file up to the beginning of the samples at byte 9276. The instructions are broken up into chunks called 'Patterns'. Each 'Pattern' describes a passage of music, using the samples to play notes across four separate audio channels at once. Another way of thinking of a 'Pattern' is like a piece of sheet music that describes the notes to play and the instrument to play them for four different players. In this case our instruments are the 'samples'. To give you an idea of what I mean here is the first note for all four players (which we call channels) given in the first 16 bytes of the data for the first of the seven patterns in `tune13.mod`.

	Bytes	Sample	Note	Effect
Channel 1	007f1000	1	A3	
Channel 2	01944f09	4	C#2	Set Speed 09
Channel 3	011d5000	5	G2	
Channel 4	00000000			

Decoding the first 16 bytes (4 per channel) in Pattern 000.

One the following page we see the complete decoded set of instructions contained in Pattern 000. As you look through it you can get a sense of the note being played on each channel and the instrument used, along with any special instruction (effect) specified.

tempest 2000

Channel 1			Channel 2			Channel 3			Channel 4		
Sample	Note	Effect	Sample	Note	Effect	Sample	Note	Effect	Sample	Note	Effect
1	A3	4	C#2	Set Speed	09	5	G2				
8	A#3					5	G2	Set Volume	15		
6	A3					5	G2				
6	A3					5	G2	Set Volume	25		
8	A#3					5	G2	Set Volume	15		
8	A#3					5	G2				
6	A3					5	G2				
1	A3					5	G2	Set Volume	25		
2	D3										
8	B3					5	G2	Set Volume	15	G2	
8	B3					5	G2		10	D3	
6	A3					5	G2	Set Volume	25	10	G2
0									10	D3	
8	B3					5	G2	Set Volume	15	10	G2
6	A3					5	G2		10	D#3	
1	A3					5	G2	Set Volume	25	10	D3
1	A3										
8	B3					5	G2	Set Volume	15		
6	A3					5	G2				
6	A3					5	G2	Set Volume	25		
8	A#3										
8	A#3					5	G2	Set Volume	15		
6	A3					5	G2				
1	A3					5	G2	Set Volume	25		
2	D3										
0						5	G2	Set Volume	15		
6	A3					5	G2			F2	
6	A3					5	G2	Set Volume	25		
8	A#3										
8	A#3	4	D2			5	F2	Set Volume	15	F2	
6	A3					5	F2				
1	A3					5	F2	Set Volume	25		
1	A3	4	C#2							F2	
8	A#3					5	G2	Set Volume	15		G2
6	A3					5	G2				
6	A3					5	G2	Set Volume	25		
8	A#3										
8	A#3					5	G2	Set Volume	15		
6	A3					5	G2				
1	A3					5	G2	Set Volume	25		
2	D3										
0						5	G2	Set Volume	15		
6	A3					5	G2			D3	
6	A3					5	G2	Set Volume	25		G2
8	A#3										D3
8	A#3					5	G2	Set Volume	15		G2
6	A3					5	G2				D#3
1	A3					5	G2	Set Volume	25		B2
1	A3										
8	A#3					5	G2	Set Volume	15		
6	A3					5	G2				
6	A3					5	G2	Set Volume	25		
8	A#3										
8	A#3					5	G2	Set Volume	15		
6	A3					5	G2				
1	A3					5	G2	Set Volume	25		
2	D3										
0						5	G2	Set Volume	15		
6	A3					5	G2				
6	A3					5	G2	Set Volume	25		
8	A#3										
1	A3					5	G2	Set Volume	15		
6	A3					5	G2				
1	A3					5	G#2	Set Volume	25		

Decoding the full instructions for Pattern 000.

What if we attempted to turn the notes in Pattern 0 into some sheet music? It's just a question of representing each channel as a stave and transcribing the notes from the pattern. We can't display the sample that's being used to play the note, but we do get a sense of the music that's being played.

Derived from tune13.mod Tempest 2000 Title Theme (Howe,Lindsay,Saville)
Imagitec Design

Pattern 0

The sheet music consists of four staves, each representing a different channel. The first staff (Channel 1) has a treble clef, the second (Channel 2) has a bass clef, the third (Channel 3) has a bass clef, and the fourth (Channel 4) has a treble clef. The music is in common time. The notes are represented by vertical stems with horizontal dashes indicating pitch and duration. The overall pattern is a repeating sequence of notes across all four channels.

Pattern 0 rendered as sheet music.

With a single chunk of music like this we now have the makings of a tune, built up from using our samples in a series of machine-readable sheet music instructions in a pattern.

Our next step is to come up with as many of these patterns as we need to form a complete tune. In theory we could mix and match the patterns in any sequence we like and play them one after the other to form the finished article. In the case of `tune13.mod`, the structure is quite simple. At byte 952 in our `tune13.mod` file we list the order in which the patterns are to be played. This also lets us know how many patterns in total the file contains. The full list is: 0, 1, 2, 3, 4, 3, 6, 7, 5. With this information we can illustrate the title tune in its entirety by piecing together the transcriptions of each of the patterns.

tempest 2000

Derived from tune13.mod

Tempest 2000 Title Theme

(Howe,Lindsay,Siville)
Imagitec Design

Pattern 0

Pattern 0

Derived from tune13.mod

Tempest 2000 Title Theme

(Howe,Lindsay,Siville)
Imagitec Design

Pattern 2

Pattern 2

Derived from tune13.mod

Tempest 2000 Title Theme

(Howe,Lindsay,Siville)
Imagitec Design

Pattern 4

Pattern 4

Derived from tune13.mod

Tempest 2000 Title Theme

(Howe,Lindsay,Siville)
Imagitec Design

Pattern 1

Pattern 1

Derived from tune13.mod

Tempest 2000 Title Theme

(Howe,Lindsay,Siville)
Imagitec Design

Pattern 3

Pattern 3

Derived from tune13.mod

Tempest 2000 Title Theme

(Howe,Lindsay,Siville)
Imagitec Design

Pattern 3

Pattern 3

Derived from tune13.mod Tempest 2000 Title Theme *(Howe,Lindsay,Swiller)
Imagitec Design*

Pattern 6

Derived from tune13.mod Tempest 2000 Title Theme *(Howe,Lindsay,Swiller)
Imagitec Design*

Pattern 7

Pattern 6

Derived from tune13.mod Tempest 2000 Title Theme *(Howe,Lindsay,Swiller)
Imagitec Design*

Pattern 5

Pattern 5

The patterns in the order that they're played, giving the complete tune.

ntsc pal

In the 1980s and 1990s games were played on cathode ray tube televisions. Like with everything, there were competing standards. And like with everything, Americans did things one way (NTSC) and Europe did things another (PAL). The rest of the world got to choose which horse to back.

There are many technical differences between NTSC and PAL, but only one that Jeff Minter really needed to worry about in *Tempest 2000*: how many lines of pixels each standard allowed. NTSC allowed a measly 241, while PAL televisions came with a princely 287.

Faced with releasing a video game that could be played on televisions in both America and Europe, the developer needed to figure out which of the two it was playing on and adjust the dimensions of the screen it was writing to accordingly.

When launched one of the earliest tasks is to do this figuring out. Fortunately there was a magic number available from the Jaguar hardware that would tell it what was what. This magic number lived at the hardware address register \$F14002. Here we can see the VideoIni routine take care of this business, querying the register, setting some dimensions based on the reading it gets, and most importantly for the rest of the game, setting a global variable called `pal` that can be used to perform any adhoc fiddling during gameplay to ensure a display consistent with the player's ageing television hardware.

```
VideoIni:  
    movem.l d0-d6,-(sp)          ; Store d0-d6 in the stack.  
    clr pal                      ; Clear the global pal variable
```

```

        move.w  BASE+$14002,d0      ; Check the JOY2(!) port.
        and.w   #$10,d0            ; Check if the magic bit is set..
        beq    ispal              ; .. if not, it's PAL so skip to ispal.

        ; Otherwise it's NTSC.
        move.w  #ntsc_hmid,d2      ; Set the Horizontal mid-point
        move.w  #ntsc_width,d0      ; Set the width.
        move.w  #ntsc_vmid,d6      ; Set the vertical mid-point.
        move.w  #ntsc_height,d4      ; Set the height.

        bra     doit                ; Actually apply the above values.

ispal:
        move.w  #pal_hmid,d2      ; Set the Horizontal mid-point
        move.w  #pal_width,d0      ; Set the width.
        move.w  #pal_vmid,d6      ; Set the vertical mid-point.
        move.w  #pal_height,d4      ; Set the height.
        move #1,pal                ; Set the global pal variable.

```

There are numerous places where this kind of fiddling will be necessary. Early on in the game's launch sequence we will query this pal setting to set various fix-up variables that can be used to centre text and images depending on the height of the screen:

```

; ****
; Set up the NTSC/PAL options.
; ****
spall:
        btst.b #6,sysflags          ; Are controllers enabled?
        beq  slopt
        move.l #o2s3,option2+16

slopt:
        clr  palside
        clr  paltop
        bclr.b #5,sysflags          ; Flag PAL as not enabled.
        tst  pal
        beq  notpal1
        ; Set screen adjustments for PAL
        move #6,palside
        move #10,paltop
        move #40,palfix1             ; Y centre fix for rectangles in PAL
        move #20,palfix2             ; Y centre fix for text in PAL
        move.l #$140000,palfix3             ; Y centre fix for PAL
        bset.b #5,sysflags          ; Set PAL as enabled.

notpal1: rts

```

Here's an example where we're going to display some text on the screen ("Press Option for Beastly Mode!") and need to figure out whether we're on a PAL or NTSC screen to decide where to place the text so that it's vertically centred.

```

draw_o:
        bsr  draw_o                  ; Call the common draw_o routine first.
        btst.b #2,sysflags          ; Is Beastly Mode enabled?

```

```

        beq rrrts          ; If so, return early.
        tst h2h            ; Are we playing a head-to-head game?
        bne rrrts          ; If so, return early.
        lea bstymsg,a0    ; "PRESS option FOR BEASTLY MODE!"
        lea cfont,a1       ; Load the small regular font to a1.
        move #180,d0       ; Set Y position of text.
        tst pal            ; Are we on a PAL screen?
        beq gnopal2        ; If not, skip.
        add palfix2,d0     ; If we are, adjust Y POS for PAL screens

gnopal2: jmp centext      ; Display horizontally centred text.
; Returns


```

The rest of the code is littered with such fix-ups. We even find an artefact of the PAL/NTSC schism when we look at the image data used in our '[Object Lists](#)'. The image data assumes a screen that is 287 pixels high, i.e. one that will display a full screen on PAL. When playing on NTSC this area of the image is not written to and instead we find a bar of pleasing junk pixels.

Raw Data	Parsed Data	Image at DATA: 0x134800
13,48,00,1D,6E,45,C1,60	TYPE: 0 LINK: 0xeb70 YPOS: 44 DATA: 0x134800 HEIGHT: 279	
00,00,80,06,01,80,CF,F8	XPOS: -7 REFLECT: 0 DEPTH: 4 RMW: 0 PITCH: 1 TRANS: 1 DWIDTH: 96 RELEASE: 0 IWIDTH: 96 FIRSTPIX: 0 INDEX: 0	

Figure .1: First Bit Mapped Object in the Object List

On PAL this field of colourful noise would be written to properly and the full game field would extend for the whole 287 pixels of screen height. The solution when playing on an NTSC television screen was simply to truncate the height and give the player just enough space to play the game while missing out on activity that extended below the direct field of play.




approaching logo process




Figure .1: The Tempest Title Rainbow.

The *TEMPEST* title sequence is a thing of simple beauty. It cascades towards us from afar in a rainbow of shifting colours. It is a glimpse into the future of video games, where the illusion of three dimensions will become commonplace. Despite its rawness, and even its crudity, it beguiles us into accepting Tempest's world of near infinite depth. A world in which things that are so far away they are almost invisible quickly come up on us from the deep.


To understand how it was achieved we must start with the bare building blocks. The routines that get the letters onto the screen. There are five individual letters in Tempest, and we have a routine for each.




```
T: VCTR 0,80,CB ;T
    VCTR -50,0,0
    VCTR 0A0,0,CB
```




```
E: VCTR -50,0,CB ;E
    VCTR -14,-40,CB
    VCTR 70,0,CB
    VCTR -70,0,0
    VCTR -14,-40,CB
    VCTR 84,0,CB
```



```
M: VCTR -20,0,CB ;M
    VCTR 30,80,CB
    VCTR 10,0,CB
    VCTR 20,-58,CB
    VCTR 20,58,CB
    VCTR 10,0,CB
    VCTR 30,-80,CB
    VCTR -20,0,CB
    RTSL
```




```
P: VCTR -10,0,CB ;P
    VCTR 0,80,CB
    VCTR 5C,0,CB
    VCTR 1A,-48,CB
    VCTR -76,0,CB
    RTSL
```



```
S: VCTR -10,-28,CB ;S
    VCTR 90,0,CB
    VCTR 0,38,CB
    VCTR -70,20,CB
    VCTR 10,28,CB
    VCTR 64,0,CB
    VCTR -0C,-20,CB
    RTSL
```

If you look closely you can see each letter is a routine in its own right. Each consists of the vector commands we encountered in '**cursors**' and '**wells**'. Each concludes with an RTSL statement that returns from the routine. This mean that in order to build the **TEMPEST** title image we can create a routine called **VORLIT** that writes it out in full.



```
VORLIT:
CNTR
VCTR -1B0,100,0
JSRL T
VCTR 60,0,0
JSRL E
VCTR 24,0,0
JSRL M
VCTR 34,0,0
JSRL P
VCTR 0F8,48,0
JSRL E
VCTR 16,28,0
JSRL S
VCTR 60,-60,0
JMPL T
```

Equipped with **VORLIT** as a shorthand for drawing the **TEMPEST** title logo onto the screen we can use two other Vector Drawing instructions to alter the scale of the logo and colour of the logo when we draw it. If we adjust the scale and the colour as we go along we can create the effect of a cascading series of differently coloured titles receding into the distance.

Fortunately there are vector commands that will allow us to do all of this. Each command is a two-byte Op Code. There's one for setting the color to be used when drawing, one for setting the scale to use when drawing (allowing us to make the title big or small), and one for calling our **VORLIT** routine to actually draw the title logo in our selected colour and scale.

OpCode	Description	OpCode	How We Use It.
New color/intensity	0x6000	000	will give the color and intensity.
New scale	0x7000	000	will give the scale to draw at.
Jump to subroutine	0xA000	000	can be populated with an address, e.g. VORLIT.

So we can imagine issuing a series of three vector commands such as the ones given below to draw our logo.

Op Code	Parsed Data	Written to Screen
7220	Set Scale: 220	
6800	New Color: White Intensity: 00	
AFA7	Jump to Routine: VORLIT	

If we repeat these commands, changing the color and reducing the scale as we go, we should expect to achieve the effect we're looking for. And this is exactly what we find in SCARNG. This routine builds the title rainbow by adding the three commands above to the vector list and it does this 19 times, adjusting the color and reducing the scale as it goes.

```
.SBTTL LOGO RAINBOW BUILDER
SCARNG:
; DRAW THE TEMPEST TITLE RAINBOW. THIS CONSISTS OF 19 ITERATIONS
; OF THE TITLE IMAGE IN EVER INCREASING DEPTH AND VARYING COLORS.
; A - CONTAINS THE LOW BYTE OF THE ADDRESS FOR VORLIT.
; X - CONTAINS THE HI BYTE OF THE ADDRESS FOR VORLIT.
STA PYL           ; STORE LOW BYTE OF VORLIT IN PYL
STX PXL           ; STORE HI BYTE OF VORLIT IN PYL
LDA NEARY         ; PUT THE NEAREST ALLOWED POSITION IN A
STA INDEX1        ; STORE IT IN INDEX1

; THIS IS THE LOOP THAT DRAWS THE TITLE RAINBOW, ONE TITLE
; AT A TIME, UNTIL WE REACH THE FURTHEST ALLOWED POSITION.
; WITH THE NEAREST POSITION STORED IN INDEX1 WE INCREMENT
; IT AT EACH ITERATION TO DRAW THE NEXT TITLE A LITTLE
; FURTHER AWAY UNTIL IT EQUALS FARY.
BEGIN             ; START OF THE LOOP

; FIRST WE SET THE SCALE OF THE TITLE IMAGE TO DRAW IN THIS
; ITERATION. THIS WILL DETERMINE HOW FAR AWAY IT SEEKS. TO DO
; THIS WE BASE IT ON THE DISTANCE VALUE IN INDEX1 AND CALL
; VGSCAL TO ADD A VECTOR COMMAND THAT SETS THE SCALE WE WILL
; DRAW THE TITLE AT. WE SET TWO SCALE TYPES, LINEAR AND BINARY.
; LINEAR SCALE FACTOR (0=FULL SIZE, FF=1/256 SIZE, FE=2/256, ETC.)
; BINARY SCALE FACTOR (0=FULL SIZE, 1=1/2 SIZE, ETC.)
LDA INDEX1        ; GET THE CURRENT DISTANCE VALUE.
ASL               ; SHIFT LEFT ONE BIT (I.E. MULTIPLY BY 2).
ASL               ; MULTIPLY BY 2 AGAIN.
AND I,7F          ; AND WITH 7F TO GET OUR SCALE VALUE.
TAY               ; SET RESULT AS OUR LINEAR SCALE VALUE.
LDA INDEX1        ; GET THE CURRENT DISTANCE VALUE.
LSR               ; DIVIDE BY 32 BY SHIFTING RIGHT FIVE TIMES.
LSR               ; TWO TIMES.
LSR               ; THREE TIMES.
LSR               ; FOUR TIMES.
```

```

LSR          ; FIVE TIMES.
JSR VGSCAL   ; ADD THE COMMAND SETTING BINARY AND LINEAR SCALE.

; NEXT WE SET THE COLOR WE WILL DRAW THE TITLE WITH. WE USE THE
; CURRENT DISTANCE TO CHOOSE THIS COLOR AND 'SET' IT BY CALLING
; VGSTAT. WE TURN THE DISTANCE INTO A COLOR BY SHIFTING IT RIGHT
; THREE BITS - THIS GIVES US A COLOUR VALUE THAT VGSTAT CAN USE.
LDA INDEX1    ; GET THE CURRENT DISTANCE VALUE.
CMP NEARY     ; IS IT EQUAL TO THE NEAREST POINT (I.E. THE TOP)?
IFEQ          ; IF SO..
LDA I,WHITE   ; .. PAINT IT WHITE.
ELSE          ; OTHERWISE..
LSR           ; DIVIDE BY 8 BY SHIFTING LEFT THREE TIMES.
LSR           ; TWO TIMES.
LSR           ; THREE TIMES.
NOP           ; A NO-OP.
AND I,7       ; MAKE THE RESULT BETWEEN 0 AND 7
CMP I,7       ; IS IT EQUAL TO 7 (I.E. BLACK)?
IFEQ          ; IF SO..
LDA I,RED     ; USE RED FOR BLACK
ENDIF         ;
ENDIF         ;
TAY           ; STORE THE COLOR IN Y
LDA I,68      ; SET THE COMMAND BITS FOR VGSTAT
JSR VGSTAT    ; SET COLOR BY CALLING VGSTAT







; NOW WE'RE READY TO PAINT THE TITLE. REMEMBER WE STORED THE ADDRESS
; TO VORLIT IN PYL AND PXL. SO VGJSRL WILL USE THAT TO DRAW THE VECTORS
; CONTAINING THE TEMPEST TITLE IN THE VORLIT ROUTINE.
LDA PYL        ; LOAD THE LOW BYTE OF VORLIT TO A.
LDX PXL        ; LOAD THE HI  BYTE OF VORLIT TO A.
JSR VGJSRL    ; DRAW THE TEMPEST TITLE.

; NOW GET READY TO DRAW THE NEXT ITERATION. WE INCREMENT INDEX1
; BY 2 AND DRAW ANOTHER TITLE FURTHER AWAY, UNTIL WE REACH THE FARDEST
; ALLOWED DISTANCE (FARY).
LDA INDEX1    ; GET THE CURRENT DISTANCE VALUE.
CLC           ; CLEAR THE CARRY SO ADDING WILL HAPPEN OK.
ADC I,2       ; ADD 2 TO THE CURRENT DISTANCE.
STA INDEX1    ; AND STORE IT IN INDEX1.
CMP FARY      ; IF IT IS EQUAL TO FARY, EXIT..
CSEND         ; OTHERWISE LOOP AND DRAW ANOTHER TITLE.

```




We can see how the commands mutate and slowly build up the rainbow piece by piece for all 19 iterations in the table that follows.

tempest

Vector Commands	Command Meaning	Result on Screen
7228	Set Scale: 228	
6801	New Color: Yellow Intensity: 00	
AFA7	Jump to Routine: VORLIT	
7230	Set Scale: 230	
6801	New Color: Yellow Intensity: 00	
AFA7	Jump to Routine: VORLIT	
7238	Set Scale: 238	
6801	New Color: Yellow Intensity: 00	
AFA7	Jump to Routine: VORLIT	
7240	Set Scale: 240	
6802	New Color: Magenta Intensity: 00	
AFA7	Jump to Routine: VORLIT	
7248	Set Scale: 248	
6802	New Color: Magenta Intensity: 00	
AFA7	Jump to Routine: VORLIT	
7250	Set Scale: 250	
6802	New Color: Magenta Intensity: 00	
AFA7	Jump to Routine: VORLIT	

Vector Commands	Command Meaning	Result on Screen
7258	Set Scale: 258	
6802	New Color: Magenta Intensity: 00	
AFA7	Jump to Routine: VORLIT	
7260	Set Scale: 260	
6803	New Color: Red Intensity: 00	
AFA7	Jump to Routine: VORLIT	
7268	Set Scale: 268	
6803	New Color: Red Intensity: 00	
AFA7	Jump to Routine: VORLIT	
7270	Set Scale: 270	
6803	New Color: Red Intensity: 00	
AFA7	Jump to Routine: VORLIT	
7278	Set Scale: 278	
6803	New Color: Red Intensity: 00	
AFA7	Jump to Routine: VORLIT	
7300	Set Scale: 300	
6804	New Color: Cyan Intensity: 00	
AFA7	Jump to Routine: VORLIT	

tempest

Vector Commands	Command Meaning	Result on Screen
7308	Set Scale: 308	
6804	New Color: Cyan Intensity: 00	
AFA7	Jump to Routine: VORLIT	
7310	Set Scale: 310	
6804	New Color: Cyan Intensity: 00	
AFA7	Jump to Routine: VORLIT	
7318	Set Scale: 318	
6804	New Color: Cyan Intensity: 00	
AFA7	Jump to Routine: VORLIT	
7320	Set Scale: 320	
6805	New Color: Green Intensity: 00	
AFA7	Jump to Routine: VORLIT	
7328	Set Scale: 328	
6805	New Color: Green Intensity: 00	
AFA7	Jump to Routine: VORLIT	
7330	Set Scale: 330	
6805	New Color: Green Intensity: 00	
AFA7	Jump to Routine: VORLIT	


We now have a sense of how to draw the rainbow, but in a way this is just one iteration of the effect. As we know from experience, the logo gets gradually larger approaching us from the black depths of the screen. The way this is achieved is by running the process we've described above over and over, while adjusting the values that define the nearest and farther point on the screen. These two values (NEARY and FARY) control the overall maximum and minimum scale that we draw the logo with in SCARNG. So by incrementing them in the routine LOGPRO below, then calling SCARNG to draw the logo with these values as their limits, we get it to draw the rainbow again, but this time slightly larger. If we do this every frame or two, the logo will appear to get nearer and nearer to us. Since the color we draw with is a function of distance we are drawing at this has the added bonus of also gradually changing the colours each instance of the title is drawn with: creating the cascading effect, as opposed to a rainbow with static colors.

```
.SBTTL LOGO - LOGO RAINBOW ; APPROACHING LOGO PROCESS
LOGPRO:
; THE LOGO RAINBOW CONSISTS OF 19 ITERATIONS OF THE TEMPEST
; LOGO IN VARYING COLORS. EACH CALL TO THIS ROUTINE MOVES
; THE LOGO RAINBOW GRADUALLY CLOSER TO THE VIEWER.

; HERE WE SET 'TEMPEST' AS ADDRESS OF PICTURE TO DRAW IN
; IN A AND X REGISTERS, AND THEN DRAW THE FULL RAINBOW WITH IT.
LDAH VORLIT+1      ; LOW BYTE OF VORLIT ADDRESS GOES IN A
LXL VORLIT         ; HI BYTE OF VORLIT ADDRESS GOES IN X
JSR SCARNG         ; DRAW RAINBOW OF LOGO


; CALCULATE POSITION OF RAINBOW FOR NEXT TIME.
; NEARY IS THE NEAREST POINT DEPTH-WISE TO DRAW THE LOGO.
; FARY IS THE FARDEST POINT DEPTH-WISE TO DRAW THE LOGO.
LDA NEARY          ; PUT NEARY IN A
CMP I,30            ; HAS IT REACHED 48 (THE NEAREST POINT) YET?
IFCS               ; IF NOT..
SBC I,1              ; NO. BRING IT CLOSER BY 1 PIXEL.
STA NEARY          ; STORE NEW VALUE IN NEARY
ENDIF
; CHECK IF WE'VE REACHED THE FINAL PHASE WHERE THE
; RAINBOW REDUCES IN DEPTH.
CMP I,80            ; HAS THE FAR POINT REACHED ITS LIMIT OF 128?
IFCC               ; IF SO BRING FAR PT. CLOSER
LDA FARY            ; BY LOADING FARY TO A
SEC                ; SET THE CARRY
SBC I,1              ; REDUCE IT BY 1
CMP NEARY           ; IS IT THE SAME AS THE NEAR POINT YET?
IFCC               ; IF SO SET NEARY AND FARY TO BE THE SAME BY..
LDA NEARY          ; LOADING NEARY TO A
ENDIF
STA FARY            ; STORE A AS THE NEW VALUE FOR FARY
ENDIFF
RTS                ; WE'RE DONE FOR THIS ROUND, RETURN.
```


player charges



Cursor Firing Charges, or as we like to call them: Bullets.

For some reason, the player's bullets in *TEMPEST*, are referred to as 'charges'. When fired they run along the rail the player occupies, into the well in the hope of striking an unwitting alien. Let's take a look at how we get these charges on to the screen.



The bullet as drawn by DIARA2 opposite.

```

DIARA2: ICVEC
CSTAT PSHCTR
SCDOT 0,0
SCDOT 7,0
SCDOT 5,5
SCDOT 0,7
SCDOT -5,5
SCDOT -7,0
SCDOT -5,-5
SCDOT 0,-7
SCDOT 5,-5
CSTAT YELLOW
SCDOT OF,0
SCDOT OB,OB
SCDOT O,OF
SCDOT -OB,OB
SCDOT -OB,O
SCDOT -OB,-OB
SCDOT O,-OB
SCDOT OB,-OB
RTSL

```

We draw the charges at a given point in time by keeping a list of their depth in the well in CHARY (short for CHARGES Y POSITION). This means that drawing them is as simple as looping through this list and drawing any entry that has a non-zero depth value.

```

.SBTTL DISPLAY-CHARGES
; THIS ROUTINE WILL LOOP THROUGH ALL THE CHARGES
; AND DRAW ANY OF THEM THAT ARE ACTIVE.

DSPCHG:
    LDX I,NCHARG-1 ; GET THE NUMBER OF CHARGES (IT'S 8).
    STX INDEX1        ; STORE IT AS OUR INDEX.

    ; LOOP THROUGH ALL 8 CHARGES (BULLETS) AND DRAW
    ; ANY THAT ARE ACTIVE.
    BEGIN            ; START OF LOOP
    LDX INDEX1        ; GET CURRENT INDEX AND USE IT TO..
    LDA X,CHARY      ; ..GET THE VALUE IN THE CURRENT SLOT.
    IFNE              ; IS IT ACTIVE?
    STA PYL          ; IF YES, SCAPIC EXPECTS THE DEPTH IN PYL.
    LDY X,CHARL1      ; GET THE LINE THE BULLET IS ON.
    LDA I,PTCURS      ; GET THE PICTURE IN DIARA2.
    JSR SCAPIC        ; DRAW THE BULLET USING THE VALUES ABOVE.
    ENDIF
    DEC INDEX1        ; DECREMENT OUR INDEX..
    MIEND             ; AND LOOP AGAIN IF STILL NOT ZERO.

    ; SET THE COLOR OF THE CENTRE OF THE BULLET TO INDICATE
    ; HOW MANY SHOTS ARE AVAILABLE TO THE PLAYER.
    LDY I,ZYELLO      ; BY DEFAULT, YELLOW INDICATES PLENTY.
    LDA CHACOU        ; GET THE NO. OF BULLETS CURRENTLY IN PLAY.
    CMP I,NPCHARG-2   ; IF BETWEEN 6 AND 8..
    IFCS              ; THEN IT'S 'LOW' SO..
    LDY I,ZBLUE        ; SET COLOR TO BLUE ('LOW').
    CMP I,NPCHARG     ; IF IT'S 8...

```

```

IFCS          ; THEN THERE ARE NO MORE LEFT..
LDY I,ZRED   ; SET COLOR TO RED ('OUT').
ENDIF
ENDIF
; SET THE COLOR WE CHOSE FOR CENTER OF BULLET.
STY COLPOR+PSHCTR ; NOTICE PSHCTR IS REF'D BY DIARA2 ABOVE.
RTS

```

There is a little-noticed subtlety in how we draw the bullets in the code above. Since there are eight bullets available to the player at any one time, we colour-code the bullets according to how many are currently in play. If all eight bullets are currently in play, we will colour them red. If there are between 5 and 7, we colour them blue. Otherwise we colour them all yellow. Personally I'd never realized this was happening until I read the code!




Figure .1: Different colours of the bullets depending on how many shots are left.

The actual drawing to the screen of each bullet is worth pausing over. First of all, what do we want to do? We want to draw each bullet a little further along the line it was originally fired on. This means that we have to pick a position that is a combination of horizontal (x), vertical (z), and depth (y) points and draw the bullet centred at that point.

You may have noticed that we were using the list of 'depth' points (CHARY) to decide whether or not to draw a bullet in the first place, so we already have the depth value we want to draw at stored in PYL. This leaves the vertical and horizontal positions. To retrieve this for a particular bullet we used a list called CHARL1 that stores the line that each bullet belongs to. Armed with this 'line number' in the Y register the first part of our drawing routine (SCAPIC), we can use it to get the appropriate horizontal and vertical position for that line from a pair of special tables. Now that we have all three points we can draw the bullet in its position.

```

;FUNCTION: DISPLAY A PICTURE CENTERED BETWEEN 2 POINTS AND SCALED
;          DOWN ACCORDING TO ITS DEPTH
;INPUT: X = INDEX INTO LINEX,Z OF 1ST PT'S X & Z WC WORDS
;       Y = INDEX INTO LINEX,Z OF 2ND PT'S X & Z WC WORDS
;       COLOR=COLOR OF OBJECT
;       PYL = Y WC COORD FOR BOTH PTS.
;       ACC = CODE FOR PICTURE TO DISPLAY (INDEX INTO PICLO)
SCAPIC:
    STA OBJIND      ; STORE THE PICTURE INDEX IN OBJIND
    LDA Y,LINEXM    ; GET THE X COORD. OF MIDWAY PT.
    STA PXL         ; STORE IN PXL
    LDA Y,LINEZM    ; GET THE Z COORD OF MIDWAYPT.
    STA PZL         ; STORE IN PZL

    ; USE PXL,PYL, AND PZL TO CALCULATE THE CENTRE POINT FOR
    ; THE BULLET ON THE SCREEN. THE RESULT IS STORED IN
    ; SXL AND SZL. THEN MOVE THE BEAM TO THE DESIRED POSITION
    ; BY ADDING SOME OPCODE INSTRUCTION TO THE VECTOR GENERATOR
    ; LIST.
    JSR WORSCR      ; CALCULATE SXL AND SZL.
    LDX I,SXL       ; GET THE CALCULATED CENTRE PT.
    JSR VGYAB1      ; DRAW A BLANK VECTOR TO IT SO WE START THERE.




    ; CALCULATE THE SCALE AND ADD IT AS AN OPCODE TO THE VGLIST.
    LDA I,0          ; VGY IS OUR INDEX INTO THE VGLIST..
    STA VGY         ; ..SO SET IT TO 0 SO WE'RE AT THE START.
    JSR CASCAL      ; CALCULATE SCALE

    ; CALCULATE THE BRIGHTNESS AND ADD IT AS AN OPCODE TO THE VGLIST.
    LDA BFACTR      ; GET THE BRIGHTNESS FACTOR.
    EOR I,7
    ASL
    CMP I,OA
    IFCC
    LDA I,OA
    ENDIF
    ASL
    ASL
    ASL
    ASL
    STA NY,VGLIST   ; STORE THE BRIGHTNESS BYTE
    INY             ; MOVE TO NEXT BYTE OF OPCODE
    LDA I,60          ; '60' IS THE ID OF THE BRIGHTNESS OPCODE.
    STA NY,VGLIST   ; STORE THE ID IN THE OPCODE LIST.
    INY             ; INCREMENT THE INDEX.
    STY VGY         ; STORE IT.




    ; ADD THE OPCODE TO THE VGLIST FOR DRAWING THE BULLET.
    LDY OBJIND      ; GET THE INDEX OF THE BULLET PIC.
    LDX Y,PICHI     ; GET THE HI BYTE OF THE BULLET PIC.
    LDA Y,PICLO     ; GET THE LO BYTE OF THE BULLET PIC.
    LDY VGY         ; GET THE CURRENT INDEX IN THE OPCODE LIST.
    JMP VGADD3      ; DRAW PIC AT PT.

```

So the result of a typical visit to the SCAPIC routine is to generate a series of Op-Codes that are added to the Vector Generator list. The first set is generated by calling VGYAB1 and is responsible for moving the beam to the desired position on screen.

Vector Commands	Command Meaning	Result on Screen
7100	Set Scale: 00	
8040	Move to Center	
1F2AE4AB	Draw Relative Vector:	

We follow that by setting the scale again, then the brightness, and finally drawing the bullet itself.

Vector Commands	Command Meaning	Result on Screen
7308	Set Scale: 308	
680A	New Color: Not Set Intensity: 0A	
AFA7	Jump to Routine: DIARA2	

1880 10101 EJD
2

